

AMSTRAD

TECHNIQUES DE PROGRAMMATION DE JEUX

GEORGES FAGOT - BARRALY



AMSTRAD
TECHNIQUES DE PROGRAMMATION
DES JEUX

DANS LA MÊME COLLECTION

Amstrad jeux d'action, P. Monsaut
Amstrad premiers programmes, R. Zaks
Amstrad 56 programmes, S.R. Trost
Amstrad exploré, J. Braga
Amstrad programmation en assembleur, G. Fagot-Barraly
Amstrad guide du graphisme, J. Wynford
Amstrad CP/M 2.2, A. d'Hardancourt
Amstrad astrologie, numérologie, biorythmes, P. Bourgault
Amstrad graphisme en trois dimensions, T. Lachant-Robert
Amstrad Multiplan, Amstrad
Amstrad CP/M plus, A. d'Hardancourt
Amstrad Astrocalc, G. Blanc/P. Destrebecq
Amstrad gagnez aux courses, J.-C. Despoine
Amstrad créer de nouvelles instructions, J.-C. Despoine
Amstrad Locoscript, B. Le Dù
Amstrad mise au point des programmes BASIC, C. Vivier/Y. Jacob
Amstrad routines en assembleur, J.-C. Despoine
Amstrad mieux programmer en assembleur, T. Lachant-Robert
Amstrad jeux de réflexion, G. Fagot-Barraly
Amstrad jeux en assembleur, E. Ravis
Amstrad Logo, A. d'Hardancourt (à paraître)
Amstrad programmes en langage machine, S. Webb (à paraître)
Amstrad guide du DOS, Amstrad (à paraître)
Amstrad introduction à la programmation en assembleur du Z80, A. d'Hardancourt (à paraître)
Amstrad systèmes d'exploitation, Amstrad (à paraître)

Georges FAGOT-BARRALY

AMSTRAD

TECHNIQUES DE PROGRAMMATION
DES JEUX



Paris • Berkeley • Düsseldorf

Sybex n'est lié à aucun constructeur.

Tous les efforts ont été faits pour fournir dans ce livre une information complète et exacte. Néanmoins, SYBEX n'assume de responsabilités, ni pour son utilisation, ni pour les contrefaçons de brevets ou atteintes aux droits de tierces personnes qui pourraient résulter de cette utilisation.

Copyright © Sybex 1986.

Tous droits réservés. Toute reproduction, même partielle, par quelque procédé que ce soit, est interdite sans autorisation préalable. Une copie par xérographie, photographie, film, bande magnétique ou autre, constitue une contrefaçon passible des peines prévues par la loi sur la protection des droits d'auteur.

ISBN : 2-7361-0208-8

*A Vincent, pour son aide dans la mise au point
des programmes de ce livre.*

S O M M A I R E

| | |
|---------------------------------------|-----|
| INTRODUCTION | 7 |
| APOCALYPSE, NO ? | 11 |
| BRANCHÉES | 21 |
| ALDO L'ÉPICIER | 31 |
| ELLES DANSENT, MARIE | 43 |
| L'ALBATROS | 53 |
| NUAGES ET SOUCOUPES | 63 |
| POKE-MAN | 73 |
| LE LANCE-MISSILES | 83 |
| L'ÉCUREUIL ET LES LUTINS | 93 |
| TIR SUR CIBLES | 103 |
| LE POISSON PILOTE | 113 |
| LA VIE DE CHÂTEAU | 123 |
| LE BASSET, LE HÉRON ET LE LAPIN | 133 |
| VOL DE NUIT | 141 |
| LE COMBAT DES CHEFS | 151 |

INTRODUCTION

Ce livre regroupe quinze programmes de jeux d'action écrits pour les micro-ordinateurs AMSTRAD. Ces ordinateurs possèdent d'exceptionnelles caractéristiques graphiques et sonores et ont des vitesses d'exécution tout à fait honorables. Dès lors, efficacement programmés, ils ont toutes les qualités pour devenir nos partenaires privilégiés dans le seul domaine où l'informatique arrive à nous faire oublier qu'elle est la fille rieuse des deux monuments de froideur que sont l'électronique et les mathématiques. Je veux parler du domaine de l'ordinateur joueur ¹, de ce monde dans lequel les programmeurs réussissent le tour de force de nous faire croire que l'on est assis devant une machine qui n'est pas tenue de suivre la route qui lui a été tracée. Dans ce monde, l'ordinateur ne fait plus peur et certaines questions n'appellent pas la réponse stéréotypée que l'on a l'habitude d'entendre. Est-ce qu'il arrive à ces machines de prendre des initiatives ? Peuvent-elles, dans leur précipitation, commettre des erreurs ? Sont-elles capables de commenter autrement que par "vrai" ou "faux", comme dans les programmes que l'on se plaît à appeler "pédagogiques", nos réponses ? Bref, sont-elles capables de faire montre de fraîcheur et de spontanéité ? Pour ma part, je pense que oui, mais je n'empêche personne de penser que, chez moi, le cœur l'emporte sur la raison.

Revenons au sujet de cet ouvrage. Quinze programmes donc, tous relatifs à des jeux extrêmement animés. De thèmes résolument variés et nouveaux, ils exigeront de la part de leurs utilisateurs des réflexes et une vigilance à toute épreuve. Un haut niveau dans le domaine du *self-control* sera le bienvenu. Pas d'inquiétude exagérée toutefois. Il n'y a, dans ce livre, aucun programme dans lequel l'ordinateur soit imbattable. De plus, la plupart du temps, le choix du niveau de difficulté est donné au joueur. Celui-ci pourra donc, quel que soit son âge ou son habileté, trouver "chaussure à son pied".

Le lecteur ne manquera pas de s'apercevoir que tous les programmes sont écrits sur le même modèle, respectant tous un certain nombre de règles. Cette homogénéité n'est pas due à une quelconque manie de l'auteur. Elle tient au fait que la conception de ces jeux s'est réalisée dans un plan d'ensemble et qu'il était alors aisé de définir à l'avance les contraintes que devaient respecter les différents programmes.

Voici la liste de ces règles et leurs raisons d'être :

1. Chaque programme fonctionne sur le mode 1. L'utilisation de cette très bonne résolution associée à l'excellence du BASIC Amstrad confère aux jeux présentés dans ce livre une précision graphique et une rapidité d'action que bien peu d'ordinateurs sont capables aujourd'hui d'égaliser. Est-il besoin de dire que les couleurs ont été choisies de telle sorte que les teintes correspondantes sur un moniteur monochrome soient nettement discernables ?
2. Les premiers numéros de lignes ont toujours été attribués à la boucle principale. Cela s'explique d'une part parce qu'il est ainsi plus facile de la retrouver dans le listing et d'autre part parce que c'est la partie du programme qui doit être exécutée le plus rapidement possible. Sans entrer dans les détails, disons que l'ordinateur, réalisant une commande telle que GOTO, retrouve une ligne d'autant plus vite qu'elle est écrite en début de programme.
3. Les variables numériques, à moins qu'elles ne dépassent un certain seuil, sont toujours déclarées comme entières (au moyen de l'instruction DEFINT), et l'on préfère utiliser parfois plusieurs d'entre elles plutôt qu'un tableau. Là encore, la raison est que l'on obtient ainsi un gain de temps appréciable.
4. Tous les listings font apparaître une abondante documentation. Chaque sous-programme étant précédé d'un commentaire, la faculté est offerte aux adeptes de la programmation de retrouver sans difficulté les phases de jeu qu'ils souhaitent analyser ou modifier. Notons qu'aucun branchement n'est "réalisé sur une REM" et que, de ce fait, toutes les lignes dont le numéro est suivi d'une apostrophe peuvent très bien ne pas être tapées au clavier.
5. Les programmes occupent tous à peu près le même volume mémoire 3 ou 4 blocs. Cet ordre de grandeur, raisonnable quant au nombre de lignes qu'il faudra entrer dans l'ordinateur, permet de ne pas sacrifier ce qui est essentiel dans un jeu d'action : le décor. Un effort tout particulier a été fait pour rendre la présentation de ces jeux la plus attrayante possible. Cela prend bien sûr de la place dans les programmes, mais le résultat est à la hauteur de la peine que l'on s'est donnée.
6. Chaque listing est accompagné, c'est la moindre des choses, d'un mode d'emploi détaillé. S'y ajoutent des indications concernant la manière dont les jeux se terminent ainsi que les modifications que l'on peut faire sans problème. On trouve aussi la façon de

s'y prendre pour utiliser les touches du clavier à la place de la manette. Enfin, un organigramme décrivant la structure de la partie principale du programme est toujours donné.

7. Le dernier point a son importance : le noir a été choisi comme couleur de fond pour tous les jeux ; c'est plus reposant pour la vue des utilisateurs et cela accroît la durée de vie des moniteurs.

Voilà. Et maintenant, place aux jeux.

1. Du même auteur : *Jeux de réflexion sur Amstrad*.

APOCALYPSE, NO ?

On dispose, quand le jeu démarre, de six hélicoptères dessinés en bas et à gauche de l'écran. Le premier hélicoptère prend son envol, atteint son altitude de croisière et attend qu'on lui donne le signal de départ. De l'autre côté de l'écran est apparue une soucoupe bleue qui change constamment de position ; elle va servir de cible. Dès qu'on appuie sur la manette de jeu, l'hélicoptère se dirige de lui-même sur la droite. On peut le faire monter ou descendre en poussant la manette vers le haut ou le bas, il envoie un missile quand on appuie sur le bouton FIRE. Le but du jeu est de détruire le maximum de soucoupes bleues tout en évitant les obstacles formés par les immeubles ou les mines qui flottent en plein ciel. Attention en particulier aux mines rouges dont la mise à feu se déclenche si un objet les rase d'un peu trop près. Lorsque l'on a traversé l'écran sans avoir pu toucher la soucoupe, cette dernière essaie de prendre sa revanche. On ne peut alors rien faire d'autre que de la regarder se diriger sur la gauche, bombarder dans un bruit de sirène notre base, et regagner ensuite son territoire. Naturellement, ceci nous aura coûté un hélicoptère sauf... si la soucoupe a elle-même rencontré durant son trajet une mine ou le toit d'un immeuble.

FIN DU JEU

L'ordinateur affiche le message "vous avez perdu" quand les six hélicoptères ont été utilisés ou quand le nombre de missiles est passé à zéro (50 de ceux-ci sont alloués au début du jeu).

On est déclaré vainqueur si l'on arrive à détruire 10 soucoupes bleues.

Le niveau de difficulté détermine le nombre de mines qui apparaissent dans le ciel.

MODIFICATIONS POSSIBLES

| | Instructions BASIC | N° de ligne |
|---|--------------------|-----------------------|
| Nombre de tirs | C2 = 50 | 11150 |
| Nombre de soucoupes à détruire | IF C1 = 10 | 9110 |
| Nombre de mines au départ | FOR I = 1 TO 2*N | 11070 |
| Nombre de mines qui apparaissent en cours de partie | FOR I = 1 TO N | 1070, 4070 et 6100 |
| Proportion de mines rouges | IF RND > 0.7 | 11080 |

JEU SANS MANETTE_____

Pour faire monter ou descendre l'hélicoptère, il faudra appuyer sur les flèches verticales. Le tir sera déclenché par appui de la barre d'espace.

- Dans la ligne 1020, remplacer

```
YH=YH+(JOY(0)=1-(JOY(0)=2))
```

par :

```
W=INKEY$: YH=YH+(W=CHR$(240))-(W=CHR$(241))
```

- Dans la ligne 1050, remplacer

```
IF JOY(0) > 15
```

par :

```
IF W=CHR$(32)
```

- Dans la ligne 3030, remplacer

```
U="Bougez la manette"
```

par :

```
U="Appuyez sur COPY"
```


- Dans la ligne 3050, remplacer

```
IF JOY(0)
```

par :

```
IF INKEY$=CHR$(224)
```

LE CŒUR DU PROGRAMME



```

940 '
950 '
960 '      APOCALYPSE , NO ?
970 '
980 '
990 GOTO 10000
1000 PAPER 0:G=0:FOR XH=1 TO 18:PEN 1:LOCATE 2*X
H-1,YH+1:PRINT " "+R(-C)
1010 SOUND 1,400,3:SOUND 2,410,2:FOR I=1 TO 5:NE
XT:Y0=YH:X0=XH
1020 YH=YH+(JOY(0)=1)-(JOY(0)=2):YH=YH-(YH=-1)
1030 C=NOT C:LOCATE 2*X0,Y0+1:PRINT "      ":LOCAT
E 2*XH+1,YH+1:PRINT R(-C)
1040 C=NOT C:SOUND 1,400,3:SOUND 2,410,2:IF C(XH
,YH) OR C(XH+1,YH) THEN 7000
1050 IF JOY(0)>15 THEN 9000 ELSE IF G THEN FOR I
=1 TO 50:NEXT ELSE GOSUB 2000
1060 NEXT XH:LOCATE 37,YH+1:PRINT "      ":IF G=0
THEN 4000
1070 FOR I=1 TO NV:GOSUB 11080:NEXT:GOTO 3040
1970 '
1980 '      Mouvements de la soucoupe
1990 '
2000 C(XS,YS)=0:PEN 3:LOCATE 2*XS+1,YS+1:PRINT "
":XS=18-(XS=18)
2010 IF RND<0.8 THEN 2020 ELSE YS=YS+INT(RND*3)-
1:YS=YS+(YS=14)-(YS=0)
2020 LOCATE 2*XS+1,YS+1:PRINT S:C(XS,YS)=2:RETUR
N
2970 '
2980 '      Attente ( helico en l'air )
2990 '
3000 LOCATE 2*XS+1,YS+1:PRINT "      ":C(XS,YS)=0:PE
N 1:YH=INT(RND*10)+2
3010 FOR I=E TO YH STEP -1:FOR K=1 TO 2:LOCATE 1
,I+1:PRINT R(-C):SOUND 1,400,4
3020 SOUND 2,409,4:FOR J=1 TO 70:NEXT:C=NOT C:NE
XT
3030 LOCATE 1,I+1:PRINT "      ":NEXT:U="BOUGEZ LA
MANETTE"
3040 XS=18:YS=INT(RND*10)+2:LOCATE 1,YH+1:PRINT
R(-C)
3050 IF JOY(0) THEN PAPER 2:PEN 2:LOCATE 1,23:PR
INT U:GOTO 1000

```

```

3060 PEN -2*C+1:PAPER 2:LOCATE 1,23:PRINT U:PEN
1:PAPER 0
3070 LOCATE 1,YH+1:PRINT R(-C):C=NOT C:GOSUB 200
0:GOTO 3050
3970 '
3980 '      Trajet de la soucoupe
3990 '
4000 C(XS,YS)=0:PEN 3:FOR I=XS TO 2 STEP -1:LOCA
TE 2*I+1,YS+1:PRINT S
4010 SOUND 1,300,5:SOUND 2,310,4:YS1=YS+(1-INT(R
ND*3))*(RND<0.4)
4020 YS1=YS1-(YS1=-1):F=C(I,YS):LOCATE 2*I+1,YS+
1:PRINT " "
4030 IF F THEN 6000 ELSE YS=YS1:NEXT:GOTO 5000
4040 PEN 3:FOR I=1 TO 18:LOCATE 2*I+1,YS+1:PRINT
S:SOUND 1,300,5:SOUND 2,310,4
4050 YS1=YS+(1-INT(RND*3))*(RND<0.4):YS1=YS1-(YS
1=-1)
4060 F=C(I,YS):LOCATE 2*I+1,YS+1:PRINT " ":IF F
THEN 6000 ELSE YS=YS1
4070 NEXT:G=1:FOR I=1 TO NV:GOSUB 11080:NEXT:GOT
O 3040
4970 '
4980 '      Bombarde ment
4990 '
5000 LOCATE 3,YS+1:PRINT S:PEN 2:FOR I=YS+2 TO E
:LOCATE 3,I+1:PRINT CHR$(248)
5010 SOUND 1,150-3*I,5:FOR J=1 TO 20:NEXT:LOCATE
3,I+1:PRINT " ":NEXT
5020 IF E=21 THEN 4040 ELSE E=E+1:FOR I=1 TO 8:P
EN I:LOCATE 1,E+1
5030 PRINT R(C+1):SOUND 1,300+10*I,5:SOUND 2,305
+10*I,4
5040 FOR J=1 TO 20:NEXT:C=NOT C:NEXT:GOTO 4040
5970 '
5980 '      Explosion soucoupe
5990 '
6000 IF F>4 THEN 6050 ELSE FOR J=1 TO 8:PEN INT(
RND*3)+1:LOCATE 2*I+1,YS+1
6010 PRINT S:SOUND 1,200+20*J,8:FOR K=1 TO 100:N
EXT:NEXT
6020 LOCATE 2*I+1,YS+1:PRINT " ":C(I,YS)=0:YS=Y
S-1

```

```

6030 IF C(I,YS)>2 AND C(I,YS)<5 THEN 6020 ELSE G
=1
6040 FOR I=1 TO NV:GOSUB 11080:NEXT:YH=INT(RND*1
0)+2:GOTO 3040
6050 PEN 1:FOR L=A(F)-1 TO A(F)+1:FOR M=B(F)-1 T
O B(F)+1:LOCATE 2*L+1,M+1
6060 PRINT ".":SOUND 1,200,1:NEXT:NEXT:FOR J=1 T
O 8:LOCATE 2*I+1,YS+1
6070 PEN INT(RND*3)+1:PRINT S:SOUND 1,200+20*J,8
:FOR K=1 TO 100:NEXT:NEXT
6080 LOCATE 2*I+1,YS+1:PRINT "  ":FOR L=A(F)-1 T
O A(F)+1:FOR M=B(F)-1 TO B(F)+1
6090 LOCATE 2*L+1,M+1:PRINT "  ":C(L,M)=0:NEXT:N
EXT
6100 G=1:FOR I=1 TO NV:GOSUB 11080:NEXT:GOTO 304
0
6970 '
6980 '      Helico contre obstacle
6990 '
7000 FO=C(XH,YH):F1=C(XH+1,YH):IF FO>4 OR F1>4 T
HEN 8000
7010 FOR I=1 TO 12:PEN C+2:LOCATE 2*XH+1,YH+1:PR
INT R(C+1):SOUND 1,200+10*I,9
7020 IF SQ(1)>127 THEN 7020 ELSE SOUND 2,202+10*
I,5:C=NOT C:NEXT
7030 LOCATE 2*XH+1,YH+1:PRINT "  ":C(XH,YH)=0:
C(XH+1,YH)=0:YH=YH-1
7040 IF (C(XH,YH)>2 AND C(XH,YH)<5) OR (C(XH+1,Y
H)>2 AND C(XH+1,YH)<5) THEN 7030
7050 E=E+1:IF E=22 THEN 12000 ELSE 3000
7970 '
7980 '      Helico detruit par une mine
7990 '
8000 H=-FO*(FO>=F1)-F1*(FO<F1):PEN 1:FOR L=A(H)-
1 TO A(H)+1
8010 FOR M=B(H)-1 TO B(H)+1:LOCATE 2*L+1,M+1:PRI
NT ".":SOUND 1,200,1
8020 NEXT:NEXT:FOR M=1 TO 10:LOCATE 2*XH+1,YH+1:
PRINT R(C+1):SOUND 1,300,8
8030 IF SQ(1)>127 THEN 8030 ELSE SOUND 2,305,3:P
EN INT(RND*3)+1
8040 NEXT:FOR L=A(H)-1 TO A(H)+1:FOR M=B(H)-1 TO
B(H)+1

```

```

8050 LOCATE 2*L+1,M+1:PRINT "   ":C(L,M)=0:NEXT:N
EXT:E=E+1:LOCATE 2*XH+1,YH+1
8060 PRINT "   ":C(XH,YH)=0:C(XH+1,YH)=0:IF E=2
2 THEN 12000 ELSE 3000
8970 '
8980 '      Tir
8990 '
9000 PEN 3:LOCATE 2*XH+5,YH+1:H=0:J=YH:FOR I=XH+
2 TO 19
9010 IF C(I,YH)=0 THEN PRINT "--":SOUND 1,100,1
:NEXT:H=1
9020 LOCATE 2*XH+5,YH+1:PRINT SPACE$(2*(I-XH-2))
:IF H THEN 9110 ELSE F=C(I,YH)
9030 IF F<5 THEN 9070 ELSE FOR L=A(F)-1 TO A(F)+
1:FOR M=B(F)-1 TO B(F)+1
9040 PEN 1:LOCATE 2*L+1,M+1:PRINT ".":SOUND 1,20
0,1:NEXT:NEXT
9050 FOR L=A(F)-1 TO A(F)+1:FOR M=B(F)-1 TO B(F)
+1:LOCATE 2*L+1,M+1
9060 PRINT "   ":C(L,M)=0:NEXT:NEXT:GOTO 9110
9070 IF F<>2 THEN 9100 ELSE FOR I=1 TO 12:PEN I:
LOCATE 2*XS+1,YS+1
9080 PRINT S:SOUND 1,400+5*I,3:NEXT:C(XS,YS)=0:G
=1
9090 C1=C1+1:PEN 3:LOCATE 21,25:PRINT C1::GOTO 9
110
9100 IF F=1 THEN LOCATE 2*I+1,J+1:PRINT "   ":C(I
,J)=0
9110 IF C1=10 THEN 12000 ELSE C2=C2-1:PEN 1:LOCA
TE 36,25:PRINT C2:
9120 IF INKEY$<>" " THEN 9120 ELSE IF C2 THEN 106
0
9130 FOR I=1 TO 15:PEN I:LOCATE 26,25:PRINT "MIS
SILES : 0":SOUND 1,100+10*I,15
9140 IF SQ(1)>127 THEN 9140 ELSE SOUND 2,110+10*
I,8:NEXT:GOTO 12000
9970 '
9980 '      Initialisation
9990 '
10000 MODE 1:INK 0,0:INK 1,16:INK 2,9:INK 3,23:P
APER 0:BORDER 0:SPEED KEY 3,3
10010 SYMBOL 244,0,32,65,70,45,31,34,0:SYMBOL 24
5,0,4,130,98,180,248,68

```

```

10020 SYMBOL 246,0,24,24,24,24,24,24,0:SYMBOL 24
7,0,0,60,195,60,0,0,0
10030 SYMBOL 248,0,0,9,6,6,6,0,0:SYMBOL 249,0,0,
0,3,255,3,0,0
10040 SYMBOL 250,7,4,255,240,240,240,255,0:SYMBOL
L 251,255,0,240,48,255,240,192,0
10050 SYMBOL 252,0,0,0,192,127,192,0,0:SYMBOL 25
3,63,0,0,3,255,3,0,0
10060 SYMBOL 254,252,4,255,240,240,240,255,0:SYM
BOL 255,0,0,240,48,255,240,192,0
10070 DEFINT B-P,X-Z:DEFSTR R-W:DIM A(50),B(50),
C(19,21)
10080 R(0)=CHR$(252)+CHR$(253)+CHR$(254)+CHR$(25
5):S=CHR$(244)+CHR$(245)
10090 R(1)=CHR$(252)+CHR$(249)+CHR$(250)+CHR$(25
1):T=CHR$(214)+CHR$(215)
10100 PEN 1:LOCATE 1,15:PRINT "CHOISISSEZ LA DIF
FICULTE ( DE 1 A 3 )"
10110 U=INKEY$:IF U="" THEN 10110 ELSE NV=VAL(U)
:IF NV<1 OR NV>3 THEN 10110
10970 '
10980 '      Decor
10990 '
11000 CLS:FOR I=3 TO 17:PAPER 0:PEN 1:J=INT(RND*
7)+14:LOCATE 2*I+1,J+1
11010 PRINT T:C(I,J)=3:PAPER 3:PEN 0:FOR K=J+1 T
O 21
11020 LOCATE 2*I+1,K+1:IF RND<0.7 THEN PRINT CHR
$(246); ELSE PRINT " ";
11030 IF RND<0.7 THEN PRINT CHR$(246) ELSE PRINT
" "
11040 C(I,K)=4:NEXT:NEXT:PAPER 2:WINDOW 1,40,23,
23:CLS:WINDOW 1,40,1,25
11050 PAPER 0:PEN 3:LOCATE 1,25:PRINT "SOUCOUPES
TOUCHEES : 0";
11060 PEN 1:LOCATE 26,25:PRINT "MISSILES : 50":;
LOCATE 1,17:FOR I=1 TO 6
11070 C=NOT C:PRINT R(C+1):NEXT:D=5:FOR I=1 TO 2
*NV
11080 IF RND>0.7 THEN 11120 ELSE J=INT(RND*15)+3
:K=INT(RND*13)+1
11090 IF C(J,K) THEN 11150 ELSE PEN 2:LOCATE 2*J
+2,K+1

```

```

11100 PRINT CHR$(247):C(J,K)=1:GOTO 11150
11120 J=INT(RND*13)+4:K=INT(RND*11)+2:PEN 1:FOR
L=J-1 TO J+1:FOR M=K-1 TO K+1
11130 IF C(L,M) THEN 11150 ELSE NEXT:NEXT:A(D)=J
:B(D)=K:FOR L=J-1 TO J+1
11140 FOR M=K-1 TO K+1:C(L,M)=D:NEXT:NEXT:LOCATE
2*J+2,K+1:PRINT CHR$(247):D=D+1
11150 IF G THEN RETURN ELSE NEXT I:E=16:C2=50:GO
TO 3000
11970 '
11980 '      Nouvelle partie
11990 '
12000 WINDOW 10,26,1,3:PAPER 2:CLS:PEN 3:LOCATE
2,2
12010 IF C1=10 THEN PRINT "VOUS AVEZ GAGNE" ELSE
PRINT "VOUS AVEZ PERDU"
12020 FOR I=1 TO 400:NEXT:WINDOW 1,40,1,25:PAPE
R 0
12030 U=INKEY$:IF U="0" THEN RUN ELSE IF U="N" T
HEN SPEED KEY 10,2:CLS:END
12040 PEN INT(RND*3)+1:FOR I=1 TO 300:NEXT:LOCAT
E 8,8
12050 PRINT "UNE AUTRE PARTIE ? ( TAPEZ 0 ou N )
":GOTO 12030

```

BRANCHÉES

Trois soucoupes partent du haut de l'écran et se déplacent en convoi. Elles n'ont pas de but précis et se laissent glisser au gré des obstacles qu'elles rencontrent.

1. Elles peuvent tout d'abord se trouver nez à nez avec un missile que l'appui sur la touche FIRE aura fait s'envoler. C'est, bien entendu, le cas le plus ennuyeux pour elles ; elles n'en réchappent pas et laissent la place à d'autres consœurs. Par contre, c'est pour nous le cas le plus intéressant : le but du jeu est en effet d'atteindre le plus possible de soucoupes. Notons que le choix du missile qui sera tiré se fait en poussant la manette à gauche, en haut ou à droite.
2. Elles peuvent aussi se présenter au bord de l'écran : là, pas de problème, elles descendent d'un cran et repartent dans la direction opposée.
3. Il leur arrive aussi de rencontrer des nuages clairs. Aucune importance pour elles, elles se contentent de les contourner en perdant un peu d'altitude.
4. On les voit parfois devant les gros nuages blancs. Enfin, on les voit... pas très longtemps, car elles ont l'habitude de se cacher dans ce genre d'endroit. Elles peuvent soit en ressortir immédiatement, soit s'y tenir tranquilles pendant quelques instants. Un autre convoi prend alors la relève, peut-être en sortant d'un nuage blanc dans lequel il s'était caché un peu auparavant.
5. Dernière chose enfin, les nuages bleu foncé ; ce sont ceux qu'elles préfèrent. Ils ont la particularité de se réduire en pluie quand les soucoupes y pénètrent. Et cette pluie va remplir des réservoirs qui, au bout de quelques averses, vont déborder dans... les aires de lancement des missiles, les mettant hors d'état. C'est désagréable, non ?

Ah, n'oublions pas : les soucoupes ont un sixième sens et il leur arrive souvent, au moment de traverser les lignes de tir, de changer brusquement d'avis et de repartir en sens contraire. Si on se laisse un peu trop souvent prendre à ce jeu, le stock de missiles risque fort de n'être pas suffisant pour résister longtemps à ces envahisseuses.

FIN DU JEU_____

Le jeu s'arrête et l'ordinateur nous déclare perdant lorsqu'il ne reste plus aucun missile (50 d'entre eux sont attribués au départ) ou lorsque les trois bases de lancement sont submergées par l'eau qui déborde des barrages.

La partie est gagnée si l'on arrive à toucher 25 soucoupes.

La vitesse de déplacement des soucoupes et leurs hésitations avant de traverser les lignes de tir sont fonction du niveau de difficulté choisi.

MODIFICATIONS POSSIBLES_____

| | Instructions BASIC | N° de ligne |
|-------------------------------|--------------------|-------------|
| Nombre de missiles | MI=50 | 10210 |
| Nombre de soucoupes à toucher | IF MS < 25 | 5030 |

JEU SANS MANETTE_____


On changera la position des missiles en appuyant sur les flèches droite, gauche et haute. La barre d'espacement permettra de déclencher le tir.

- Supprimer totalement la ligne 1090 et la remplacer par les trois lignes suivantes :

```
1090 W=INKEY$: IF W="" THEN 1110
```

```
1095 IF W=CHR$(32) THEN GOSUB 4000
```

```
1096 M1=- (W=CHR$(240))-4*(W=CHR$(242))-8*(W=CHR$(243))
```



```

940 '
950 '
960 '          BRANCHEES
970 '
980 '
990 GOTO 9000
1000 M=0:L=INT(RND*4):XA=-1:XB=-1:XC=-1:IF YA(L)
>7 THEN 6000
1010 N=1:IF L(L)=0 THEN FOR I=1 TO 20:SOUND 1,10
0-3*I,2:NEXT:GOTO 1040
1020 FOR I=1 TO 6:PEN I:FOR K=1 TO 50:NEXT:LOCAT
E X(L(L))+1,Y(L(L)):PRINT R1
1030 LOCATE X(L(L))+1,Y(L(L))+1:PRINT R2:SOUND 1
,200-5*I,8:NEXT
1040 PEN 1:IF YA(L)=8 THEN 1130 ELSE FOR X=X0(L)
TO X1(L) STEP X2(L)
1050 P=A(X+X2(L),YA(L)):IF P THEN 2000
1060 IF XC>-1 THEN LOCATE XC+1,YC+1:PRINT " "
1070 LOCATE X+X2(L)+1,YA(L)+1:PRINT CHR$(255)
1080 XC=XB:YC=YB:XB=XA:YB=YA:YA=YA(L):XA=X+X2(L)
:N=0
1090 M1=JOY(0):IF M1=0 THEN 1110 ELSE IF M1>15 T
HEN GOSUB 4000
1100 IF M1=1 OR M1=4 OR M1=8 THEN GOSUB 3000
1110 IF N THEN RETURN ELSE IF M THEN 5000 ELSE N
EXT X:X1(L)=39-X1(L)
1120 X2(L)=-X2(L):X0(L)=X1(L)-39*X2(L):YA(L)=YA(
L)+1:GOTO 1040
1130 IF XA=39 THEN LOCATE 38,8:PRINT " ":GOTO
1000 ELSE N=1:X0=XA-1
1140 FOR X=X0 TO 35:LOCATE X+2,9:PRINT CHR$(255)
:LOCATE XC+1,YC+1:PRINT " "
1150 XC=XB:YC=YB:XB=XA:YB=YA:XA=X+1:YA=8
1160 GOSUB 1090:IF M THEN 5000 ELSE NEXT:LOCATE
35,9:PRINT " ":GOTO 1000
1970 '
1980 '          Un obstacle
1990 '
2000 IF P<>1 THEN 2040 ELSE IF RND<NV/4 THEN 202
0
2010 IF XC>-1 THEN LOCATE XC+1,YC+1:PRINT " ":GO
TO 1070 ELSE 1070
2020 X2(L)=-X2(L):X1(L)=1-37*(X2(L)=1)

```

```

2030 XO(L)=XA-X2(L):YA(L)=YA(L)+1:GOTO 1040
2040 IF P<>2 THEN 2060 ELSE X2(L)=1+2*(RND<0.5)
2050 X1(L)=1-37*(X2(L)=1):XO(L)=XA-X2(L):YA(L)=Y
A(L)+1:GOTO 1040
2060 N=1:LOCATE XC+1,YC+1:PRINT " ":GOSUB 1090:L
OCATE XB+1,YB+1
2070 PRINT " ":GOSUB 1090:LOCATE XA+1,YA(L)+1:PR
INT " "
2080 XA=XA+2*X2(L):X2(L)=1+2*(RND<0.5)
2090 XO(L)=XA+X2(L):X1(L)=1-37*(X2(L)=1):L(L)=P-
2:GOTO 1000
2970 '
2980 '      Choix du missile
2990 '
3000 IF M(M1)=0 THEN RETURN
3010 M2=10-10*(M1=1)-20*(M1=8):FOR I=10 TO 30 ST
EP 10
3020 LOCATE I,22:IF I=M2 THEN PRINT CHR$(254):GO
TO 3040
3030 IF TEST(16*I-2,62)<>2 THEN PRINT " "
3040 NEXT:RETURN
3970 '
3980 '      Tir
3990 '
4000 SOUND 1,30,30,0,1,1:LOCATE M2,22:PRINT " ":
FOR I=18 TO YA(L)+2 STEP -1
4010 LOCATE M2,I:PRINT CHR$(254):LOCATE M2,I:PRI
NT " ":NEXT
4020 LOCATE M2,22:PRINT CHR$(254):M=0:IF XA+1=M2
OR XB+1=M2 OR XC+1=M2 THEN M=1
4030 MI=MI-1:LOCATE 28,25:PRINT MI::IF MI THEN R
ETURN
4040 FOR I=0 TO 15:LOCATE 29,25:PRINT "O MISSIL
E "":SOUND 1,300+10*I,15
4050 IF SQ(1)>127 THEN 4050 ELSE PEN I:NEXT:W="P
ERDU":GOTO 11000
4970 '
4980 '      Soucoupes touchees
4990 '
5000 FOR I=1 TO 8:PEN I:LOCATE XA+1,YA(L)+1:PRIN
T CHR$(255):LOCATE XB+1,YB+1
5010 PRINT CHR$(255):LOCATE XC+1,YC+1:PRINT CHR$
(255):SOUND 1,100+20*I,4

```

```

5020 IF SQ(1)>127 THEN 5020 ELSE SOUND 2,300-10*
I,3:NEXT:MS=MS+1:LOCATE 21,25
5030 PEN 1:PRINT MS::IF MS<25 THEN N=1:GOTO 1019
0 ELSE W="GAGNE":GOTO 11000
5970 '
5980 '      Nuage de droite
5990 '
6000 YO=INT(RND*3):IF P(YO)=2 THEN 6000 ELSE Y=Y
0+10:YA(L)=Y-1:YB=Y-1:YC=Y-1
6010 N=1:T=CHR$(255)+CHR$(255)+CHR$(255)+" ":FOR
X=34 TO 1 STEP -1:PEN 1
6020 LOCATE X,Y:PRINT T:IF TEST(16*X-20,16*(25-Y
)+8) THEN 7000
6030 XA=X-1:XB=X:XC=X+1:GOSUB 1090:IF M THEN 500
0
6040 NEXT X:LOCATE 1,Y:PRINT " ":GOTO 10190
6970 '
6980 '      Il pleut ...
6990 '
7000 LOCATE X+2,Y:PRINT " ":GOSUB 1090:LOCATE X+
1,Y
7010 PRINT " ":GOSUB 1090:LOCATE X,Y:PRINT " ":K
=3
7020 FOR I=0 TO 7:FOR J=1 TO 7:MOVE 16*X-8*J-16,
16*(25-Y)-4
7030 DRAW 16*X-8*J-16,100,K:K=5-K:NEXT:Y1=64+16*
P(YO)+2*I
7040 X1=16*(X-4)-16*P(YO)-2*I-16:MOVE X1,Y1
7050 DRAWR 64+32*P(YO)+4*I,0,2:NEXT:WINDOW X-4,X
-1,Y+1,19
7060 CLS:WINDOW 1,40,1,25:P(YO)=P(YO)+1:IF P(YO)
=2 THEN 8000
7070 M=0:X=X-7:GOTO 6030
7970 '
7980 '      Et ca deborde
7990 '
8000 PEN 2:LOCATE X+2,22:PRINT S:LOCATE X+2,21:P
RINT S:LOCATE X+2,20:PRINT S
8010 M(-4*(YO=0)-8*(YO=1)-(YO=2))=0:MY=MY+1
8020 IF MY=3 THEN W="PERDU":GOTO 11000
8030 M1=2^INT(RND*4):IF M(M1)=0 THEN 8030 ELSE P
EN 1:GOSUB 3010:GOTO 7070
8970 '

```

```

8980 '          Initialisation
8990 '
9000 MODE 1:INK 0,0:INK 1,16:INK 2,23:INK 3,10:P
APER 0:BORDER 0
9010 SYMBOL 245,3,15,31,63,127,127,255,255:SYMBO
L 246,195,255,255,255,255,255,255,255
9020 SYMBOL 247,192,240,248,252,252,254,255,255:
SYMBOL 248,255,255,127,127,63,31,15,3
9030 SYMBOL 249,255,255,255,255,255,255,255,195:
SYMBOL 250,255,255,254,252,252,248,240,192
9040 SYMBOL 251,54,127,255,255,255,127,127,30:SY
MBOL 252,195,255,255,255,255,255,255,195
9050 SYMBOL 253,232,252,254,255,255,255,254,108:
SYMBOL 254,24,24,24,24,24,24,36,36
9060 SYMBOL 255,0,0,0,56,108,214,0,0:DEFINT A-P,
X-Z:DEFSTR R-W
9070 PEN 2:LOCATE 1,10:PRINT "CHOISISSEZ LA DIFF
ICULTE ( DE 1 A 3 )"
9080 W=INKEY$:IF W="" THEN 9080 ELSE NV=VAL(W):I
F NV<1 OR NV>3 THEN 9080
9090 R1=CHR$(245)+CHR$(246)+CHR$(247):R2=CHR$(24
8)+CHR$(249)+CHR$(250)
9100 R=CHR$(251)+CHR$(252)+CHR$(252)+CHR$(253):S
=CHR$(143)
9110 DIM A(40,20):ENT 1,70,1,1:ENV 1,2,5,1,1,0,7
0
9970 '
9980 '          Decor
9990 '
10000 CLS:PEN 2:FOR I=1 TO 6:READ X,Y:LOCATE X+1
,Y+1:PRINT CHR$(251)+CHR$(253)
10010 A(X,Y)=2:A(X+1,Y)=2:NEXT:DATA 22,0,4,1,2,4
,30,5,4,7,34,6
10020 FOR I=1 TO 4:READ X(I),Y(I):LOCATE X(I)+1,
Y(I):PRINT R1
10030 LOCATE X(I)+1,Y(I)+1:PRINT R2:A(X(I),Y(I))
=2+I:A(X(I),Y(I)-1)=2+I
10040 A(X(I)+2,Y(I))=2+I:A(X(I)+2,Y(I)-1)=2+I:NE
XT:DATA 13,3,23,5,13,7,32,2
10050 LOCATE 38,9:PRINT R1:LOCATE 38,10:PRINT R2
:LOCATE 38,11:PRINT R1
10060 LOCATE 38,12:PRINT R2:PEN 3:FOR I=10 TO 12
:READ X:DATA 4,24,14

```

```

10070 LOCATE X,I:PRINT R:NEXT:FOR I=1 TO 21 STEP
  10:LOCATE I,19:PRINT CHR$(247)
10080 LOCATE I,20:PRINT S+CHR$(247):LOCATE I+8,2
0:PRINT CHR$(245):LOCATE I,21
10090 PRINT S+S+CHR$(247):LOCATE I+7,21:PRINT CH
R$(245)+S:LOCATE I,22
10100 PRINT STRING$(9,S):NEXT:LOCATE 38,15:PRINT
  CHR$(245)+CHR$(247)+CHR$(245)
10110 LOCATE 37,16:PRINT CHR$(245)+STRING$(3,S):
  LOCATE 34,17
10120 PRINT CHR$(245)+CHR$(247)+CHR$(245)+STRING
$(4,S)
10130 LOCATE 31,18:PRINT CHR$(245)+CHR$(247)+CHR
$(245)+STRING$(7,S)
10140 WINDOW 31,40,19,22:PAPER 3:CLS:WINDOW 1,40
,1,25
10150 LOCATE 1,23:PRINT SPACE$(40):PEN 1:PAPER 0
:LOCATE 1,25
10160 PRINT "SOUCOUPES TOUCHEES : 0      50 MISS
ILES";
10170 FOR I=9 TO 29 STEP 10:FOR J=0 TO 7
10180 A(I,J)=1:NEXT:NEXT:FOR L=0 TO 3
10190 YA(L)=0:L(L)=0:X2(L)=1+2*(RND<0.5):X0(L)=-
1-41*(X2(L)=-1)
10200 X1(L)=X0(L)+39*X2(L):IF N THEN 1000 ELSE N
EXT
10210 M(1)=1:M(4)=1:M(8)=1:MI=50:M1=1:GOSUB 3010
:GOTO 1000
10970 '
10980 '      Nouvelle partie
10990 '
11000 WINDOW 10,26,1,3:PAPER 1:PEN 0:CLS:LOCATE
2,2:PRINT "VOUS AVEZ ";W
11010 FOR I=1 TO 3000:NEXT:WINDOW 1,40,1,25:PAPE
R 0
11020 W=INKEY$:LOCATE 1,16:PRINT "UNE AUTRE PART
IE ( 0 ou N )"
11030 IF W="0" THEN RUN ELSE IF W="N" THEN CLS:EN
D
11040 PEN INT(RND*3)+1:FOR I=1 TO 500:NEXT:GOTO
11020

```

ALDO L'ÉPICIER

Ce programme a été conçu pour deux joueurs qui vont devoir affronter, chacun à leur tour, Aldo l'épicier. Un drôle d'épicier, puisqu'il n'a sur ses étagères que des conserves. Oh, pas trente-six marques de conserves, non, deux seulement. Des rouges et des blanches. Et encore, les blanches deviennent rouges au bout d'un moment ! Ne vous avisez pas de critiquer l'extrême pauvreté du stock d'Aldo ; il vous lancerait une boîte de conserve à la figure. Déjà qu'il le fait sans qu'on lui ait adressé la parole, alors...

Analysons les habitudes commerciales de notre épicier. Comme nous allons le voir, il ne vend peut-être pas grand'chose, mais au moins il rigole.

Au début du jeu, nous remarquons une étagère placée en haut de l'écran. Sur celle-ci apparaissent les deux catégories de conserves que propose Aldo : des rouges et des blanches donc. Repérez bien la place des blanches, ces boîtes ont été remplies de pétards par l'épicier farceur. Il ne s'agira pas, tout à l'heure, de poser la main dessus...

Quand Aldo estime que l'on a suffisamment regardé son étalage, il baisse la bâche de son magasin jusqu'à ce qu'elle cache l'étagère. Il choisit alors trois boîtes de conserve qu'il dépose à portée de main d'une cliente que nous allons devoir guider. Mais attention, une de ces boîtes contient des pétards et il nous est malheureusement impossible de savoir laquelle car l'épicier a pris la précaution de coller une étiquette rouge sur la boîte en question. Alors, faut-il considérer ce jeu comme tout à fait aléatoire et prendre une boîte au hasard ? Si vous êtes chanceux, oui, assurément. Et si vous ne l'êtes pas, essayez de vous rappeler les emplacements qu'avaient les boîtes blanches avant qu'Aldo ne baisse sa bâche. Vous verrez que cela rendra votre choix plus sûr !

Cela étant dit, il ne vous reste plus qu'à savoir comment se déplace la ménagère dans le magasin. Si l'on pousse la manette à gauche ou à droite, elle avance ou elle recule. Cela lui permet soit de se diriger vers les conserves, soit de se placer en dehors de la trajectoire des boîtes qu'Aldo fait de temps en temps dégringoler sur elle. Si l'on pousse la manette vers le haut, la cliente lève son bras pour prendre une boîte de conserve. Et si l'on tire la manette vers le bas, notre ménagère baisse la tête ; cela lui évite de prendre en pleine figure certaines boîtes qui volent à travers la boutique...

FIN DU JEU

Le jeu s'arrête et l'ordinateur propose une nouvelle partie dès que l'un des deux joueurs atteint le score de 30 points. Ces points sont distribués de la façon suivante :

- 2 points quand on prend la première conserve.
- 3 points quand on prend la deuxième.
- 1 point est donné à l'autre joueur si l'on prend une boîte contenant des pétards.
- 1 point est aussi attribué à l'autre joueur si les boîtes lancées par Aldo atteignent la cliente.

La proportion de boîtes blanches parmi les rouges ainsi que la fréquence d'apparition d'Aldo sont déterminées par le niveau de difficulté.

MODIFICATIONS POSSIBLES

| | Instructions BASIC | N° de ligne |
|------------------------------|-----------------------|-------------|
| Score à atteindre | IF P(L) > 29 | 3120 |
| | IF P(1-L) > 29 | 3180 |
| Fréquence des sorties d'Aldo | IF RND < 0.05+0.03*NV | 1000 |
| | IF RND < 0.03+0.03*NV | 1000 |
| Durée de chaque partie | C=100 | 7240 |

JEU SANS MANETTE

Les deux joueurs utiliseront le clavier à tour de rôle. Ils devront se servir des deux flèches horizontales pour faire avancer ou reculer l'acheteuse. L'appui sur la flèche montante fera prendre à celle-ci

une boîte de conserve et elle baissera la tête si la quatrième flèche est enfoncée.

- Dans la ligne 1010, remplacer

```
D=JOY(0)
```

par :

```
W=INKEY$:D=- (W=CHR$(240))-4*(W=CHR$(242))-8*(W=CHR$(243))
```

- Dans les lignes 4030 et 4100, remplacer

```
IF JOY(0) < > 2
```

par :

```
IF INKEY$ < > CHR$(241)
```


- Dans la ligne 5030, remplacer

```
D=JOY(0)
```

par :

```
W=INKEY$:D=-4*(W=CHR$(242))-8*(W=CHR$(243))
```

LE CŒUR DU PROGRAMME



```

940 '
950 '
960 '          ALDO L'EPICIER
970 '
980 '
990 GOTO 9000
1000 IF RND<0.05+0.03*NV THEN 4000 ELSE IF RND<0
.03+0.03*NV THEN 5000
1010 D=JOY(0):IF D=8 THEN X=X+1:GOSUB 2000:GOTO
1030 ELSE IF D=1 THEN 3000
1020 IF D=4 THEN X=X-1:GOSUB 2000 ELSE FOR I=1 T
O 200-50*NV:NEXT
1030 GOSUB 1050:IF C THEN 1000 ELSE FOR I=1 TO 1
5:SOUND 1,90,8:LOCATE 28,20+4*L
1040 IF SQ(1)>127 THEN 1040 ELSE PAPER 3:PEN I:P
RINT "TEMPS : 0":NEXT:GOTO 6000
1050 C=C-1:C=-C*(C>0):PAPER 3:PEN 2:LOCATE 35,20
+4*L:PRINT C
1060 FOR K=1 TO 100:NEXT:PAPER 0:RETURN
1970 '
1980 '          Deplacement lateral
1990 '
2000 X=X+(X=30)-(X=6):PAPER 0:PEN 1:LOCATE X+3,1
5:PRINT R0:LOCATE X+2,16
2010 PRINT R1:LOCATE X,17:PRINT R2:LOCATE X,18:P
RINT R(M):M=1-M:RETURN
2970 '
2980 '          Le bras se leve
2990 '
3000 PAPER 0:LOCATE X+3,16:PRINT " ":MOVE 16*(X+
3)-2,158:DRAWR -16,-4,1
3010 GOSUB 1050:DRAWR 16,4,0:DRAWR -12,10,1:GOSU
B 1050
3020 DRAWR 12,-10,0:DRAWR -4,16,1:GOSUB 1050
3030 FOR J=1 TO 3:IF X(J)<>X+3 THEN NEXT:GOTO 31
90
3040 PEN 3:LOCATE X+3,14:PRINT CHR$(237):PEN 2:L
OCATE X+2,15
3050 PRINT CHR$(238):DRAWR 4,-16,0:DRAWR -12,10,
1
3060 GOSUB 1050:LOCATE X+2,15:PRINT " ":CHR$(8);
CHR$(10);CHR$(238)
3070 DRAWR 12,-10,0:DRAWR -16,-4,1:GOSUB 1050:PE
N 1

```

```

3080 LOCATE X+2,16:PRINT R1:X(J)=0:IF Y(J) THEN
3130
3090 P=P+1:FOR I=1 TO 15:PEN I:PAPER 3:LOCATE 19
,20+4*L
3100 PRINT "+":P:SOUND 1,200+3*I,4:SOUND 2,300+3
*I,3
3110 IF SQ(1)>127 THEN 3110 ELSE NEXT:P(L)=P(L)+
P:PEN 2:LOCATE 13,20+4*L
3120 PRINT P(L):IF P(L)>29 THEN 8000 ELSE IF P=3
THEN 6000 ELSE 1030
3130 FOR I=1 TO 20:J=INT(RND*3)+1:LOCATE X+1,17:
SOUND 1,800+100*J,5
3140 IF SQ(1)>127 THEN 3140 ELSE PEN J:PRINT CHR
$(216)+CHR$(216):NEXT
3150 FOR I=1 TO 100:NEXT:FOR I=1 TO 15:PEN I:PAP
ER 3
3160 LOCATE 19,24-4*L:PRINT "+1":SOUND 1,300+5*I
,6:SOUND 2,400+8*I,5
3170 IF SQ(2)>127 THEN 3170 ELSE NEXT:P(1-L)=P(1
-L)+1:PEN 2:PAPER 3
3180 LOCATE 13,24-4*L:PRINT P(1-L):IF P(1-L)>29
THEN L=1-L:GOTO 8000 ELSE 6000
3190 DRAWR 4,-16,0:DRAWR -12,10,1:GOSUB 1050:DR
AWR 12,-10,0:DRAWR -14,-4,1
3200 GOSUB 1050:PEN 1:LOCATE X+3,16:PRINT CHR$(2
51):GOTO 1030
3970 '
3980 '      Aldo lance une boite
3990 '
4000 PAPER 0:SOUND 1,400,0,4,1:SOUND 2,405,0,4,1
:IF RND<0.5 AND X<25 THEN 4080
4010 F=0:LOCATE 2,15:PRINT CHR$(243):LOCATE 2,16
:PRINT CHR$(242):LOCATE 2,17
4020 PRINT CHR$(241):LOCATE 2,18:PRINT CHR$(240)
:FOR I=3 TO X+6:LOCATE I,15
4030 PEN 2:PRINT CHR$(238):IF JOY(0)<>2 THEN 405
0 ELSE PEN 1:LOCATE X+4,15
4040 PRINT " ":LOCATE X+3,16:PRINT CHR$(254):F=1
:PEN 2
4050 IF F=0 AND I=X+4 THEN 4150 ELSE LOCATE I,15
:PRINT " ":NEXT
4060 FOR I=15 TO 18:LOCATE 2,I:PRINT " ":NEXT:PE
N 1

```

```

4070 LOCATE X+2,16:PRINT R1:LOCATE X+3,15:PRINT
RO:GOTO 1030
4080 F=0:LOCATE 39,15:PRINT CHR$(239):LOCATE 39,
16:PRINT CHR$(242):LOCATE 39,17
4090 PRINT CHR$(241):LOCATE 39,18:PRINT CHR$(240
):FOR I=38 TO X STEP -1:PEN 2
4100 LOCATE I,15:PRINT CHR$(238):IF JOY(0)<>2 TH
EN 4120 ELSE PEN 1:LOCATE X+4,15
4110 PRINT " ":LOCATE X+3,16:PRINT CHR$(254):F=1
:PEN 2
4120 IF F=0 AND I=X+4 THEN 4150 ELSE LOCATE I,15
:PRINT " ":NEXT
4130 FOR I=15 TO 18:LOCATE 39,I:PRINT " ":NEXT:P
EN 1
4140 LOCATE X+2,16:PRINT R1:LOCATE X+3,15:PRINT
RO:GOTO 1030
4150 FOR I=1 TO 20:J=INT(RND*3)+1:LOCATE X+4,15:
SOUND 1,800+100*J,5
4160 IF SQ(1)>127 THEN 4160 ELSE PEN J:PRINT CHR
$(216):NEXT
4170 FOR I=15 TO 18:LOCATE 2,I:PRINT " ":LOCATE
39,I:PRINT " ":NEXT:GOTO 3150
4970 '
4980 '      Chute de boites
4990 '
5000 PAPER 0:SOUND 1,405,0,4,1:SOUND 2,395,0,4,1
:J=X+1+INT(RND*4)
5010 FOR I=5 TO 18:PEN 2:IF J<X+1 OR J>X+4 OR I<
15 THEN 5030
5020 IF (J<X+3 AND I>16) OR (J=X+3 AND I>15) OR
(J=X+4 AND I>14) THEN 5060
5030 LOCATE J,I:PRINT CHR$(238):GOSUB 1050:PEN 1
:LOCATE J,I:PRINT " ":D=JOY(0)
5040 IF D=4 THEN X=X-1:GOSUB 2000 ELSE IF D=8 TH
EN X=X+1:GOSUB 2000
5050 I=I-(I=13):NEXT:GOTO 1030
5060 FOR IO=1 TO 20:K=INT(RND*3)+1:LOCATE J,I:SO
UND 1,800+100*K,5
5070 IF SQ(1)>127 THEN 5070 ELSE PEN K:PRINT CHR
$(216):NEXT:GOTO 3150
5970 '
5980 '      Joueur suivant
5990 '

```

```

6000 P=1:LOCATE 28,20+4*L:PAPER 3:PRINT SPACE$(1
0):FOR I=1 TO 500:NEXT:PAPER 0
6010 I=X:FOR X=I TO 29:GOSUB 2000:SOUND 1,100,3:
SOUND 2,102,2:FOR K=1 TO 150
6020 NEXT K,X:LOCATE 9,14:PRINT SPACE$(23):MOVE
128,176:DRAW 496,176,3:L=1-L
6970 '
6980 '      Etagere et bache
6990 '
7000 WINDOW 1,40,1,5:CLS:FOR I=1 TO 15:LOCATE 1,
2:SOUND 1,395+5*I,5
7010 IF SQ(1)>127 THEN 7010 ELSE PRINT "A VOUS ,
";T(L):PEN I:NEXT:CLS
7020 WINDOW 1,40,1,25:X(1)=INT(RND*15)+13
7030 X(2)=INT(RND*15)+13:IF ABS(X(2)-X(1))<2 THE
N 7030
7040 X(3)=INT(RND*15)+13:IF ABS(X(3)-X(1))<2 OR
ABS(X(3)-X(2))<2 THEN 7040
7050 J=INT(RND*3)+1:FOR I=1 TO 3:IF I=J THEN Y(I
)=1 ELSE Y(I)=0
7060 NEXT:LOCATE 11,4:FOR I=11 TO 29
7070 IF I<>X(1) AND I<>X(2) AND I<>X(3) THEN PEN
1-(RND<0.1*NV):GOTO 7090
7080 PEN 1-(X(J)=I)
7090 PRINT CHR$(238)::NEXT:MOVE 144,336:DRAWR 34
8,0,3:FOR I=1 TO 6000-500*NV
7100 NEXT:PEN 1:LOCATE 2,14:PRINT CHR$(236):MOVE
23,399:DRAWR 0,-208,1
7110 MOVE 24,399:DRAWR 0,-208,1:PEN 2:LOCATE 2,1
5:PRINT CHR$(239)
7120 LOCATE 2,16:PRINT CHR$(242):LOCATE 2,17:PRI
NT CHR$(241):LOCATE 2,18
7130 PRINT CHR$(240):J=4:FOR I=0 TO 15:PEN 2:LOC
ATE 2,15:PRINT CHR$(239+J)
7140 PEN 1:LOCATE 2,14:PRINT CHR$(236-J/4):MOVE
0,399-4*I
7150 DRAWR 640,0,J/4+2:MOVE 0,397-4*I:DRAWR 640,
0,J/4+2
7160 FOR K=1 TO 250:NEXT:J=4-J:NEXT:LOCATE 1,14:
FOR I=1 TO 5:PRINT " ";NEXT
7170 MOVE 23,334:DRAWR 0,-144,0:MOVE 25,334:DRAW
R 0,-144,0
7180 FOR I=1 TO 1000:NEXT:PEN 1:FOR I=1 TO 3:FOR
J=5 TO 13

```

```

7190 LOCATE X(I),J:PRINT CHR$(238):SOUND 1,300+2
0*J,2
7200 IF SQ(1)>127 THEN 7200 ELSE LOCATE X(I),J:P
RINT " ":NEXT
7210 LOCATE X(I),J:PRINT CHR$(238):MOVE 16*X(I),
16*(25-J)
7220 DRAWR -16,0,3:NEXT:FOR I=2 TO 10:SOUND 1,30
0+2*I,4:SOUND 2,302+2*I,4
7230 IF SQ(1)>127 THEN 7230 ELSE PAPER 3:PEN I:L
OCATE 2,20+4*L:PRINT T(L)
7240 LOCATE 28,20+4*L:PRINT "TEMPS : 100":NEXT:C
=100:X=29:GOTO 1000
7970 '
7980 '      Nouvelle partie
7990 '
8000 WINDOW 1,40,1,3:PAPER 3:CLS:PEN 0:LOCATE 5,
2:PRINT T(L);" A GAGNE"
8010 WINDOW 1,40,1,25:PAPER 0:R="UNE AUTRE PARTI
E (O ou N)"
8020 W=INKEY$:IF W="O" THEN RUN ELSE IF W="N" TH
EN CLS:END
8030 PEN INT(RND*3)+1:LOCATE 8,8:PRINT R:FOR I=1
TO 150:NEXT:GOTO 8020
8970 '
8980 '      Initialisation
8990 '
9000 MODE 1:INK 0,0:INK 1,16:INK 2,26:INK 3,9:SY
MBOL AFTER 234
9010 SYMBOL 235,24,24,24,24,31,1,1,1:SYMBOL 236,
24,24,24,24,248,128,128,128
9020 SYMBOL 237,0,0,0,0,0,0,0,255:SYMBOL 238,0,0
,0,3,3,3,0,0
9030 SYMBOL 239,128,132,142,159,158,78,68,68:SYM
BOL 240,36,36,36,100,68,68,68,198
9040 SYMBOL 241,60,60,124,124,124,124,124,124:SY
MBOL 242,124,120,120,120,120,56,56,56
9050 SYMBOL 243,1,33,113,249,121,114,34,34:SYMBOL
L 244,255,255,255,127,127,127,63,63
9060 SYMBOL 245,255,255,255,254,254,254,252,252:
SYMBOL 246,63,31,31,0,0,15,4,4
9070 SYMBOL 247,252,248,248,8,8,248,16,16:SYMBOL
248,96,96,96,96,96,96,96,224
9080 SYMBOL 249,144,144,136,136,132,132,130,130:
SYMBOL 250,248,248,252,252,254,254,255,255

```

```

9090 SYMBOL 251,0,0,1,3,6,28,112,192:SYMBOL 252,
240,240,248,248,248,248,248
9100 SYMBOL 253,2,5,57,120,248,120,112,32:SYMBOL
254,120,125,127,61,16,0,0,0
9110 DEFSTR R-W:DEFINT A-P,X-Z:PAPER 0:BORDER 0:
R0=" "+CHR$(253)+" "
9120 R1=" "+CHR$(251)+CHR$(252)+" ":R2=" "+CHR$(
244)+CHR$(245)+" "
9130 R2=R2+CHR$(250)+" ":R(0)=" "+CHR$(246)+CHR$(
247)+" "
9140 R(1)=R(0)+CHR$(248)+" ":R(0)=R(0)+CHR$(249)
+" ":PEN 2:LOCATE 1,9
9150 LINE INPUT "NOM DU PREMIER JOUEUR ":T(0):T(
0)=LEFT$(T(0),10)
9160 LOCATE 1,13:LINE INPUT "NOM DU DEUXIEME JOU
EUR ":T(1):T(1)=LEFT$(T(1),10)
9170 W=INKEY$:Q=RND:IF W<>" " THEN 9190 ELSE PEN
INT(RND*3)+1:LOCATE 1,21
9180 PRINT"CHOISISSEZ LA DIFFICULTE ( DE 1 A 3
)":FOR I=1 TO 90:NEXT:GOTO 9170
9190 NV=VAL(W):IF NV<1 OR NV>3 THEN 9170 ELSE EN
V 1,7,1,6,3,-3,5
9970 '
9980 '      Decor
9990 '
10000 CLS:WINDOW 1,40,19,21:PAPER 3:CLS:PEN 2:LO
CATE 2,2:PRINT T(0)
10010 LOCATE 12,2:PRINT ": 0":WINDOW 1,40,23,25:
CLS:LOCATE 2,2:PRINT T(1)
10020 LOCATE 12,2:PRINT ": 0":WINDOW 1,40,1,25:L
=1:X=29:GOTO 6000

```

**ELLES DANSENT,
MARIE**

Les trois soucoupes, encadrées de chaque côté par des nuages qu'elles ne peuvent traverser, descendent en zigzaguant au milieu de l'écran. Elles sont programmées pour détruire la ville située au-dessous d'elles, mais ne peuvent le faire de façon directe car une barrière de nuages, infranchissable elle aussi, les en empêche. Il leur faudra donc respecter les trois phases suivantes :

1. Elles devront attendre qu'un nuage veuille bien disparaître pour leur laisser un passage sur la droite ou sur la gauche. Elles seront à ce moment-là vulnérables puisqu'elles devront passer dans le champ de tir de la batterie de missiles. Cinq de ceux-ci s'élèvent dans le même temps lorsque l'on appuie sur la touche FIRE. Naturellement, en poussant la manette à droite ou à gauche, on fait le choix des missiles qui décolleront. Il se peut que les soucoupes ne soient pas touchées toutes les trois : cela ne gênera pas les survivantes qui continueront leur mission.
2. Elles se trouveront à cet instant sur l'un des bords verticaux de l'écran, dans une totale quiétude. Elles pourront monter ou descendre à leur guise mais, tôt ou tard, il leur faudra continuer leur route. Pour, une nouvelle fois, survoler nos lignes, avec bien sûr, tous les risques que cela comporte pour elles : cinq missiles se lanceront dans les airs à la moindre sollicitation du bouton FIRE. Si aucune des soucoupes n'a pu résister à ce traitement, le jeu reprend au début et trois nouveaux objets volants se dessinent en haut de l'écran. En revanche, s'il en reste au moins une, il nous faudra assister de façon passive à la dernière phase.
3. Les soucoupes, en fin de parcours, procèdent au bombardement de la ville dont on a la garde. Se jetant du haut du ciel, comme des kamikazes, elles mettent en pièces un immeuble à chaque fois. Une nouvelle vague de trois ennemis apparaît alors dans le ciel.

FIN DU JEU

La partie se termine lorsque l'on a épuisé sa réserve de missiles (130 au départ) ou lorsque toute la ville a été bombardée (12 soucoupes ont alors réussi à s'échapper). On est déclaré vainqueur si l'on arrive à détruire 40 soucoupes.

La vitesse de déplacement des soucoupes est déterminée par le niveau de difficulté.

MODIFICATIONS POSSIBLES_____

| | Instructions BASIC | N° de ligne |
|---------------------------------------|------------------------|-------------|
| Nombre de missiles | MI = 130 | 11090 |
| Nombre de soucoupes à toucher | IF E > 39 | 8040 |
| Hésitation avant de sortir des nuages | IF RND < 0.7 | 4030 |

JEU SANS MANETTE_____


Le déplacement des missiles à droite ou à gauche se fera à l'aide des deux touches flèches horizontales. Le tir sera déclenché en appuyant sur la barre d'espace.

- Dans la ligne 2000, remplacer

D=JOY(0)

par :

W=INKEY\$:D=-4*(W=CHR\$(242))-8*(W=CHR\$(243))-16*(W=CHR\$(32))



```

940 '
950 '
960 '      ELLES DANSENT , MARIE
970 '
980 '
990 GOTO 10000
1000 FOR I=20 TO XD:LOCATE I,L:PRINT R1:GOSUB 20
00:NEXT
1010 FOR I=XD TO XG STEP -1:LOCATE I,L:PRINT R0:
GOSUB 2000:NEXT
1020 FOR I=XG TO 17:LOCATE I,L:PRINT R1:GOSUB 20
00:NEXT:RETURN
1030 FOR I=17 TO XG STEP -1:LOCATE I,L:PRINT R0:
GOSUB 2000:NEXT
1040 FOR I=XG TO XD:LOCATE I,L:PRINT R1:GOSUB 20
00:NEXT
1050 FOR I=XD TO 20 STEP -1:LOCATE I,L:PRINT R0:
GOSUB 2000:NEXT:RETURN
1060 PEN 3:LOCATE X(0,LO),LO:PRINT RN:LOCATE X(1
,L0),LO:PRINT RN
1070 PO=2*INT(RND*2)-1:LOCATE 18-2*(PO=1),1:PRIN
T R0:L=1:LO=INT(RND*7)+3
1080 R0=CHR$(253)+CHR$(253)+CHR$(253):R1=" "+R0:
R0=R0+" ":PEN 1
1090 XG=X(0,L)+2:XD=X(1,L)-4:IF PO=1 THEN GOSUB
1000 ELSE GOSUB 1030
1100 L=L+1:P1=2*INT(RND*2)-1:IF L=LO THEN P1=2*P
2-1
1110 IF L=LO-1 THEN P2=INT(RND*2):LOCATE X(P2,LO
),LO:PRINT " "
1120 FOR J=0 TO 2:LOCATE 20+(J-2)*PO,L-1:PRINT "
":LOCATE 20+J*P1,L
1130 PRINT CHR$(253):GOSUB 2000:NEXT:PO=P1:IF L<
LO THEN 1090 ELSE 4000
1970 '
1980 '      Base de tir
1990 '
2000 NS=0:FOR K=1 TO NV:D=JOY(0):IF D>15 THEN GO
SUB 3000:RETURN
2010 IF D<>12-P THEN NEXT:RETURN ELSE PEN 1:LOCA
TE 4-29*(P=4),22
2020 PRINT R:LOCATE 4-29*(P=8),22:PRINT SPACE$(5
):P=12-P:RETURN

```

```

2970 '
2980 '      Tir
2990 '
3000 PEN 1:ENT 1,70,1,1:ENV 1,2,5,1,1,0,70:SOUND
      1,70,30,0,1,1
3010 AM=4-29*(P=8):FOR K=22 TO L+1 STEP -3:LOCAT
E AM,K:PRINT R
3020 LOCATE AM,K:PRINT "      ":NEXT:LOCATE AM,22
:PRINT R:MI=MI-5:PEN 3
3030 LOCATE 26,25:PRINT MI::FOR AS=AM TO AM+4
3040 IF TEST(16*AS-8,16*(26-L)-8) THEN NS=NS+1:N
A=AS
3050 NEXT:IF MI THEN PEN 1:RETURN ELSE FOR I=1 T
O 15:SOUND 1,200+20*I,6
3060 IF SQ(1)>127 THEN 3060 ELSE LOCATE 27,25:PE
N I
3070 PRINT "O  MISSILE ";;NEXT:W="PERDU":GOTO 9
000
3970 '
3980 '      Hesitation
3990 '
4000 PEN 1:PO=INT(RND*4):FOR J=1 TO PO
4010 IF P1=1 THEN XD=29:XG=X(0,L)+2:GOSUB 1000:G
OTO 4030
4020 XD=X(1,L)-4:XG=8:GOSUB 1030
4030 IF RND<0.7 THEN 4050 ELSE LOCATE X(P2,L),L:
PEN 3:PRINT RN
4040 P1=-P1:P2=1-P2:LOCATE X(P2,L),L:PRINT "  ":
SOUND 1,100,8:PEN 1:GOTO 4000
4050 LOCATE 20-2*P1,L:PRINT "  ":LOCATE 20+P1,L:P
RINT CHR$(253):GOSUB 2000
4060 LOCATE 20-P1,L:PRINT "  ":LOCATE 20+2*P1,L:P
RINT CHR$(253):GOSUB 2000:NEXT
4970 '
4980 '      Deplacement a gauche
4990 '
5000 IF P1=1 THEN 6000 ELSE P1=0:FOR J=17 TO 9 S
TEP -1
5010 LOCATE J,L:PRINT R0:GOSUB 2000:NEXT
5020 X0=8:X1=P1+1:X2=-1:R2=R0:GOSUB 8000
5030 IF P1=3 THEN 1060 ELSE IF J<4 THEN X2=1:GOT
O 5090
5040 FOR J=9 TO X(1,L)-LEN(R1):LOCATE J,L:PRINT
R1:GOSUB 2000:NEXT

```

```

5050 FOR J=X(1,L)-LEN(R1) TO 9 STEP -1:LOCATE J,
L:PRINT RD:GOSUB 2000
5060 IF J<>X(0,L)+2 OR RND>0.8 THEN NEXT:IF RND>
0.3 THEN 5040 ELSE 5020
5070 PEN 3:LOCATE X(0,L),L:PRINT RN:LOCATE X(1,L
),L:PRINT " "
5080 SOUND 1,295,5:SOUND 2,300,5:SOUND 3,305,5:P
EN 1:GOTO 6050
5090 FOR J=L TO 12 STEP X2:LOCATE P1+1,J:PRINT R
D:GOSUB 2000
5100 LOCATE 1,J:PRINT " ":NEXT:I=1
5110 FOR L=15-2*I TO 15+3*I STEP I:LOCATE P1+1,L
:PRINT RD:GOSUB 2000
5120 LOCATE 1,L:PRINT " ":IF RND>0.15 THEN NEX
T:I=-I:GOTO 5110
5130 X0=P1+1:X1=8:X2=1:R2=R1:GOSUB 8000:X1=9
5140 IF P1=3 THEN 1060 ELSE IF J<4 THEN X2=-1:GO
TO 5090 ELSE 7000
5970 '
5980 '      Deplacement a droite
5990 '
6000 P1=0:FOR J=20 TO 29:LOCATE J,L:PRINT R1:GOS
UB 2000:NEXT
6020 X0=30+P1:X1=37:X2=1:R2=R1:GOSUB 8000
6030 IF P1=3 THEN 1060 ELSE IF J>33 THEN X2=1:GO
TO 6090
6040 FOR J=33-LEN(R0) TO X(0,L)+2 STEP-1:LOCATE
J,L:PRINT RD:GOSUB 2000:NEXT
6050 FOR J=X(0,L)+2 TO 33-LEN(R0):LOCATE J,L:PRI
NT R1:GOSUB 2000
6060 IF J<>X(1,L)-LEN(R1) OR RND>0.3 THEN NEXT:I
F RND<0.7 THEN 6040 ELSE 6020
6070 PEN 3:LOCATE X(1,L),L:PRINT RN:LOCATE X(0,L
),L:PRINT " ":PEN 1
6080 SOUND 1,295,5:SOUND 2,300,5:SOUND 3,305,5:G
OTO 5050
6090 FOR J=L TO 12 STEP X2:LOCATE 37,J:PRINT R1:
GOSUB 2000
6100 LOCATE 37,J:PRINT " ":NEXT:I=1
6110 FOR L=15-2*I TO 15+3*I STEP I:LOCATE 37,L:P
RINT R1:GOSUB 2000
6120 LOCATE 37,L:PRINT " ":IF RND>0.15 THEN N
EXT:I=-I:GOTO 6110

```

```

6130 X0=37:X1=30+P1:X2=-1:R2=RO:GOSUB 8000:X1=29
+P1
6140 IF P1=3 THEN 1060 ELSE IF J>33 THEN X2=-1:G
OTO 6090
6970 '
6980 '      Destruction de la ville
6990 '
7000 R2=R1+" ":FOR I=1 TO 3-P1:READ P0:X2=SGN(P0
-X1)
7010 FOR J=X1 TO P0 STEP X2:LOCATE J,L:PRINT R2:
GOSUB 2000:NEXT
7020 FOR J=L TO 22:LOCATE P0+1,J:PRINT CHR$(253)
:SOUND 1,200-8*J,5
7030 IF SQ(1)>127 THEN 7030 ELSE LOCATE P0,J:PRI
NT CHR$(253)+" ":SOUND 1,400,5
7040 IF SQ(1)>127 THEN 7040 ELSE LOCATE P0,J:PRI
NT " ":NEXT
7050 IF P0=23 THEN W="PERDU":GOTO 9000 ELSE X1=P
0+1
7060 R2=LEFT$(R2,LEN(R2)-2)+" ":NEXT:GOTO 1060
7070 DATA 17,27,13,29,9,25,21,15,11,19,31,23
7970 '
7980 '      Soucoupes touchees
7990 '
8000 FOR J=X0 TO X1 STEP X2:LOCATE J,L:PRINT R2:
GOSUB 2000
8010 IF NS=0 THEN NEXT:RETURN ELSE FOR K=1 TO 12
:SOUND 1,400+20*K,3
8020 IF SQ(1)>127 THEN 8020 ELSE PEN K:LOCATE NA
-NS+1,L:PRINT LEFT$(RO,NS)
8030 NEXT:E=E+NS:PEN 3:LOCATE 20,25:PRINT E;
8040 IF E>39 THEN W="GAGNE":GOTO 9000 ELSE P1=P1
+NS
8050 RO=RIGHT$(RO,LEN(RO)-NS):R1=LEFT$(R1,LEN(R1
)-NS):PEN 1:RETURN
8970 '
8980 '      Nouvelle partie
8990 '
9000 WINDOW 10,26,1,3:PAPER 2:CLS:PEN 0:LOCATE 2
,2:PRINT "VOUS AVEZ ";W
9010 WINDOW 1,40,1,25:PAPER 0:FOR I=1 TO 3000:NE
XT
9020 W=INKEY$:IF W="0" THEN RUN ELSE IF W="N" TH
EN CLS:END

```

```

9030 LOCATE 7,15:PRINT "UNE AUTRE PARTIE ( 0 ou
N )
9040 FOR I=1 TO 150:NEXT:PEN INT(RND*3)+1:GOTO 9
020
9970 '
9980 '      Initialisation
9990 '
10000 MODE 1:INK 0,0:INK 1,16:INK 2,9:INK 3,23:P
APER 0:BORDER 0
10010 SYMBOL 249,255,159,159,159,159,255,255,255
:SYMBOL 250,255,249,249,249,249,255,255,255
10020 SYMBOL 251,255,153,153,153,153,255,255,255
:SYMBOL 252,24,24,24,24,24,24,36,36
10030 SYMBOL 253,0,0,0,56,124,170,0,0:SYMBOL 254
,30,63,127,255,255,127,31,14
10040 SYMBOL 255,28,254,254,254,255,255,126,24:D
EFINT A-P,X-Z:DEFSTR R-W
10050 RN=CHR$(254)+CHR$(255):R=CHR$(252)+CHR$(25
2)+CHR$(252)+CHR$(252)+CHR$(252)
10060 PEN 3:LOCATE 1,10:PRINT "CHOISISSEZ LA DIF
FICULTE ( DE 1 A 3 )"
10070 W=INKEY$:IF W="" THEN Q=RND:GOTO 10070 ELS
E NV=VAL(W)
10080 IF NV<1 OR NV>3 THEN 10070 ELSE NV=(4-NV)*
2-1:CLS:RESTORE 11010
10970 '
10980 '      Decor
10990 '
11000 PEN 3:FOR I=1 TO 9:FOR J=0 TO 1:READ X(J,I
):LOCATE X(J,I),I:PRINT RN
11010 NEXT:NEXT:DATA 12,31,10,28,13,25,13,30,9,2
6,11,29,14,28,12,26,15,30
11020 LOCATE 14,10:PRINT RN;"      ";RN;" ";RN;"
";RN
11030 LOCATE 11,11:PRINT RN;"      ";RN:RN;" ";RN:
"      ";RN
11040 PEN 1:LOCATE 4,22:PRINT R:LOCATE 1,23:PAPE
R 2:PRINT SPACE$(40)
11050 FOR I=9 TO 32:J=INT(RND*3)+19:PEN 1:PAPER
0:LOCATE I,J
11060 PRINT CHR$(244):PEN 3:FOR K=J+1 TO 22:Q=RN
D:LOCATE I,K
11070 PRINT CHR$(250+(Q<0.3)-(Q>0.7)):NEXT:NEXT

```

```
11080 LOCATE 1,25:PRINT "SOUCOUPES TOUCHEES: 0";  
:LOCATE 27,25  
11090 PRINT "130 MISSILES";:P=4:MI=130:LO=1:REST  
ORE 7070:GOTO 1060
```

L'ALBATROS

Plusieurs joueurs peuvent prendre part successivement à ce jeu. Ils devront faire de leur mieux pour aider un albatros à se poser sans trop de casse sur la terre ferme. Cela n'est pas de tout repos car cet oiseau utilise une technique très particulière pour atterrir. Il jette généralement son dévolu sur un champ de fleurs et il se met en devoir de le transformer en piste d'atterrissage ; une piste qui sera obligatoirement débarrassée de toute forme de végétation, la moindre brindille risquant de déséquilibrer un albatros en train de se poser. Quoi, qu'est-ce que vous me demandez ? Comment s'y prend l'oiseau pour détruire des plantes situées plusieurs mètres au-dessous de ses ailes ? Ben, il lance des doses de désherbant...

Vous ne me croyez pas ! Eh bien entrez ce programme dans votre ordinateur et faites-le exécuter. Quand l'albatros sera en plein vol, poussez la manette vers le bas ; vous verrez si notre oiseau rare n'est pas capable de générer spontanément des petits bidons de désherbant. Et tenez, si des fois l'albatros se trouve nez-à-nez avec un bourdon mécontent qu'on détruise ses fleurs préférées, appuyez sur FIRE. Et vous verrez ce que vous verrez : un albatros doté d'une langue de caméléon. Non, mais !

FIN DU JEU

L'albatros traverse l'écran de gauche à droite puis de droite à gauche en perdant chaque fois un peu d'altitude. Il poursuit son vol tant qu'il ne rencontre aucun obstacle ; sa course s'arrête s'il heurte une fleur ou le bourdon. Une deuxième chance est alors donnée au joueur et un nouvel albatros remplace le précédent.

A chaque fois que l'oiseau, après avoir détruit toutes les plantes, arrive à atterrir, il reprend son vol en haut de l'écran et le décor est reconstitué. Avec des fleurs dont les tiges sont de plus en plus grandes.

Le niveau de difficulté détermine dans ce programme la hauteur initiale des fleurs.

MODIFICATIONS POSSIBLES _____

| | Instructions BASIC | N° de ligne |
|--|--------------------|--------------|
| Nombre d'albatros par joueur | IF N < > 2 | 7030 |
| Hauteur initiale des fleurs | Z=18-3*NV | 10020 |
| Fréquence d'apparition du bourdon | RND < 0.6 | 1000 |
| Allongement des fleurs après chacun des atterrissages | Z=Z-2 | 2000 |
| Pouvoir de destruction des bidons de désherbant | Z0=3 Z1=3 | 4020 4050 |

JEU SANS MANETTE _____

On devra appuyer sur la flèche du bas pour commander le lâcher des bidons de désherbant et sur la barre d'espacement pour dérouler la langue de l'albatros jusqu'au bourdon.

- Dans la ligne 3000, remplacer

```
IF JOY(0) < > 2
```

par :

```
IF INKEY$ < > CHR$(241)
```


- Dans la ligne 5040, remplacer

```
IF JOY(0) > 15
```

par :

```
IF INKEY$=CHR$(32)
```

LE CŒUR DU PROGRAMME



```

940 '
950 '
960 '           L'ALBATROS
970 '
980 '
990 GOTO 9000
1000 FOR Y=1 TO 21:IF RND<0.6 AND Y<Z-1 AND Y>1
THEN 5000 ELSE LOCATE 1,Y
1010 PRINT CHR$(253)+CHR$(254)+CHR$(255):X=1:GOS
UB 3000:GOSUB 4000:H=1
1020 FOR X=3 TO 37 STEP 2:LOCATE X-2,Y:PRINT R(H
):H=1-H
1030 IF TEST(16*(X+2)+8,16*(25-Y)+10) THEN M=1:G
OTO 7000
1040 GOSUB 3000:GOSUB 4000:NEXT:LOCATE 37,Y:PRIN
T " "+CHR$(253)+CHR$(254)
1050 GOSUB 3000:GOSUB 4000:LOCATE 39,Y:PRINT "
":GOSUB 6000
1970 '
1980 '           Droite vers gauche
1990 '
2000 Y=Y+1:IF Y=22 THEN Z=Z-2:GOTO 11020 ELSE L
OCATE 38,Y
2010 PRINT CHR$(247)+CHR$(248)+CHR$(249):X=39:GO
SUB 3000:GOSUB 4000:H=3
2020 FOR X=37 TO 3 STEP -2:LOCATE X-1,Y:PRINT R(H
):H=5-H
2030 IF TEST(16*(X-2)-2,16*(25-Y)+12) THEN M=0:G
OTO 7000
2040 GOSUB 3000:GOSUB 4000:NEXT:LOCATE 1,Y:PRINT
CHR$(243)+CHR$(254)+" "
2050 GOSUB 3000:GOSUB 4000:LOCATE 1,Y:PRINT " "
:GOSUB 6000:NEXT Y
2970 '
2980 '           Nouveau bidon
2990 '
3000 IF JOY(0)<>2 THEN FOR I=1 TO 20:NEXT:RETURN
ELSE IF BD THEN 3020
3010 PEN 3:LOCATE X,Y+1:PRINT R:A=A+1:X0=X:Y0=Y+
1:BD=1:PEN 1:RETURN
3020 IF B1 THEN RETURN ELSE PEN 3:LOCATE X,Y+1
3030 PRINT R:A=A+1:X1=X:Y1=Y+1:B1=1:PEN 1:RETURN
3970 '

```

```

3980 '      Deplacements bidons
3990 '
4000 FOR I=1 TO 50*(2-A):NEXT:IF B0=0 THEN 4040
4010 PEN 3:LOCATE X0,Y0:PRINT " ":Y0=Y0+1-(Y0<Z-
1)
4020 IF Y0>21 OR Z0=3 THEN B0=0:Z0=0:A=A-1:GOTO
4040 ELSE LOCATE X0,Y0
4030 PRINT R:IF TEST(16*X0-2,16*(25-Y0)-2) THEN
Z0=Z0+1:Z2=Z2+1
4040 IF B1=0 THEN 4070 ELSE PEN 3:LOCATE X1,Y1:P
RINT " ":Y1=Y1+1-(Y1<Z-1)
4050 IF Y1>21 OR Z1=3 THEN B1=0:Z1=0:A=A-1:GOTO
4070 ELSE LOCATE X1,Y1
4060 PRINT R:IF TEST(16*X1-2,16*(25-Y1)-2) THEN
Z1=Z1+1:Z2=Z2+1
4070 PEN 1:RETURN
4970 '
4980 '      Vol du bourdon
4990 '
5000 IF B0 OR B1 THEN GOSUB 4000:GOTO 5000 ELSE
D=0::LOCATE 1,Y
5010 PRINT CHR$(253)+CHR$(254)+CHR$(255):SOUND 1
,395,0,4,1:YB=Y-1
5020 H=0:FOR X=3 TO 37 STEP 2:PEN 1:LOCATE X-2,Y
:PRINT R(H):H=1-H
5030 IF D=1 THEN GOSUB 3000:GOSUB 4000:GOTO 5070
5040 IF X=33 THEN M=1:GOTO 7000 ELSE IF JOY(0)>1
5 THEN 5090
5050 LOCATE 35,YB:PRINT " ":IF RND<0.5 THEN YB=
Y ELSE YB=Y-1-2*(RND<0.5)
5060 PEN 2:LOCATE 35,YB:PRINT CHR$(238+H)+CHR$(2
40)
5070 NEXT:LOCATE 37,Y:PRINT " "+CHR$(253)+CHR$(
254):GOSUB 3000:GOSUB 4000
5080 LOCATE 39,Y:PRINT " ":GOSUB 6000:GOTO 2000
5090 W=STRING$(32-X,"-"):PEN 3:LOCATE X+3,Y:PRIN
T W:PEN 0:LOCATE X+3,Y:PRINT W
5100 PEN 2:IF YB<>Y THEN 5050 ELSE FOR I=1 TO 12
:SOUND 1,500+20*I,5
5110 IF SQ(1)>127 THEN 5110 ELSE PEN I:LOCATE 35
,Y
5120 PRINT CHR$(238+H)+CHR$(240):H=1-H:NEXT:D=1:
GOTO 5070

```

```

5970 '
5980 '      Affichage score
5990 '
6000 Z(G)=Z2*NV*10:PAPER 1:PEN 0:LOCATE 13,24:PR
INT Z(G):PAPER 0:PEN 1:RETURN
6970 '
6980 '      Albatros contre obstacle
6990 '
7000 H=3:FOR I=1 TO 12:SOUND 1,195+2*I,5:SOUND 2
,200+2*I,5:SOUND 4,205+2*I,5
7010 IF SQ(1)>127 THEN 7010 ELSE PEN I:LOCATE X-
M-1,Y:PRINT R(H-2*M)
7020 NEXT:LOCATE X-M-2,Y:PRINT SPACE$(7):LOCATE
35,YB
7030 PRINT "   ":N=N+1:IF N<>2 THEN 7050 ELSE GOS
UB 6000
7040 FOR I=1 TO 1000:NEXT:IF Z(G)>ZG THEN ZG=Z(G
):GOTO 10000 ELSE 10000
7050 IF B0 OR B1 THEN GOSUB 4000:GOTO 7050 ELSE
PEN 1:Y=INT((Y+1)/2)*2-3:H=0
7060 Y=-Y*(Y>1)-(Y<2):FOR I=1 TO Y-1:LOCATE 1,I:
PRINT R(H):SOUND 1,100-4*I,4
7070 IF SQ(1)>127 THEN 7070 ELSE LOCATE 3,I:PRIN
T "   ":H=1-H:NEXT:GOTO 1020
7970 '
7980 '      Nouvelle partie
7990 '
8000 W=INKEY$:IF W="0" THEN RUN ELSE IF W="N" TH
EN CLS:END
8010 PEN INT(RND*3)+1:LOCATE 3,24:PRINT "UNE AUT
RE PARTIE ? ( O ou N )"
8020 FOR I=1 TO 150 :NEXT:GOTO 8000
8970 '
8980 '      Initialisation
8990 '
9000 MODE 1:INK 0,0:INK 1,16:INK 2,9:INK 3,23:BO
RDER 0:PAPER 0:SYMBOL AFTER 237
9010 SYMBOL 238,128,64,32,16,121,255,121,0:SYMBOL
L 239,1,2,4,8,121,255,121,0
9020 SYMBOL 240,0,0,0,248,252,254,255,1:SYMBOL 2
41,0,7,7,7,7,7,7,0
9030 SYMBOL 242,193,225,114,58,26,6,1,1:SYMBOL 2
43,0,0,12,28,56,112,96,128

```

```

9040 SYMBOL 244,1,0,0,56,255,0,0,0:SYMBOL 245,12
8,96,28,3,255,15,7,0
9050 SYMBOL 246,1,3,56,192,255,252,240,0:SYMBOL
247,0,56,255,0,0,0,3,12
9060 SYMBOL 248,0,0,255,15,55,195,12,48:SYMBOL 2
49,0,0,255,252,240,0,0,0
9070 SYMBOL 250,128,96,28,3,255,63,15,0:SYMBOL 2
51,1,6,56,192,255,240,224,0
9080 SYMBOL 252,128,0,0,28,255,0,0,0:SYMBOL 253,
0,0,255,63,15,0,0,0
9090 SYMBOL 254,0,0,255,240,236,195,48,12:SYMBOL
255,0,28,255,0,0,0,192,48
9100 DEFINT A-P,X-Z:DEFSTR R-W:ENV 1,7,1,7,3,-3,
5
9110 R(0)=CHR$(253)+CHR$(254)+CHR$(255):R(1)=CH
R$(250)+CHR$(251)+CHR$(252)
9120 R(2)=CHR$(247)+CHR$(248)+CHR$(249):R(3)=CHR
$(244)+CHR$(245)+CHR$(246)
9130 FOR I=0 TO 1:R(I)=" "+R(I):R(I+2)=R(I+2)+"
":NEXT:R=CHR$(241)+" "
9140 PEN 1:LOCATE 1,23:PRINT "CHOISISSEZ LA DIFF
ICULTE ( DE 1 A 3 )"
9150 W=INKEY$:IF W="" THEN 9150 ELSE NV=VAL(W):I
F NV<1 OR NV>3 THEN 9150
9160 CLS:PRINT "QUEL EST LE NOMBRE DE JOUEURS"
9170 W=INKEY$:IF W="" THEN Q=RND:GOTO 9170 ELSE
E=VAL(W):IF E=0 THEN 9170
9180 PEN 3:FOR I=1 TO E:LOCATE 1,2*I+4:PRINT "NO
M DU JOUEUR NUMERO";I;
9190 LINE INPUT V(I):V(I)=LEFT$(V(I),11):NEXT
9970 '
9980 '      Joueur suivant
9990 '
10000 CLS:FOR I=1 TO E:PEN 1:LOCATE 1,2*I+1:PRIN
T V(I):PEN 3:LOCATE 13,2*I+1
10010 PRINT Z(I):LOCATE 22,2*I+1:PRINT "POINT":;
IF Z(I)>1 THEN PRINT "S"
10020 NEXT:G=G+1:IF G>E THEN 8000 ELSE P=0:Z2=0:
Z=18-3*NW:YB=1
10030 W=INKEY$:IF W=CHR$(224) THEN 11000 ELSE PE
N INT(RND*3)+1:LOCATE 1,25
10040 PRINT V(G);" APPUYEZ SUR COPY":;FOR I=1 TO
150:NEXT:GOTO 10030

```

```
10970 '
10980 '      Decor
10990 '
11000 CLS:WINDOW 1,40,23,25:PAPER 1:CLS:WINDOW 1
,40,1,25:PEN 0:LOCATE 2,24
11010 PRINT V(G):LOCATE 12,24:PRINT": 0":PEN 3:L
OCATE 25,24:PRINT "RECORD :";ZG
11020 N=0:P=P+1:Z1=0:BO=0:B1=0:A=0:PAPER 2:LOCAT
E 1,22:PRINT SPACE$(40):PEN 3
11030 PAPER 1:LOCATE 25,25:PRINT "PASSAGE :";P;;
PAPER 0:FOR I=9 TO 31 STEP 2
11040 Z0=Z+INT(RND*(18-Z)):Z0=-Z0*(Z0>1)-(Z0<2):
PEN 1:LOCATE I,Z0
11050 PRINT CHR$(221)+CHR$(220):PEN 2:FOR J=Z0+1
TO 21:LOCATE I,J
11060 PRINT CHR$(242)+CHR$(243):NEXT:NEXT:PEN 1:
Z0=0:GOTO 1000
```

NUAGES ET SOUCOUPES

Les envahisseurs ont mis au point un plan diabolique : ils ont observé que les bases de tir qui assurent notre défense ont été édifiées, bien imprudemment il est vrai, à proximité immédiate de deux barrages remplis d'eau. Ordre a donc été donné aux soucoupes de bombarder sans relâche ces barrages jusqu'à ce que l'eau qui s'en échappe submerge totalement le champ de tir.

L'idée générale de ce jeu est donc la suivante : les soucoupes partent du milieu de l'écran et se dirigent soit à droite, soit à gauche. Elles sont alors obligées de passer dans la ligne de mire d'un des deux lance-missiles. Si vous avez de bons réflexes, elles volent en éclats et ne vont pas plus loin. Sinon, elles poursuivent leur route et ne tardent pas à survoler un barrage qu'elles détruisent en partie. Le principe des vases communicants est alors respecté : l'eau perdue par le barrage va se retrouver quelque part ; l'ennui c'est que le "quelque part" en question, c'est notre surface de tir ! D'accord, les lance-missiles sont prévus pour résister aux inondations, mais tout de même pas au-delà d'un certain niveau !

Cessons d'envisager le pire et voyons ce qu'il faut faire pour sortir vainqueur du combat. On doit détruire la totalité de la flotte ennemie, ce qui représente 10 soucoupes à abattre. C'est très faisable, comme vous le voyez. Surtout si l'on comprend comment la mise à feu d'un missile s'effectue : on pousse la manette du côté de la flèche qui est dessinée sous le lance-missile. Par exemple, si l'on veut tirer avec la base de droite et si, sous cette base, est tracée une flèche qui descend, il faut pousser la manette vers le bas. OK ?

Un dernier détail concernant les nuages dessinés dans le haut de l'écran. Ils servent d'une part de cache naturelle aux soucoupes mais, d'autre part, les envahisseurs les utilisent parfois pour "tromper l'ennemi". On ne sait par quel mystère ils arrivent à colorer certains d'entre eux en rouge et à les faire se déplacer dans notre ligne de tir. Toujours est-il que, pris par le feu de l'action, nous risquons fort d'épuiser notre stock de missiles en "tirant dans les nuages".

FIN DU JEU

On ne peut gagner la partie qu'en détruisant 10 soucoupes. On est perdant si l'on épuise son stock de missiles (25 de ceux-ci sont alloués au départ) ou si le champ de tir est entièrement submergé par l'eau qui s'écoule des barrages. Le niveau de difficulté influence directement la vitesse de déplacement des soucoupes.

MODIFICATIONS POSSIBLES _____

| | Instructions BASIC | N° de ligne |
|-------------------------------------|--------------------|-------------|
| Nombre de missiles | MI = 25 | 9000 |
| Nombre de soucoupes à détruire | S(C) = 10 | 4020 |
| Fréquence de changement des flèches | RND < 0.3 | 1000 |

JEU SANS MANETTE _____


Le départ d'un missile sera commandé par appui sur la flèche que l'on voit dessinée sous la base de tir choisie.

- Dans les lignes 1060, 1130 et 2050, remplacer

S = JOY(0)

par :

W = INKEY\$:S - - (W = CHR\$(240)) - 2 * (W = CHR\$(241)) - 4 * (W = CHR\$(242)) - 8 * (W = CHR\$(243))



```

940 '
950 '
960 '      NUAGES      ET      SOUCOUPES
970 '
980 '
990 GOTO 8000
1000 IF RND<0.3 THEN GOSUB 6000 ELSE IF RND<0.5
THEN GOSUB 2000
1010 E=INT(RND*4)+1:IF Z(E) THEN 1030 ELSE PEN 3
1020 J=INT(RND*4)+8:LOCATE 2*J+1,E+1:PRINT U(0):
A(E,J)=1:Z(E)=1
1030 Y(E)=2*(RND<0.5)+1:FO=0:PEN 1:ENV 1,6,1,4,3
,-3,5:SOUND 1,410,0,4,1
1040 IF SQ(1)>127 THEN 1040 ELSE FOR I=X(E)+Y(E)
TO -19*(Y(E)=1) STEP Y(E)
1050 IF A(E,I)=0 THEN LOCATE 2*I+1,E+1:PRINT U(4
+Y(E))
1060 K=4*(4-NV)-3:FOR J=1 TO K:S=JOY(0)
1070 IF (S=C(1) OR S=C(2)) AND FO=0 THEN F1=0:F2
=I:GOTO 3000
1080 NEXT:IF I*Y(E)=-2 OR I*Y(E)=17 THEN 5000
1090 IF A(E,I)=0 THEN LOCATE 2*I+1,E+1:PRINT "
"
1100 NEXT:J=800*(INT(RND*4)+1):FOR K=1 TO J:NEXT
1110 Y(E)=-Y(E):FO=0:FOR J=I+Y(E) TO X(E) STEP Y
(E)
1120 IF A(E,J)=0 THEN LOCATE 2*J+1,E+1:PRINT U(4
+Y(E))
1130 K=4*(4-NV)-3:FOR L=1 TO K:S=JOY(0)
1140 IF (S=C(1) OR S=C(2)) AND FO=0 THEN F1=1:F2
=J:GOTO 3000
1150 NEXT:IF A(E,J)=0 THEN LOCATE 2*J+1,E+1:PRIN
T " "
1160 NEXT:GOTO 1000
1970 '
1980 '      Deplacement d'un nuage
1990 '
2000 I=INT(RND*4)+1:J=INT(RND*20):IF A(I,J)<2 TH
EN 2000
2010 FO=0:A(I,J)=0:K=-2*(RND<0.5)-1:F3=0:KO=INT(
RND*20)+10
2020 FOR L=1 TO KO:PEN 1:IF F3=0 THEN LOCATE 2*J
+1,I+1:PRINT " "

```

```

2030 J=J+K:IF (J+1)*(J-20)=0 THEN K=-K:GOTO 2030
2040 F3=(A(I,J)>0):IF F3=0 THEN LOCATE 2*J+1,I+1
:PRINT U(0)
2050 S=JOY(0):IF (S=C(1) OR S=C(2)) AND F0=0 THE
N F4=1:GOTO 3000
2060 NEXT:PEN 3:LOCATE 2*J+1,I+1:PRINT U(0)
2070 IF A(I,J)=1 THEN 2100
2080 IF J=2 OR J=4 OR J=15 OR J=17 THEN LOCATE 2
*J+1,I+1:PRINT " ":GOTO 2100
2090 A(I,J)=2:RETURN
2100 J=J+K:LOCATE 2*J+1,I+1:PRINT U(0):GOTO 2070
2970 '
2980 '     Tir
2990 '
3000 ENT 1,70,1,1:ENV 1,2,4,1,1,0,60:SOUND 1,30,
40,0,1,1:F0=1
3010 FOR I1=18 TO 6 STEP -2:LOCATE T(S),I1:PRINT
CHR$(255)
3020 FOR I2=1 TO 30:NEXT:LOCATE T(S),I1:PRINT "
"
3030 NEXT:LOCATE 9,18:PRINT CHR$(255):LOCATE 31,
18:PRINT CHR$(255)
3040 MI=MI-1:PEN 0:PAPER 1:LOCATE 1,25:PRINT MI;
:IF MI=0 THEN 3070
3050 PAPER 0:PEN 1:IF F4 THEN F4=0:GOTO 2060
3060 IF F2=(T(S)-1)/2 THEN 4000 ELSE IF F1 THEN
1150 ELSE 1080
3070 FOR I=1 TO 15:LOCATE 2,25:PAPER 1:PRINT "0
MISSILE ";;SOUND 1,400+20*I,6
3080 IF SQ(1)>127 THEN 3080 ELSE PEN I:NEXT:GOTO
7000
3970 '
3980 '     Soucoupe touchee
3990 '
4000 FOR I=1 TO 12:PEN I:LOCATE 2*F2+1,E+1:PRINT
U(4+Y(E)):SOUND 1,80+20*I,2
4010 IF SQ(1)>127 THEN 4010 ELSE NEXT:S(C)=S(C)+
1:Z(E)=0:PAPER 1:PEN 0
4020 LOCATE 36,25:PRINT S(C);:PAPER 0:PEN 1:IF S
(C)=10 THEN 7000 ELSE 1000
4970 '
4980 '     Bombar dement
4990 '

```

```

5000 FOR I1=E+2 TO 15:LOCATE 2*I+1,I1:PRINT CHR$(
(58):SOUND 1,100-INT(RND*80),3
5010 IF SQ(1)>127 THEN 5010 ELSE LOCATE 2*I+1,I1
:PRINT " ":NEXT
5020 ENV 2,15,-1,4:SOUND 4,0,0,0,2,0,10:FOR I2=1
TO 3:FOR I1=1 TO 12:PEN I1
5030 LOCATE 2*I+1,15:PRINT CHR$(218)+CHR$(218):N
EXT:NEXT:P=-(Y(E)=1)
5040 FOR I1=1 TO 2:MOVE 544*P,H(P):DRAWR 96,0,0:
H(P)=H(P)-1:NEXT:PEN 1
5050 MOVE 96,H:DRAWR 448,0,3:H=H+2:IF H=112 THEN
7000 ELSE 1090
5970 '
5980 '     Nouvelles fleches
5990 '
6000 PEN 0:PAPER 3:FOR I1=1 TO 6:FOR I2=9 TO 31
STEP 22
6010 I3=2^INT(RND*4):LOCATE I2,22:PRINT U(I3):SO
UND 1,200+INT(RND*100),5
6020 IF SQ(1)>127 THEN 6020 ELSE NEXT:NEXT:C(1)=
2^INT(RND*4)
6030 C(2)=2^INT(RND*4):IF C(2)=C(1) THEN 6030 EL
SE LOCATE 9,22:PRINT U(C(1))
6040 LOCATE 31,22:PRINT U(C(2)):T(C(1))=9:T(C(2)
)=31:PAPER 0:PEN 1:RETURN
6970 '
6980 '     Nouvelle partie
6990 '
7000 WINDOW 10,26,8,10:PAPER 1:PEN 0:CLS:LOCATE
2,2
7010 PRINT "VOUS AVEZ "':IF S(C)=10 THEN PRINT "
GAGNE" ELSE PRINT "PERDU"
7020 WINDOW 1,40,1,25:PAPER 0:FOR I=1 TO 6000:NE
XT:IF C<>N THEN 9000
7030 CLS:FOR I=1 TO N:PEN 1:LOCATE 1,2*I:PRINT W
(I):LOCATE 16,2*I
7040 PRINT S(I):"POINTS "':IF S(I)=10 THEN PRINT
"(GAGNE)" ELSE PRINT "(PERDU)"
7050 NEXT:U="UNE AUTRE PARTIE ( O ou N )"
7060 W=INKEY$:IF W="0" THEN RUN ELSE IF W="N" TH
EN CLS:END
7070 PEN INT(RND*3)+1:LOCATE 5,14:PRINT U:FOR I=
1 TO 150:NEXT:GOTO 7060

```

```

7970 '
7980 '      Initialisation
7990 '
8000 MODE 1:INK 0,0:INK 1,16:INK 2,10:INK 3,26:B
ORDER 0:PAPER 0
8010 SYMBOL AFTER 238:DEFINT A-T,X-Z:DEFSTR U-W:
DIM A(4,19)
8020 SYMBOL 239,0,0,0,3,15,255,24,0:SYMBOL 240,0
,15,112,192,248,240,56,0
8030 SYMBOL 241,0,1,3,6,12,24,63,63:SYMBOL 242,2
4,12,6,3,1,0,0,0
8040 SYMBOL 243,0,128,192,96,48,24,252,252:SYMBOL
L 244,24,48,96,192,128,0,0,0
8050 SYMBOL 245,99,51,27,15,7,3,0,0:SYMBOL 246,2
4,48,96,192,128,0,0,0
8060 SYMBOL 247,0,0,3,7,15,27,51,99:SYMBOL 248,0
,0,0,128,192,96,48,24
8070 SYMBOL 249,62,127,255,255,255,127,60,0:SYMB
OL 250,56,252,254,254,254,252,120,0
8080 SYMBOL 251,0,240,14,3,31,15,28,0:SYMBOL 253
,0,0,0,192,240,255,24,0
8090 SYMBOL 254,63,63,63,57,54,57,54,57,54:SYMBOL 2
55,6,6,6,6,6,6,9,9
8100 U(0)=CHR$(249)+CHR$(250):U(3)=CHR$(239)+CHR
$(240):U(5)=CHR$(251)+CHR$(253)
8110 U(1)=CHR$(247)+CHR$(248)+CHR$(8)+CHR$(8)+CH
R$(10)+CHR$(209)+CHR$(32)
8120 U(2)=CHR$(209)+CHR$(32)+CHR$(8)+CHR$(8)+CHR
$(10)+CHR$(245)+CHR$(246)
8130 U(4)=CHR$(241)+CHR$(210)+CHR$(8)+CHR$(8)+CH
R$(10)+CHR$(242)+CHR$(32)
8140 U(8)=CHR$(210)+CHR$(243)+CHR$(8)+CHR$(8)+CH
R$(10)+CHR$(32)+CHR$(244)
8150 PEN 1:LOCATE 1,1:PRINT "QUEL EST LE NOMBRE
DE JOUEURS ?"
8160 W=INKEY$:Q=RND:IF W="" THEN 8160 ELSE N=VAL
(W):IF N=0 THEN 8160
8170 PEN 2:FOR I=1 TO N:LOCATE 1,2*I+2:PRINT "NO
M DU JOUEUR NUMERO";I;
8180 LINE INPUT W(I):W(I)=LEFT$(W(I),10):NEXT
8190 PEN 1:LOCATE 1,22:PRINT "CHOISISSEZ LA DIFF
ICULTE ( DE 1 A 3 )"
8200 W=INKEY$:IF W="" THEN 8200 ELSE NV=VAL(W):I
F NV<1 OR NV>3 THEN 8200

```

```

8970 '
8980 '   Decor
8990 '
9000 MI=25:CLS:PAPER 1:WINDOW 1,40,21,25:CLS:PAP
ER 2:WINDOW 6,6,16,20
9010 CLS:WINDOW 1,6,21,21:CLS:WINDOW 35,35,16,20
:CLS:WINDOW 35,40,21,21
9020 CLS:PAPER 3:WINDOW 1,5,16,20:CLS:WINDOW 36,
40,16,20:CLS
9030 WINDOW 1,40,1,25:PAPER 1:PEN 0:LOCATE 2,25:
PRINT "25 MISSILES";
9040 LOCATE 18,25:PRINT "SOUCOUPES TOUCHEES 0";:
PAPER 0:PEN 2
9050 LOCATE 9,19:PRINT CHR$(254)+CHR$(211):LOCAT
E 9,20
9060 PRINT CHR$(219)+CHR$(219):LOCATE 31,19:PRIN
T CHR$(254)+CHR$(211)
9070 LOCATE 31,20:PRINT CHR$(219)+CHR$(219):PEN
1:LOCATE 9,18
9080 PRINT CHR$(255):LOCATE 31,18:PRINT CHR$(255
)
9090 PEN 3:C=C+1:FOR I=1 TO N:LOCATE 1,I:PRINT W
(I):LOCATE 14,I
9100 PRINT S(I):LOCATE 23,I:PRINT "POINT";:IF S(
I)>1 THEN PRINT"S"
9110 NEXT:FOR I=1 TO 4:FOR J=0 TO 19:A(I,J)=0:NE
XT:NEXT:RESTORE
9120 IF INKEY$=CHR$(224) THEN 9140 ELSE LOCATE 1
,14:PEN INT(RND*3)+1
9130 PRINT W(C);" , APPUYEZ SUR COPY":FOR I=1 TO
100:NEXT:GOTO 9120
9140 WINDOW 1,40,1,14:CLS:WINDOW 1,40,1,25:PEN 3
:FOR I=1 TO 10
9150 J=INT(RND*4)+1:K=INT(RND*20):IF K=2 OR K=4
OR K=15 OR K=17 THEN 9150
9160 IF A(J,K) THEN 9150 ELSE A(J,K)=2:LOCATE 2*
K+1,J+1:PRINT U(0):NEXT
9170 FOR I=1 TO 6:READ J:LOCATE J,1:PRINT U(0):N
EXT:DATA 5,9,17,23,31,35
9180 FOR I=1 TO 4:J=INT(RND*4)+8:LOCATE 2*J+1,I+
1:PRINT U(0):Z(I)=1
9190 A(I,J)=1:X(I)=J:NEXT:GOSUB 6000:H(0)=158:H(
1)=158:H=80:GOTO 1000

```

POKE-MAN

Voici la version BASIC d'un jeu célèbre qui se déroule dans un labyrinthe :

PREMIÈRE PARTIE DU JEU

Poke-man, représenté par un losange jaune, est poursuivi par trois fantômes. Le joueur doit, à l'aide de la manette, guider ce malheureux qui peut monter, descendre, aller à gauche ou à droite ; il a la possibilité de se diriger vers n'importe quel endroit de l'écran, mais ne peut en aucune façon traverser les murs du labyrinthe. En ce qui concerne les poursuivants, de couleur rose, on ne peut pas dire que les murs les gênent beaucoup : ils ont la faculté, comme tous les fantômes, n'est-ce pas ? de passer à travers tous les types d'obstacles. On connaîtra tout d'eux quand on saura qu'ils se rendent parfois invisibles et qu'ils se déplacent plutôt lentement. Lentement, oui, mais sûrement : ils ne perdent jamais la trace de leur proie et, si l'on n'y prend garde, encadrent ce pauvre Poke-man, le dévorent dans un bruit de chaînes et attendent qu'un de ses frères vienne le remplacer dans l'arène. Avec l'espoir de lui faire subir un sort comparable. Mais alors, que peut faire notre protégé, sinon fuir encore et toujours ? Eh bien, rien ne l'empêche d'avalier une de ces super-pastilles jaunes disposées aux quatre coins du décor. Et là, les choses changeront. Avant de voir en quoi, voici un petit conseil : on a intérêt à diriger notre ami vers une super-pastille et à attendre, avant de la croquer, que les fantômes se soient un peu approchés ; il sera ensuite plus facile de les rattraper.

DEUXIÈME PARTIE DU JEU

Les rôles sont inversés. C'est Poke-man qui va maintenant poursuivre ses ennemis et, puisque c'est là le but du jeu, essayer d'en attraper le plus possible. Les fantômes n'ont plus qu'une seule préoccupation : rejoindre au plus vite et sans ennui la partie centrale du labyrinthe où ils seront en sécurité. Ils ont, cette fois encore, la possibilité de traverser les murs ; la ligne droite est décidément pour eux le plus court chemin. Mais Poke-man a ses chances : il est plus rapide que ses adversaires et il lui arrive souvent d'en manger un ou deux, même s'il doit faire un trajet plus long. Mais, autant qu'il le sache, quand les fantômes auront regagné leur antre, ils reprendront le rôle des attaquants et notre cher petit losange devra, à nouveau, se diriger vers une super-pastille.

FIN DU JEU _____

Le message de fin de jeu apparaît sur l'écran lorsque trois Poke-men ont été attrapés par leurs poursuivants.

On est déclaré gagnant par la machine si l'on arrive à capturer 10 fantômes.

La vitesse de déplacement des fantômes est fonction du niveau de difficulté. Ce dernier détermine aussi la fréquence d'apparition des super-pastilles.

MODIFICATIONS POSSIBLES _____

| | Instructions BASIC | N° de ligne |
|---|--------------------|-------------|
| Nombre de Poke-men | IFPP=3 | 3040 |
| Nombre de fantômes à attraper | IFF < 10 | 4080 |
| Apparition de nouvelles super-pastilles | FORI=1TOP | 5000 |

JEU SANS MANETTE _____

Le déplacement du Poke-man sera réalisé en enfonçant l'une des quatre flèches.


- Dans les lignes 1020 et 2030, remplacer

S=JOY(0)

par :

S\$=INKEY\$:S=- (S\$=CHR\$(240))-2*(S\$=CHR\$(241))-4*(S\$=CHR\$(242))-8*(S\$=CHR\$(243))

LE CŒUR DU PROGRAMME



```

940 '
950 '
960 '          POKE - MAN
970 '
980 '
990 GOTO 7000
1000 GOSUB 5000:FOR I=1 TO 13:SOUND 1,300+10*I,4
:SOUND 2,400+10*I,4
1010 IF SQ(1)>127 THEN 1010 ELSE PEN I:LOCATE 27
,24:PRINT "SAUVEZ-VOUS !":NEXT
1020 FOR I=1 TO 3:PO=(RND<0.4):S=JOY(0):U=(S=4)-
(S=8):V=(S=1)-(S=2)
1030 IF U=0 AND V=0 THEN 1090 ELSE Z=T(X0+U,Y0+V
)
1040 IF Z=17 THEN 1070 ELSE IF Z<>15 THEN 1090
1050 LOCATE 2*X0+2,Y0+1:PEN 3:PRINT " "
1060 LOCATE 2*(X0+U)+2,Y0+V+1:PRINT C(13):GOTO 2
000
1070 PEN 3:T(X0,Y0)=17:LOCATE 2*X0+2,Y0+1:PRINT
" ":X0=X0+U:Y0=Y0+V
1080 LOCATE 2*X0+2,Y0+1:PRINT C(13):T(X0,Y0)=13
1090 IF P=1 AND Y0<>Y(I) AND X0<>X(I) THEN XI=-P
0*SGN(X0-X(I)):YI=(1+PO)*SGN(Y0-Y(I)):GOTO 1110
1100 XI=SGN(X0-X(I)):YI=SGN(Y0-Y(I))
1110 T(X(I),Y(I))=Z(I):PEN 2:LOCATE 2*X(I)+2,Y(I
)+1:PRINT C(Z(I))
1120 X(I)=X(I)+XI:Y(I)=Y(I)+YI:PEN 1:LOCATE 2*X(I
)+2,Y(I)+1
1130 PRINT C(10):Z(I)=T(X(I),Y(I)):T(X(I),Y(I))=
9+I
1140 IF Z(I)=13 THEN 3000 ELSE IF Z(I)>9 AND Z(I
)<13 THEN Z(I)=Z(Z(I)-9)
1150 NEXT:GOTO 1020
1970 '
1980 '          Poke-man est le poursuivant
1990 '
2000 FOR I=1 TO 13:SOUND 1,300+10*I,4:SOUND 2,40
0+10*I,4
2010 IF SQ(1)>127 THEN 2010 ELSE PEN I:LOCATE 27
,24:PRINT "CAPTUREZ-LES!":NEXT
2020 T(X0,Y0)=17:X0=X0+U:Y0=Y0+V:T(X0,Y0)=13
2030 FOR I=1 TO 3:PO=(RND<0.4):S=JOY(0):U=(S=4)-
(S=8):V=(S=1)-(S=2)

```

```

2040 IF U=0 AND V=0 THEN 2080 ELSE Z=T(X0+U,Y0+V
)
2050 IF Z<>17 THEN IF Z>9 AND Z<13 THEN 4000 EL
SE 2080
2060 PEN 3:T(X0,Y0)=17:LOCATE 2*X0+2,Y0+1:PRINT
" ":X0=X0+U:Y0=Y0+V
2070 LOCATE 2*X0+2,Y0+1:PRINT C(13):T(X0,Y0)=13
2080 IF X(I)=7+I AND Y(I)=11 THEN 2160
2090 IF P=1 AND Y(I)<>11 AND X(I)<>7+I THEN XI=-
PO*SGN(7+I-X(I)):YI=(1+PO)*SGN(11-Y(I)):GOTO 211
0
2100 XI=SGN(7+I-X(I)):YI=SGN(11-Y(I))
2110 IF T(X(I)+XI,Y(I)+YI)=13 THEN 2160 ELSE T(X
(I),Y(I))=Z(I)
2120 PEN 2:LOCATE 2*X(I)+2,Y(I)+1:PRINT C(Z(I)):
X(I)=X(I)+XI:Y(I)=Y(I)+YI
2130 PEN 1:LOCATE 2*X(I)+2,Y(I)+1:PRINT C(10)
2140 Z(I)=T(X(I),Y(I)):T(X(I),Y(I))=9+I
2150 IF Z(I)>9 AND Z(I)<13 THEN Z(I)=Z(Z(I)-9)
2160 NEXT:FOR J=1 TO 3:IF X(J)<>7+J OR Y(J)<>11
THEN 2030 ELSE NEXT
2170 FOR J=1 TO 3:Z(J)=17:NEXT:PEN 1:LOCATE 18,1
2
2180 PRINT C(10);C(10);C(10):GOTO 1000
2970 '
2980 '      Un fantome attrape Poke-man
2990 '
3000 Z(I)=17:FOR J=1 TO 12:SOUND 1,100-INT(RND*5
0),4
3010 IF SQ(1)>127 THEN 3010 ELSE LOCATE 2*X(I)+2
,Y(I)+1
3020 PEN J-1:PRINT C(13):SOUND 1,100-INT(RND*50)
,4
3030 IF SQ(1)>127 THEN 3030 ELSE LOCATE 2*X(I)+2
,Y(I)+1
3040 PEN J+1:PRINT C(10):NEXT:PP=PP+1:IF PP=3 TH
EN C="PERDU":GOTO 6000
3050 X0=INT(RND*17)+1:Y0=INT(RND*21)+1:IF T(X0,Y
0)<>17 OR Y0=11 THEN 3050
3060 FOR J=1 TO 11:PEN J+1:LOCATE 22,23+PP:SOUND
1,300+INT(RND*90),5
3070 IF SQ(1)>127 THEN 3070 ELSE PRINT C(13):PEN
J:LOCATE 2*X0+2,Y0+1

```

```

3080 PRINT C(13):NEXT:T(X0,Y0)=13:GOTO 1000
3970 '
3980 '     Poke-man attrape un fantome
3990 '
4000 FOR J=1 TO 3:IF Z<>9+J THEN 4080 ELSE T(X0,
Y0)=17
4010 LOCATE 2*X0+2,Y0+1:PRINT "   ":T(X0+U,Y0+V)=
13:X0=X0+U
4030 Y0=Y0+V:PEN 3:LOCATE 2*X0+2,Y0+1:PRINT C(13
):FOR I=1 TO 11
4040 PEN I-1:LOCATE 2*X0+2,Y0+1:PRINT C(10):SOUN
D 1,200+INT(RND*90),4
4050 IF SQ(1)>127 THEN 4050 ELSE PEN I:LOCATE 2*
X0+2,Y0+1
4060 PRINT C(13):SOUND 1,200+INT(RND*90),4
4070 IF SQ(1)>127 THEN 4070 ELSE NEXT:X(J)=7+J:Y
(J)=11
4080 NEXT:F=F+1:PEN 1:LOCATE 4,25:PRINT F::IF F<
10 THEN 2160
4090 C="GAGNE":GOTO 6000
4970 '
4980 '     Nouvelle super-pastille
4990 '
5000 FOR I=1 TO P:J=1-16*(RND<0.5):K=2-18*(RND<0
.5):IF T(J,K)<>17 THEN 5030
5010 FOR L=1 TO 15:PEN L:LOCATE 2*J+2,K+1:SOUND
1,70-INT(RND*30),3
5020 IF SQ(1)>127 THEN 5020 ELSE PRINT C(15):NEX
T:T(J,K)=15
5030 NEXT:RETURN
5970 '
5980 '     Nouvelle partie
5990 '
6000 WINDOW 12,28,1,3:PAPER 2:CLS:PEN 0:LOCATE 2
,2:PRINT "VOUS AVEZ ";C
6010 IF INKEY$<>" " THEN 6010 ELSE WINDOW 1,40,1,
25:FOR I=1 TO 3000:NEXT
6020 PAPER 0:C=INKEY$:IF C="O" THEN RUN ELSE IF
C="N" THEN CLS:END
6030 PEN INT(RND*3)+1:LOCATE 5,5:PRINT "UNE AUTR
E PARTIE ( O ou N )"
6040 FOR I=1 TO 150:NEXT:GOTO 6020
6970 '

```

```

6980 '      Initialisation
6990 '
7000 MODE 1:INK 0,0:INK 1,16:INK 2,9:INK 3,24:PA
PER 0:BORDER 0
7010 DEFINT D-Z:DEFSTR A-C:DIM C(17),T(18,22)
7020 SYMBOL 250,0,0,6,15,15,4,0,0:SYMBOL 251,0,2
,7,13,24,13,7,2
7030 SYMBOL 252,0,0,128,192,192,128,0,0:SYMBOL 2
53,7,15,31,50,63,127,112,63
7040 SYMBOL 254,0,128,192,96,224,240,112,224:C="
"
7050 C(0)=CHR$(154)+CHR$(154):C(1)=CHR$(149)+" "
:C(2)=CHR$(150)+CHR$(154)
7060 C(3)=CHR$(156)+" ":C(4)=CHR$(147)+CHR$(154)
:C(5)=CHR$(153)+" "
7070 C(6)=CHR$(158)+CHR$(154):C(7)=CHR$(155)+CHR
$(154):C(8)=CHR$(157)+" "
7080 C(9)=CHR$(151)+CHR$(154):C(10)=CHR$(253)+CH
R$(254):C(11)=C(10):C(12)=C(10)
7090 C(13)=CHR$(251)+CHR$(252):C(15)=CHR$(250)+"
":C(17)=C
7100 PEN 2:LOCATE 1,10:PRINT "CHOISISSEZ LA DIFF
ICULTE ( 1 ou 2 )"
7110 A=INKEY$:X=RND:IF A="" THEN 7110 ELSE P=VAL
(A):IF P<1 OR P>2 THEN 7110
7120 X0=9:Y0=17:FOR I=1 TO 3:Y(I)=11:X(I)=7+I:Z(
I)=17:NEXT
7970 '
7980 '      Decor
7990 '
8000 CLS:FOR J=0 TO 22:LOCATE 2,J+1:FOR I=0 TO 1
8:READ X
8010 PRINT C(X)::T(I,J)=X:NEXT:NEXT:PEN 3:LOCATE
4,3
8020 PRINT C(15):LOCATE 36,3:PRINT C(15):LOCATE
4,21:PRINT C(15)
8030 LOCATE 36,21:PRINT C(15):LOCATE 20,18:PRINT
C(13):PEN 1
8040 LOCATE 18,12:PRINT C(10);C(10);C(10):LOCATE
2,24
8050 PRINT "FANTOMES CAPTURES":LOCATE 5,25:PRINT
"D SUR 10";
8060 PEN 3:LOCATE 22,24:PRINT C(13):LOCATE 22,25
:PRINT C(13)::GOTO 1000

```

```

8970 '
8980 '      Datas du labyrinthe
8990 '
9000 DATA 2,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,3
9010 DATA 1,17,17,17,17,17,17,17,17,17,17,17,17,17,
17,17,17,17,1
9020 DATA 1,15,0,0,6,0,17,0,0,17,0,0,17,0,6,0,0,
15,1
9030 DATA 1,17,17,17,1,17,17,17,17,17,17,17,17,1
7,1,17,17,17,1
9040 DATA 9,0,0,17,17,17,0,0,17,0,17,0,0,17,17,1
7,0,0,8
9050 DATA 1,17,17,17,1,17,17,17,17,17,17,17,17,1
7,1,17,17,17,1
9060 DATA 1,17,0,0,5,17,0,0,0,17,0,0,0,17,4,0,0,
17,1
9070 DATA 1,17,17,17,17,17,17,17,17,17,17,17,17,
17,17,17,17,17,1
9080 DATA 1,17,0,17,0,0,0,17,0,0,0,17,0,0,0,17,0
,17,1
9090 DATA 1,17,17,17,17,17,17,17,17,17,17,17,17,
17,17,17,17,17,1
9100 DATA 9,0,17,0,0,0,17,2,0,0,0,3,17,0,0,0,17,
0,8
9110 DATA 1,17,17,17,17,17,17,1,10,11,12,1,17,17
,17,17,17,17,1
9120 DATA 9,0,0,17,0,0,17,4,0,0,0,5,17,0,0,17,0,
0,8
9130 DATA 1,17,17,17,17,17,17,17,17,17,17,17,17,
17,17,17,17,17,1
9140 DATA 1,17,0,17,0,0,0,17,0,0,0,17,0,0,0,17,0
,17,1
9150 DATA 1,17,17,17,17,17,17,17,17,17,17,17,17,
17,17,17,17,17,1
9160 DATA 1,17,0,0,3,17,0,0,0,17,0,0,0,17,2,0,0,
17,1
9170 DATA 1,17,17,17,1,17,17,17,17,13,17,17,17,1
7,1,17,17,17,1
9180 DATA 9,0,0,17,17,17,0,0,17,0,17,0,0,17,17,1
7,0,0,8
9190 DATA 1,17,17,17,1,17,17,17,17,17,17,17,17,1
7,1,17,17,17,1
9200 DATA 1,15,0,0,7,0,17,0,0,17,0,0,17,0,7,0,0,
15,1

```

9210 DATA 1,17,17,17,17,17,17,17,17,17,17,17,17,
17,17,17,17,17,1
9220 DATA 4,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,5

LE LANCE-MISSILES

PREMIER TABLEAU

Plusieurs soucoupes ennemies apparaissent en haut de l'écran et se laissent glisser vers le sol. Le joueur peut, en poussant la manette, déplacer le lance-missiles sur la droite ou sur la gauche. L'appui sur la touche FIRE provoque la mise à feu d'un missile qui détruit inmanquablement la soucoupe qui se trouve sur sa trajectoire. Toute soucoupe abattue est remplacée par un nouveau vaisseau, mais celui-ci repart du haut de l'image. Méfiez-vous des soucoupes qui deviennent brusquement bleu clair : elles se préparent à bombarder votre base de tir. Sortez le plus rapidement possible de leur ligne de mire et cela sans essayer d'appuyer sur FIRE ; les soucoupes claires ont en effet le pouvoir de paralyser momentanément votre système de tir.

DEUXIÈME TABLEAU

A certains moments l'ordinateur, sans doute devant la trop grande habileté du joueur à détruire ses soucoupes, change de tactique. Il ordonne à tous les vaisseaux d'aller se mettre à l'abri dans l'espace et lance une soucoupe kamikaze sur la base de tir. Est-il besoin de le préciser, il faudra nécessairement la toucher avant qu'elle n'atteigne son but. Le jeu reprendra alors avec le premier tableau.

TROISIÈME TABLEAU

Si l'ordinateur s'aperçoit qu'il a bien peu de chances de l'emporter dans ce jeu, il essaiera de gagner la partie en vous forçant à utiliser la totalité de votre stock de munitions. Dans ce tableau, vous aurez affaire à un couple de soucoupes qui traversent l'écran dans un sens puis dans l'autre en perdant chaque fois de l'altitude. Vous n'aurez rien d'autre à faire que d'appuyer sur FIRE quand vous verrez l'une d'elles à la verticale de votre base de tir. Naturellement, tout n'est pas si simple car les soucoupes flairent le danger. Alors qu'elles sont sur le point de traverser votre ligne de mire, vous observerez qu'elles font parfois brusquement demi-tour. Et le missile qui devait aller à leur rencontre ira se perdre dans l'espace. Que cela ne se reproduise pas trop souvent, le nombre de missiles n'est pas illimité !

FIN DU JEU

On est déclaré gagnant si l'on arrive au total de 30 soucoupes touchées.

Il y a trois possibilités pour que l'ordinateur sorte vainqueur du jeu :

- le stock de missiles est épuisé (on en a 60 au départ) ;
- une soucoupe s'est posée ;
- une soucoupe a bombardé la base de tir.

C'est en fonction du niveau de difficulté qu'est déterminée la vitesse de déplacement des soucoupes.

MODIFICATIONS POSSIBLES _____

| | Instructions BASIC | N° de ligne |
|---|--------------------|-------------|
| Nombre de missiles | MI=60 | 8010 |
| Nombre de soucoupes à abattre | Z < 30 | 2020 |
| Fréquence des changements de tableau | IF RND < 0.01 | 1040 |
| Fréquence d'apparition des soucoupes blanches | IF RND < 0.6 | 1060 |

JEU SANS MANETTE _____

La base sera déplacée en appuyant sur l'une des flèches horizontales et le tir sera déclenché en enfonçant la barre d'espace.

- Dans la ligne 1120, remplacer

D=JOY(0)

par :

D = - 16*(W=CHR\$(32))

- Dans la ligne 1180, remplacer

D=JOY(0)

par :

W=INKEY\$:D = - 4*(W=CHR\$(242)) - 8*(W=CHR\$(243))

- Dans les lignes 5040, 5050, 5110 et 5120, remplacer

GOSUB 1120


par :

GOSUB 1115

- Ajouter la ligne 1115 suivante :

1115 W=INKEY\$

LE CŒUR DU PROGRAMME



```

940 '
950 '
960 '      LE LANCE - MISSILES
970 '
980 '
990 GOTO 8000
1000 PEN 1:C=2*INT(RND*6)+1:IF Y(C)=-1 THEN Y(C)
=0:LOCATE 2*C+8,1:PRINT S:GOTO 1030
1010 LOCATE 2*C+8,Y(C)+1:PRINT "  ":Y(C)=Y(C)+1:
LOCATE 2*C+8,Y(C)+1
1020 PRINT S:IF Y(C)=19 THEN S="PERDU":GOTO 7000
1030 FOR K=1 TO NV:GOSUB 1180
1040 IF RND<0.01 THEN 3000 ELSE Y1=Y(X):GOSUB 11
20
1050 IF D THEN IF Y(X)>-1 THEN Y=Y(X):GOSUB 2000
:Y(X)=-1
1060 NEXT:IF RND<0.6 OR Y(X)=-1 THEN 1000 ELSE L
OCATE 2*X+8,Y(X)+1
1070 PEN 3:PRINT S:X0=X:Y0=Y(X):FOR I=1 TO 20*NV
:GOSUB 1180
1080 IF D THEN LOCATE 2*X0+8,Y0+1:PRINT S:GOTO 1
000
1090 NEXT:FOR I=1 TO 15:MOVE 128+32*X,16*(24-Y(X
))
1100 DRAW 128+32*X,100,I:SOUND 1,400+20*I,5
1110 IF SQ(1)>127 THEN 1110 ELSE NEXT:S="PERDU":
GOTO 7000
1120 D=JOY(0):IF D<16 THEN D=0:RETURN ELSE ENV 1
,2,5,1,1,0,70
1130 SOUND 1,50,30,0,1,1:PEN 3:FOR M=19 TO Y1+2
STEP -2:LOCATE 2*X+8,M
1140 PRINT CHR$(209):FOR L=1 TO 30:NEXT:LOCATE 2
*X+8,M:PRINT "  ":NEXT:MI=MI-1
1150 LOCATE 36,25:PRINT MI:;PEN 1:IF MI THEN RET
URN ELSE FOR I=1 TO 15
1160 PEN I:LOCATE 27,25:PRINT "MISSILE : 0":;SOU
ND 1,100-INT(RND*60),3
1170 IF SQ(1)>127 THEN 1170 ELSE NEXT:S="PERDU":
GOTO 7000
1180 D=JOY(0):D=(D=4)-(D=8):IF D=0 THEN RETURN E
LSE PEN 2:X=X+D
1190 X=X-(X=0)+(X=12):LOCATE 2*X+5,20:PRINT R:PE
N 1:FOR L=1 TO 200:NEXT:RETURN

```

```

1970 '
1980 '      Explosion d'une soucoupe
1990 '
2000 ENV 2,15,-1,6:SOUND 7,0,0,0,2,0,30:FOR I=1
TO 12:PEN I
2010 LOCATE 2*X+8,Y+1:PRINT CHR$(244)+CHR$(244):
FOR L=1 TO 20:NEXT:NEXT:Z=Z+1
2020 PEN 3:LOCATE 20,25:PRINT Z::IF Z<30 THEN RE
TURN ELSE S="GAGNE":GOTO 7000
2970 '
2980 '      Remontee des soucoupes
2990 '
3000 FOR I=1 TO 11 STEP 2:IF Y(I)<0 THEN 3030 EL
SE FOR J=Y(I) TO 0 STEP -1
3010 LOCATE 2*I+8,J+1:PRINT S:SOUND 1,80-INT(RND
*50),3
3020 IF SQ(1)>127 THEN 3020 ELSE LOCATE 2*I+8,J+
1:PRINT " ":NEXT
3030 NEXT:ENV 1,8,1,10,3,-3,5:SOUND 1,405,0,4,1
3040 IF SQ(1)>127 THEN 3040 ELSE IF RND<0.5 THEN
5000
3970 '
3980 '      Soucoupe kamikaze
3990 '
4000 J=INT(RND*9)+2:JM=J:JG=J-2:JD=J+2:FOR I=0 T
O 18:J=J+INT(RND*3)-1
4010 J=J-(J=JG)+(J=JD):LOCATE 2*J+8,I+1:PRINT S
4020 Y1=I:FOR K=1 TO 1-20*(ABS(X-JM)>1)
4030 GOSUB 1180:IF ABS(X-JM)<2 THEN K=21
4040 GOSUB 1120:PEN 1:IF D THEN IF X=J THEN Y=I:
GOSUB 2000:GOTO 6000
4050 NEXT:LOCATE 2*J+8,I+1:PRINT " ":NEXT:LOCAT
E 2*J+8,20
4060 PRINT S:S="PERDU":GOTO 7000
4970 '
4980 '      Va et vient soucoupes
4990 '
5000 PEN 2:FOR I=X TO 6 STEP SGN(6-X):LOCATE 2*I
+5,20:PRINT R
5010 FOR L=1 TO 900:NEXT:NEXT:X=6:PEN 1:FOR I=0
TO 16 STEP 2:Y1=I
5020 FOR J=7 TO 27 STEP 2:LOCATE J+1,I+1:PRINT T
:IF J<>13 OR RND>0.6 THEN 5050

```

```

5030 FOR K=15 TO 7 STEP -2:LOCATE K+1,I+1:PRINT
U:FOR J=1 TO NV
5040 GOSUB 1120:NEXT:NEXT:GOTO 5020
5050 FOR K=1 TO NV:GOSUB 1120:IF D=0 THEN 5080
5060 IF J<>15 AND J<>17 THEN 5080 ELSE Y=I:GOSUB
2000
5070 IF SQ(1)>127 THEN 5070 ELSE X=X+1+2*(J=15):
GOSUB 2000:X=6:GOTO 6000
5080 NEXT:NEXT:LOCATE 30,I+1:PRINT "    ":I=I+2
5090 FOR J=27 TO 7 STEP -2:LOCATE J+1,I+1:PRINT
U:IF J<>21 OR RND>0.6 THEN 5120
5100 FOR K=21 TO 27 STEP 2:LOCATE K+1,I+1:PRINT
T:FOR J=1 TO NV
5110 GOSUB 1120:NEXT:NEXT:GOTO 5090
5120 FOR K=1 TO NV:GOSUB 1120:IF D=0 THEN 5150
5130 IF J<>17 AND J<>19 THEN 5150 ELSE Y=I:GOSUB
2000
5140 IF SQ(1)>127 THEN 5140 ELSE X=X+1+2*(J=17):
GOSUB 2000:X=6:GOTO 6000
5150 NEXT:NEXT:LOCATE 8,I+1:PRINT "    ":NEXT
5160 LOCATE 7,20:PRINT U:S="PERDU":GOTO 7000
5970 '
5980 '     Retour des soucoupes
5990 '
6000 PEN 1:FOR I=1 TO 11 STEP 2:IF Y(I)<0 THEN 6
040 ELSE FOR J=0 TO Y(I)-1
6010 LOCATE 2*I+8,J+1:PRINT S:SOUND 1,80-INT(RND
*50),4
6020 IF SQ(1)>127 THEN 6020 ELSE LOCATE 2*I+8,J+
1:PRINT "    ":NEXT
6030 LOCATE 2*I+8,Y(I)+1:PRINT S
6040 NEXT:GOTO 1000
6970 '
6980 '     Nouvelle partie
6990 '
7000 WINDOW 12,28,1,3:PAPER 2:CLS:PEN 0:LOCATE 2
,2:PRINT "VOUS AVEZ ":S
7010 WINDOW 1,40,1,25:PAPER 0:PEN 3:FOR I=1 TO 5
000:NEXT
7020 IF INKEY$<>"" THEN 7020 ELSE S="UNE AUTRE P
ARTIE ( O ou N )"
7030 W=INKEY$:IF W="O" THEN RUN ELSE IF W="N" TH
EN CLS:END

```

```

7040 PEN INT(RND*3)+1:LOCATE 6,5:PRINT S:FOR I=1
  TO 150:NEXT:GOTO 7030
7970 '
7980 '   Initialisation
7990 '
8000 MODE 1:INK 0,0:INK 1,6:INK 2,18:INK 3,23:BO
RDER 0:PAPER 0:DEFSTR R-W
8010 DEFINT A-P,X-Z:DIM Y(12):MI=60:FOR I=0 TO 1
  2:Y(I)=-1:NEXT
8020 SYMBOL 240,0,0,0,31,63,121,224,224:SYMBOL 2
  41,3,3,15,31,248,240,224,0
8030 SYMBOL 242,0,0,192,227,127,62,28,0:SYMBOL 2
  43,0,0,0,224,240,120,28,28
8040 SYMBOL 244,20,65,20,65,20,65,8,34:SYMBOL 24
  5,0,32,65,70,45,31,34,0
8050 SYMBOL 246,0,4,130,98,180,248,68,0:SYMBOL 2
  47,192,240,252,255,255,255,255
8060 SYMBOL 248,255,255,255,255,255,255,248,192:
  SYMBOL 249,3,15,63,255,255,255,255,255
8070 SYMBOL 250,255,255,255,255,255,255,31,3:SYM
  BOL 251,255,255,153,153,153,255,255,255
8080 SYMBOL 252,255,255,159,159,159,255,255,255:
  SYMBOL 253,255,255,249,249,249,255,255,255
8090 SYMBOL 254,255,255,255,255,255,255,255,255:
  S=CHR$(245)+CHR$(246)
8100 T=" "+S+S:U=S+S+" ":R=" "+CHR$(240)+CHR$(
  241)+CHR$(242)+CHR$(243)+" "
8110 PEN 3:LOCATE 1,10:PRINT "CHOISISSEZ LA DIFF
  ICULTE (DE 1 A 3 )"
8120 X=RND:W=INKEY$:IF W="" THEN 8120 ELSE NV=VA
  L(W):IF NV<1 OR NV>3 THEN 8120
8970 '
8980 '   Decor
8990 '
9000 CLS:FOR I=0 TO 2:LOCATE 2*I+1,7+3*I:PRINT C
  HR$(251)+CHR$(247)
9010 FOR J=8+3*I TO 22-I:X=INT(RND*3):LOCATE 2*I
  +1,J:PRINT CHR$(251+X)+CHR$(254)
9020 NEXT:LOCATE 2*I+1,J:PRINT CHR$(254)+CHR$(24
  8):NEXT:FOR I=0 TO 2
9030 LOCATE 35+2*I,13-3*I:PRINT CHR$(249)+CHR$(2
  53):FOR J=14-3*I TO 20+I
9040 X=INT(RND*3):LOCATE 35+2*I,J:PRINT CHR$(254
  )+CHR$(251+X):NEXT

```

```
9050 LOCATE 35+2*I,J:PRINT CHR$(250)+CHR$(254):N  
EXT:MOVE 96,78  
9060 DRAWR 448,0,1:PEN 2:FOR I=7 TO 34 STEP 3:LO  
CATE I,22:PRINT CHR$(244)  
9070 MOVE 16*I-9,46:DRAWR 0,-16,1:MOVER 2,0:DRAW  
R 0,16:NEXT:PEN 3:LOCATE 1,25  
9080 PRINT "SOUCOUPES TOUCHEES: 0":LOCATE 27,25  
:PRINT "MISSILES: 60":  
9090 PEN 2:LOCATE 17,20:PRINT R:X=6:ENT 1,70,1,1  
:NV=2*(4-NV)-1:GOTO 1000
```

L'ÉCUREUIL ET LES LUTINS

Il était une fois, dans une forêt, quatre lutins qui voulaient attraper un écureuil. Mais celui-ci ne tenait pas une seconde en place et sautait continuellement de branche en branche. Nos quatre personnages, pas assez habiles pour poursuivre le gracieux animal dans les airs, se cachèrent donc derrière des arbres et attendirent que l'écureuil veuille bien descendre sur le sol. Ils attendraient encore aujourd'hui si la ... météo n'était venue à leur aide. Cela s'est passé de la façon suivante : le temps a brusquement changé et de gros nuages ont fait leur apparition dans le ciel. L'écureuil, pressentant l'arrivée prochaine de l'hiver, s'est souvenu qu'il lui fallait faire des provisions et il est donc descendu de ses branches pour venir ramasser noisettes et autres fruits secs. Et naturellement, les lutins l'ont attrapé.

Vous voulez des précisions sur cette capture ? Alors tenez, voici les détails : les écureuils, comme vous le savez, ne se rendent à terre qu'en empruntant l'intérieur des arbres creux. Ils s'y laissent glisser et, dès qu'ils trouvent une ouverture au ras du sol, se faufilent dehors. Et qu'ont fait les méchants lutins pendant que le gentil écureuil faisait sa cueillette ? Ils ont bouché le trou de l'arbre par lequel il était venu. Oui, ils ont fait cela !

Hein, quoi, qu'est ce qu'ils ont mes lutins ? Ils ne cherchent pas à attraper l'écureuil, dites-vous... Mais j'y pense, est-ce que vous avez poussé la manette d'un côté ou de l'autre, au moins ? Non, alors sachez que pour qu'un lutin ferme l'ouverture par laquelle vient de se glisser un écureuil, il suffit de pousser la manette dans la direction qu'indique la flèche dessinée sous l'arbre creux qui a servi de passage. Élémentaire, n'est-ce pas ?

FIN DU JEU

Chaque écureuil capturé rapporte trois points à l'heureux joueur qui aura su jouer de sa manette avec habileté. En contrepartie, l'ordinateur, à qui rien n'échappe, accordera un point à l'adversaire du joueur qui aura bougé sa manette un peu trop tôt ou un peu trop tard.

Notre écureuil est un animal doué d'un sixième sens et il est capable de savoir qu'un orage ne sera que de courte durée ; il ne descend pas à terre dans ce cas-là. Évitez donc de donner un point à votre adversaire lorsque cette situation se présente.

La partie se termine dès que l'un des joueurs a atteint le score de 30 points.

MODIFICATIONS POSSIBLES _____

| | Instructions BASIC | N° de ligne |
|--|--------------------|-------------|
| Score du vainqueur | IF L(I) < 30 | 5000 |
| Fréquence des changements de flèches | IF Q > 0.8 | 1040 |
| Fréquence d'apparition des nuages "bidons" | IF Q < 0.15 | 1030 |

JEU SANS MANETTE _____

Pour utiliser le programme tel qu'il est donné, on doit employer les quatre flèches du clavier (premier joueur) et la manette de jeu (deuxième joueur).

Si l'on ne dispose pas de manette, le second joueur devra se servir des touches 'W' (haut), 'S' (droite), 'A' (gauche) et 'Z' (bas).


Les modifications à réaliser sont les suivantes :

- Dans les lignes 1060, 1080 et 2030, remplacer

GOSUB 1120

par :

GOSUB 1140



```

940 '
950 '
960 '      L'ECUREUIL ET LES LUTINS
970 '
980 '
990 GOTO 8000
1000 FOR X=F TO G STEP H:N=0
1010 IF INT((X-1)/11)<>(X-1)/11 THEN LOCATE X+1+
(H=1),9:PRINT T(2-H):GOTO 1060
1020 N=1:LOCATE X-H+1,9:PRINT "   ":LOCATE X-H+1
,10:PRINT "   ":LOCATE X+1,9:PRINT T(1-H)
1030 LOCATE X+1,10:PRINT T(6-H):Q=RND:IF X=C THE
N 2000 ELSE IF Q<0.15 THEN 1080
1040 IF Q>0.8 THEN GOSUB 6000 ELSE FOR I=1 TO 50
:NEXT
1050 LOCATE X+1,9:PRINT "   ":LOCATE X+1,10:PRINT
"   "
1060 GOSUB 1120:IF E(0) OR E(1) THEN 3000 ELSE N
EXT X
1070 F=35-F:G=35-G:H=-H:GOTO 1000
1080 PEN 3:N=1:FOR I=1 TO 6:LOCATE I,1:PRINT R:G
OSUB 1120
1090 IF E(0) OR E(1) THEN LOCATE 1,1:PRINT SPACE
$(38):GOTO 3000
1100 NEXT:LOCATE 1,1:PRINT SPACE$(38):PEN 3::LOC
ATE X+1,9
1110 PRINT "   ":LOCATE X+1,10:PRINT "   ":GOTO 10
60
1120 E(0)=JOY(0):E(1)=0:V=INKEY$:IF V="" THEN RE
TURN ELSE P=ASC(V)
1130 E(1)=- (P=240)-2*(P=241)-4*(P=242)-8*(P=243)
:RETURN
1140 E(0)=0:E(1)=0:V=INKEY$:IF V="" THEN RETURN
ELSE P=ASC(V)
1150 E(1)=- (P=240)-2*(P=241)-4*(P=242)-8*(P=243)
1160 E(0)=- (P=87)-2*(P=90)-4*(P=65)-8*(P=83):RET
URN
1970 '
1980 '      L'ecureuil descend
1990 '
2000 LOCATE X+1,9:PRINT "   ":LOCATE X+1,10:PRINT
"   " :L=3:FOR I=1 TO NV
2010 PEN 3:LOCATE X+L,19:PRINT "   ":LOCATE X+L,2
0:PRINT "   ":L=7-L

```

```

2020 LOCATE X+L,19:PRINT T(0):LOCATE X+L,20:PRIN
T T(5):LOCATE I,1:PRINT R
2030 GOSUB 1120:IF E(0)=D OR E(1)=D THEN 4000
2040 NEXT:LOCATE X+3,19:PRINT "   ":LOCATE X+3,2
0:PRINT "   ":LOCATE X+1,9
2050 PRINT T(1-H):LOCATE X+1,10:PRINT T(6-H)
2060 LOCATE 1,1:PRINT SPACE$(38):GOSUB 6040:GOTO
1060
2970 '
2980 '   Rate ...
2990 '
3000 IF N THEN PEN 3:LOCATE X+1,9:PRINT T(1-H):L
OCATE X+1,10:PRINT T(6-H)
3010 FOR I=1 TO 12:PEN I:FOR J=0 TO 1:IF E(J)=0
THEN 3040
3020 LOCATE 11,4+J:PRINT "RATE ... ";W(J):SOUND
1,300+20*I,5
3030 IF SQ(1)>127 THEN 3030 ELSE LOCATE 38-22*J,
25:PRINT "+1"
3040 NEXT:NEXT:FOR I=0 TO 1:L(1-I)=L(1-I)-(E(I)<
>0):NEXT:GOSUB 5000
3050 IF N=0 THEN 1060 ELSE LOCATE X+1,9:PRINT "
"
3060 LOCATE X+1,10:PRINT "   ":GOSUB 6040:GOTO 10
60
3970 '
3980 '   L'ecureuil est attrape
3990 '
4000 PAPER 1:LOCATE X+1,17:PRINT "   ":LOCATE X+1
,18:PRINT "   ":PEN 2
4010 LOCATE X+1,19:PRINT T:LOCATE X+1,20:PRINT U
:PAPER 0:LOCATE X+L,19
4020 PRINT "   ":LOCATE X+L,20:PRINT "   ":MOVE 16
*X,134:DRAWR 16,4,2
4030 DRAWR 0,-32,2:FOR I=1 TO 12:PEN I:SOUND 1,8
0-INT(RND*50),4
4040 IF SQ(1)>127 THEN 4040 ELSE LOCATE X+4,19:P
RINT T(0):LOCATE X+4,20
4050 PRINT T(5):FOR J=0 TO 1:IF E(J)=D THEN LOCA
TE 16+22*J,25:PRINT "+3";
4060 NEXT:NEXT:LOCATE 1,1:PRINT SPACE$(38):FOR I
=0 TO 1:L(I)=L(I)-3*(E(I)=D)
4070 NEXT:GOSUB 5000:PAPER 1:PEN 0:LOCATE X+1,19
:PRINT T:LOCATE X+1,20

```

```

4080 PRINT U:PEN 2:LOCATE X+1,17:PRINT T:LOCATE
X+1,18:PRINT U
4090 MOVE 16*X,134:DRAWR 16,4,2:PAPER 0:GOSUB 60
40
4100 IF INKEY$<>"" THEN 4100 ELSE 1060
4970 '
4980 '   Scores
4990 '
5000 L=0:PEN 3:FOR I=0 TO 1:LOCATE 12+22*I,25:PR
INT L(I)::IF L(I)<30 THEN 5020
5010 PEN 3:LOCATE 4,2*I+2:PRINT W(I);" A GAGNE":
L=1
5020 NEXT:IF L THEN 7000 ELSE RETURN
5970 '
5980 '   Changement de fleches
5990 '
6000 FOR I=1 TO 1+INT(RND*2):FOR J=0 TO 3:C(J)=C
(J)*2:IF C(J)=16 THEN C(J)=1
6010 PEN 3:LOCATE 2+11*J,22:PRINT R(C(J))
6020 LOCATE 2+11*J,23:PRINT R(C(J)+5):SOUND 1,60
,3
6030 IF SQ(1)>127 THEN 6030 ELSE NEXT:NEXT
6040 I=INT(RND*3):IF I=I0 THEN 6040 ELSE I0=I:D=
C(I):C=1+11*I
6050 PEN 3:LOCATE X+1,9:PRINT "   ":LOCATE X+1,10
:PRINT "   ":RETURN
6970 '
6980 '   Nouvelle partie
6990 '
7000 FOR I=1 TO 2000:NEXT:U=" UNE AUTRE PARTIE (
  O ou N )"
7010 IF INKEY$<>"" THEN 7010
7020 W=INKEY$:IF W="O" THEN RUN ELSE IF W="N" TH
EN CLS:END
7030 PEN INT(RND*3)+1:LOCATE 5,14:PRINT U:FOR I=
1 TO 150:NEXT:GOTO 7020
7970 '
7980 '   Initialisation
7990 '
8000 MODE 1:INK 0,0:INK 1,16:INK 2,9:INK 3,23:PA
PER 0:BORDER 0
8010 DEFINT A-P,X-Z:DEFSTR R-W:SYMBOL AFTER 227
8020 SYMBOL 228,56,88,152,145,18,60,56,56:SYMBOL
229,48,48,24,24,48,48,56,60

```

```

8030 SYMBOL 230,0,0,0,1,3,7,15,15:SYMBOL 231,0,0
,0,128,192,224,240,240
8040 SYMBOL 232,62,127,255,255,255,127,60,0:SYMB
OL 233,56,252,254,254,254,252,120,0
8050 SYMBOL 234,24,12,6,3,1,0,0,0:SYMBOL 235,0,1
,3,6,12,24,63,63
8060 SYMBOL 236,24,48,96,192,128,0,0,0:SYMBOL 23
7,0,128,192,96,48,24,252,252
8070 SYMBOL 238,99,51,27,15,7,3,0,0:SYMBOL 239,2
4,48,96,192,128,0,0,0
8080 SYMBOL 240,0,0,3,7,15,27,51,99:SYMBOL 241,0
,0,0,128,192,96,48,24
8090 SYMBOL 242,0,192,192,63,63,79,3,0:SYMBOL 24
3,0,15,48,192,192,192,240,12
8100 SYMBOL 244,3,15,252,0,0,0,0,0:SYMBOL 245,19
2,240,63,0,0,0,0,0
8110 SYMBOL 246,0,240,12,3,3,3,15,48:SYMBOL 247,
0,3,3,252,252,242,192,0
8120 SYMBOL 248,0,0,0,48,112,240,48,48:SYMBOL 24
9,60,60,195,195,195,195,192,192
8130 SYMBOL 250,60,252,124,60,63,63,63,31:SYMBOL
251,48,48,12,12,12,12,240,240
8140 SYMBOL 252,60,60,195,195,195,195,3,3:SYMBOL
253,0,0,0,12,14,15,12,12
8150 SYMBOL 254,12,12,48,48,48,48,15,15:SYMBOL 2
55,60,63,62,60,252,252,252,248
8160 FOR I=1 TO 9:R=R+" "+CHR$(232)+CHR$(233):NE
XT:DIM R(13)
8170 R(1)=CHR$(240)+CHR$(241):R(2)=CHR$(209)+" "
:R(4)=CHR$(235)+CHR$(210)
8180 R(6)=CHR$(209)+" ":R(7)=CHR$(238)+CHR$(239)
:R(8)=CHR$(210)+CHR$(237)
8190 R(9)=CHR$(234)+" ":R(13)=" "+CHR$(236)
8200 T(0)=CHR$(252)+CHR$(253):T(1)=" "+CHR$(245)
+CHR$(246)+CHR$(247)
8210 T(2)=CHR$(248)+CHR$(249):T(3)=CHR$(242)+CHR
$(243)+CHR$(244)+" "
8220 T(5)=CHR$(254)+CHR$(255):T(7)=CHR$(250)+CHR
$(251)
8230 T=CHR$(230)+CHR$(231):U=CHR$(138)+CHR$(133)
8240 PEN 3:LINE INPUT "NOM DU PREMIER JOUEUR ";W
(0):W(0)=LEFT$(W(0),10)
8250 PRINT:LINE INPUT "NOM DU DEUXIEME JOUEUR ";
W(1):W(1)=LEFT$(W(1),10)

```

```

8260 W=INKEY$:Q=RND:IF W<>" " THEN 8290 ELSE LOCA
TE 1,20
8270 PRINT "CHOISISSEZ LA DIFFICULTE ( DE 1 A 3
)":PEN INT(RND*3)+1
8280 FOR I=1 TO 150:NEXT:GOTO 8260
8290 NV=VAL(W):IF NV<1 OR NV>3 THEN 8260
8970 '
8980 '   Decor
8990 '
9000 CLS:PAPER 2:WINDOW 1,40,21,21:CLS:WINDOW 1,
40,1,25:PAPER 0
9010 PEN 3:FOR I=0 TO 1:LOCATE 1+22*I,25:PRINT W
(I)::LOCATE 11+22*I,25
9020 PRINT ": 0":NEXT:FOR I=1 TO 34 STEP 11:PAP
ER 1:WINDOW I+1,I+2,12,20
9030 CLS:WINDOW 1,40,1,25:PEN 2:LOCATE I+1,17:PR
INT T:LOCATE I+1,18
9040 PRINT U:PAPER 0:LOCATE I,17:PRINT CHR$(228)
:LOCATE I,18:PRINT CHR$(229)
9050 PEN 3:LOCATE I,19:PRINT CHR$(143):LOCATE I,
20:PRINT CHR$(143)
9060 MOVE 16*I,134:DRAWR 16,4,2:PEN 1:LOCATE I+6
,20:PRINT CHR$(46)
9070 PAPER 1:PEN 0:LOCATE I+1,19:PRINT T:LOCATE
I+1,20:PRINT U:PAPER 0
9080 FOR J=0 TO 32 STEP 8:MOVE 16*I-16+J,238:DRA
WR 16,-16,2
9090 DRAWR 48,16,2:NEXT:NEXT:NV=2*(4-NV)-1:M=3:F
=1:G=34
9100 H=1:FOR I=0 TO 3:C(I)=2^I:NEXT:GOSUB 6000:G
OTO 1000

```

TIR SUR CIBLES

Neuf joueurs peuvent prendre part successivement à ce jeu. Il va s'agir pour chacun d'eux, d'obtenir le score le plus élevé possible. Chaque participant a la possibilité de diriger le vaisseau en poussant la manette à droite, à gauche, en haut ou en bas. Notons que, sans intervention de notre part, l'avion continue sa trajectoire en ligne droite... jusqu'à ce qu'il heurte un obstacle : il est alors détruit et le tour passe au joueur suivant. Le rôle des manettes est donc d'empêcher le vaisseau d'entrer en contact avec le pourtour de l'écran ou avec l'un des objets hétéroclites qui encombrent l'espace.

Reste à définir le rôle du bouton FIRE. L'appui de cette touche provoque, à partir de l'avion, la mise à feu d'un missile. La course de ce dernier à travers l'écran ne sera suspendue que lorsqu'il aura atteint son objectif. Voyons les différents cas possibles :

1. Un missile heurte le bord du décor : rien de particulier ne se passe.
2. Un missile rencontre une cible : le joueur se voit gratifier d'un nombre de points égal à la valeur affichée sur la cible. Allez-y donc sans retenue : détruisez le maximum de cibles.
3. Un missile heurte une mine jaune : celle-ci explose en détruisant tout ce qui se tient dans son environnement proche. Si c'est une cible, tant mieux, vous recevrez les points affichés sur elle. Si c'est une soucoupe, tant mieux aussi, vous gagnerez 50 ou même, si l'ordinateur est de bon poil, 200 points. Et si c'est votre vaisseau ? Mieux vaut ne pas y penser !
4. Un missile arrive sur une soucoupe : deux cas peuvent se présenter. Ou bien l'ordinateur vous attribue 50 points et programme l'explosion de la soucoupe, ou bien il ordonne à ladite soucoupe de riposter sur votre vaisseau. Malheureux vaisseau, il n'en réchappera pas : on ne connaît pas d'exemple de soucoupe qui ait raté sa cible. Faut-il alors en conclure qu'il vaut mieux laisser en paix les soucoupes, puisque de toute façon rien ne permet de distinguer sur l'écran à quelle catégorie elles appartiennent ? Oui, encore qu'il m'ait semblé que les soucoupes que l'on pouvait détruire sans risque avaient l'habitude, au moment où elles apparaissaient sur le téléviseur, de changer plusieurs fois de teinte avant de se colorer définitivement en rouge. Remarquez, je me fais peut-être des idées, tout cela va tellement vite !

FIN DU JEU

Le jeu s'arrête lorsque le vaisseau heurte un obstacle, lorsqu'une soucoupe riposte ou lorsque le stock de missiles (20 au départ) est épuisé.

MODIFICATIONS POSSIBLES

| | Instructions BASIC | N° de ligne |
|--------------------------------------|--------------------|-------------|
| Nombre de missiles | MI=20 | 9050 |
| Nombre de cibles | FOR I=1 TO 10 | 10070 |
| Nombre de mines | FOR I=0 TO 6 | 10100 |
| Fréquence d'apparition des soucoupes | IF RND < 0.05 | 1020 |

JEU SANS MANETTE

Il sera possible, avec les modifications suivantes, de diriger l'avion en appuyant sur l'une des quatre flèches et de lancer les missiles en enfonçant la barre d'espace.

- Dans les lignes 1000 et 10150, remplacer

```
B=JOY(0)
```

par :


```
V=INKEY$:B=- (V=CHR$(240))-2*(V=CHR$(241))-4*(V=CHR$(242))-8*(V=CHR$(243))
```

- Dans la ligne 1080, remplacer

```
IF JOY(0) > 15
```

par :

```
IF V=CHR$(32)
```



```

940 '
950 '
960 '          TIR   SUR   CIBLES
970 '
980 '
990 GOTO 8000
1000 PEN 2:B=JOY(0):FOR I=0 TO 3:IF B<>E(I) THEN
NEXT:GOTO 1020
1010 C=(B=4)-(B=8):D=(B=1)-(B=2):E=B
1020 IF RND<0.05 OR ( G(H)>800 AND RND<0.2 ) THE
N GOSUB 4000
1030 A(X,Y)=0:LOCATE 2*X+1,Y+1:PRINT "   ":X=X+C:
Y=Y+D:P=A(X,Y)
1040 IF P=0 THEN 1070 ELSE FOR I=1 TO 15:SOUND 1
,300+INT(RND*200),5
1050 IF SQ(1)>127 THEN 1050 ELSE LOCATE 2*X+1,Y+
1:IF I<9 THEN PRINT U(E) ELSE PRINT U
1060 PEN I:NEXT:GOTO 9000
1070 LOCATE 2*X+1,Y+1:PRINT U(E):A(X,Y)=1:G(H)=G
(H)+1:PEN 2:LOCATE 19,25
1080 PRINT G(H)::IF JOY(0)>15 THEN L=X+C:M=Y+D:T
=CHR$(209+55*(E>3)):GOTO 2000
1090 F=F-1:FOR I=1 TO F:NEXT:GOTO 1000
1970 '
1980 '          Missile
1990 '
2000 LOCATE 2*L+1,M+1:PRINT T:ON E GOTO 2010,20
40,1,2060,1,1,1,2030
2010 P=A(L,M):IF P THEN 2070 ELSE M=M-1:LOCATE 2
*L+1,M+1
2020 PRINT T:LOCATE 2*L+1,M+2:PRINT "   ":GOTO 20
10
2030 P=A(L,M):IF P THEN 2070 ELSE L=L+1:LOCATE 2
*L-1,M+1:PRINT "   ":T:GOTO 2030
2040 P=A(L,M):IF P THEN 2070 ELSE M=M+1:LOCATE 2
*L+1,M+1
2050 PRINT T:LOCATE 2*L+1,M:PRINT "   ":GOTO 2040
2060 P=A(L,M):IF P=0 THEN L=L-1:LOCATE 2*L+1,M+
1:PRINT T:"   ":GOTO 2060
2070 MI=MI-1:PEN 3:LOCATE 36,25:PRINT MI::IF MI
THEN 2100
2080 FOR I=1 TO 14:PEN I:LOCATE 28,25:SOUND 1,10
0+INT(RND*200),4

```

```

2090 IF SQ(1)>127 THEN 2090 ELSE PRINT "MISSILE
  0";:NEXT:GOTO 9000
2100 IF P=4 THEN PEN 3:LOCATE 2*L+1,M+1:PRINT CH
R$(143)+CHR$(143):PEN 2:GOTO 1000
2110 IF P=3 THEN PEN 1:LOCATE 2*L+1,M+1:PRINT S:
GOTO 5000
2120 IF P<10 THEN 3000 ELSE ENV 2,13,-1,6:SOUND
7,0,0,0,2,0,20
2130 FOR I=1 TO 12:PEN I:LOCATE 2*L+1,M+1:PRINT
U:NEXT:G(H)=G(H)+P
2140 PEN 2:LOCATE 19,25:PRINT G(H);:A(L,M)=0:GOT
O 1000
2970 '
2980 '      Mine touchee
2990 '
3000 PEN 2:LOCATE 2*L+1,M+1:PRINT CHR$(244):FOR
I=-2 TO 2:FOR J=-2 TO 2
3010 P=A(L+I,M+J):LOCATE 2*(L+I)+1,M+J+1:SOUND 1
,60-INT(RND*40),2
3020 IF SQ(1)>127 THEN 3020 ELSE PRINT CHR$(244)
+" ":IF P=1 THEN 1040
3030 IF P=3 THEN GOSUB 6000:PEN 2
3040 IF P<10 THEN 3070 ELSE FOR K=1 TO 15:SOUND
1,60-INT(RND*40),2
3050 IF SQ(1)>127 THEN 3050 ELSE LOCATE 2*(L+I)+
1,M+J+1
3060 PEN K:PRINT U:NEXT:G(H)=G(H)+P:PEN 2:LOCATE
19,25:PRINT G(H);:PEN 2
3070 A(L+I,M+J)=0:NEXT:NEXT:WINDOW 2*L-3,2*L+6,M
-1,M+3
3080 CLS:WINDOW 1,40,1,25:GOTO 1000
3970 '
3980 '      Nouvelle soucoupe
3990 '
4000 I=INT(RND*18)+1:J=INT(RND*22)+1:IF A(I,J) T
HEN RETURN
4010 IF RND<0.5 THEN PEN 1:LOCATE 2*I+1,J+1:PRIN
T S:A(I,J)=3:PEN 2:RETURN
4020 FOR K=5 TO 13:PEN K:LOCATE 2*I+1,J+1
4030 PRINT S:NEXT:A(I,J)=50:PEN 2:RETURN
4970 '
4980 '      Riposte d'une soucoupe
4990 '

```

```

5000 IF T=CHR$(209) THEN 5040 ELSE FOR I=1 TO 5:
IO=SGN(X-L)
5010 PEN I+2*(I=4):FOR J=2*L+2+2*IO TO 2*X+1-2*I
0 STEP 2*IO
5020 LOCATE J,Y+1:PRINT CHR$(154):SOUND 1,100+10
*J+30*I,4
5030 IF SQ(1)>127 THEN 5030 ELSE LOCATE J,Y+1:PR
INT " ":NEXT:NEXT:GOTO 1040
5040 FOR I=1 TO 5:IO=SGN(Y-M):PEN I+2*(I=4):FOR
J=M+1+IO TO Y+1-IO STEP IO
5050 LOCATE 2*X+1,J:PRINT CHR$(209):SOUND 1,100+
10*J+30*I,4
5060 IF SQ(1)>127 THEN 5060 ELSE LOCATE 2*X+1,J:
PRINT " ":NEXT:NEXT:GOTO 1040
5970 '
5980 '      Bonus + 200
5990 '
6000 FOR K=1 TO 15:PAPER 3:PEN K:LOCATE 10,1:PRI
NT "BONUS ---> 200"
6010 PAPER 0:LOCATE 2*(L+I)+1,M+J+1:PRINT S:SOUN
D 1,200-10*K,4
6020 IF SQ(1)>127 THEN 6020 ELSE NEXT:PEN 1:LOCA
TE 2*(L+I)+1,M+J+1
6030 PRINT U:G(H)=G(H)+200:PEN 2:LOCATE 19,25:PR
INT G(H):;PEN 3:RETURN
6970 '
6980 '      Nouvelle partie
6990 '
7000 IF INKEY$<>"" THEN 7000 ELSE U="UNE AUTRE P
ARTIE ? ( 0 ou N )"
7010 V=INKEY$:IF V="0" THEN RUN ELSE IF V="N" TH
EN CLS:END
7020 PEN INT(RND*3)+1:LOCATE 5,24:PRINT U:FOR I=
1 TO 150:NEXT:GOTO 7010
7970 '
7980 '      Initialisation
7990 '
8000 MODE 1:INK 0,0:INK 1,16:INK 2,24:INK 3,10:B
ORDER 0:PAPER 0:DEFSTR R-W
8010 SYMBOL 244,0,0,2,7,7,2,0,0:SYMBOL 245,0,32,
65,70,45,31,34,0
8020 SYMBOL 246,0,4,130,98,180,248,68:SYMBOL 247
,1,1,3,7,13,25,1,2

```

```

8030 SYMBOL 248,128,128,192,224,176,152,128,64:S
YMBOL 249,2,1,25,13,7,3,1,1
8040 SYMBOL 250,64,128,152,176,224,192,128,128:S
YMBOL 251,0,1,3,31,31,3,1,0
8050 SYMBOL 252,192,128,8,240,240,8,128,192:SYMB
OL 253,3,1,32,31,31,32,1,3
8060 SYMBOL 254,0,128,192,248,248,192,128,0:SYMB
OL 255,168,2,144,0,146,8,130,40
8070 U=CHR$(255)+CHR$(255):U(1)=CHR$(247)+CHR$(2
48):U(2)=CHR$(249)+CHR$(250)
8080 U(4)=CHR$(251)+CHR$(252):U(8)=CHR$(253)+CHR
$(254)
8090 PEN 3:LOCATE 1,5:PRINT "CHOISISSEZ LA DIFFI
CULTE ( DE 1 A 5 )"
8100 X=RND:V=INKEY$:IF V="" THEN 8100 ELSE F=VAL
(V):IF F<1 OR F>5 THEN 8100
8110 FO=100*(5-F):PEN 2:LOCATE 1,13:PRINT "QUEL
EST LE NOMBRE DE JOUEURS ?"
8120 V=INKEY$:IF V="" THEN 8120 ELSE HO=VAL(V):I
F HO<1 THEN 8120
8130 CLS:FOR I=1 TO HO:LOCATE 1,2*I:PRINT "NOM D
U JOUEUR NUMERO":I;
8140 LINE INPUT R(I):R(I)=LEFT$(R(I),10):NEXT
8150 FOR I=0 TO 3:E(I)=2^I:NEXT:DIM A(19,24):CLS
8970 '
8980 '           Scores
8990 '
9000 PEN 3:FOR J=0 TO 23:FOR I=0 TO 19:A(I,J)=0:
NEXT:NEXT:CLS
9010 FOR I=1 TO HO:LOCATE 1,2*I:PRINT R(I):LOCAT
E 15,2*I:PRINT "":G(I)
9020 LOCATE 23,2*I:PRINT "POINT":IF G(I)>1 THEN
PRINT "S"
9030 NEXT:H=H+1:IF H>HO THEN 7000 ELSE PEN 1:LOC
ATE 1,23
9040 PRINT R(H);" , APPUYEZ SUR COPY"
9050 IF INKEY$<>CHR$(224) THEN 9050 ELSE MI=20:F
=FO
9970 '
9980 '           Decor
9990 '
10000 CLS:PAPER 3:WINDOW 1,40,1,1:CLS:WINDOW 1,4
0,24,24:CLS

```

```

10010 WINDOW 1,2,1,24:CLS:WINDOW 39,40,1,24:CLS:
WINDOW 1,40,1,25:PAPER 0
10020 PEN 1:LOCATE 1,25:PRINT R(H)::PEN 2:LOCATE
 13,25:PRINT "SCORE: 0";
10030 PEN 3:LOCATE 28,25:PRINT "MISSILES: 20":F
OR I=0 TO 19
10040 A(I,0)=4:A(I,23)=4:NEXT:FOR J=0 TO 23:A(0,
J)=4:A(19,J)=4:NEXT
10050 PEN 1:S=CHR$(245)+CHR$(246):FOR I=1 TO 8:X
=INT(RND*18)+1
10060 Y=INT(RND*22)+1:A(X,Y)=3:LOCATE 2*X+1,Y+1:
PRINT S:NEXT
10070 PAPER 3:PEN 0:FOR I=1 TO 10:X=INT(RND*18)+
1
10080 Y=INT(RND*22)+1:IF A(X,Y) THEN 10080 ELSE
C=INT(RND*90)+10
10090 A(X,Y)=C:LOCATE 2*X+1,Y+1:PRINT RIGHT$(STR
$(C),2):NEXT
10100 PAPER 0:PEN 2:FOR I=0 TO 6:X=INT(RND*14)+3
10110 Y=INT(RND*18)+3:IF A(X,Y) THEN 10110 ELSE
A(X,Y)=2
10120 LOCATE 2*X+1,Y+1:PRINT CHR$(244):NEXT:PEN
3:X=INT(RND*18)+1
10130 Y=INT(RND*22)+1:IF A(X,Y) THEN 10130 ELSE
PEN 2:LOCATE 2*X+1,Y+1
10140 PRINT CHR$(143)+CHR$(143):A(X,Y)=1
10150 B=JOY(0):FOR I=0 TO 3:IF B<>E(I) THEN NEXT
:GOTO 10150
10160 E=B:C=(B=4)-(B=8):D=(B=1)-(B=2):GOTO 1000

```

LE POISSON PILOTE

Voici un jeu qui permettra à deux joueurs de mesurer la qualité de leurs réflexes : il va s'agir, pour chacun d'eux, d'attraper le plus possible de ces poissons que l'on voit nager dans l'étang.

Mais, autant le dire tout de suite, ces poissons sont des animaux capricieux. Ils passent le plus clair de leur temps à parcourir la mare en tournant dans un sens, puis dans l'autre. Ils ne donnent pas l'impression d'être affamés et dédaignent les appâts accrochés au bout des lignes des quatre pêcheurs assis au bord de l'eau... jusqu'au moment où ils se décident enfin. Et là, tout s'accélère : la canne de l'un des pêcheurs ploie, son bouchon s'enfonce et un bruit de sonnette retentit. Pas une seconde à perdre, il faut ferrer le poisson : un petit coup sec sur la manette de jeu fera l'affaire. Attention toutefois à pousser la manette du bon côté, à droite si c'est le pêcheur de droite qui a une touche, en haut si c'est celui du haut qui est concerné, etc. Difficile de rester maître de soi, n'est-ce pas ? Surtout quand on sait que, si l'on rate son coup, on risque fort de voir le poisson se faire attraper par son adversaire avec, pour ce dernier, un capital de trois points qui viendra s'ajouter à son score.

Ah, j'allais oublier : évitez de relever la ligne d'un pêcheur avant qu'un poisson ne s'y soit intéressé, vous ne feriez qu'ajouter un point au score de votre adversaire. Vous riez, vous pensez que cela n'arrive qu'aux autres ? Alors laissez-moi vous préciser que la sonnette qui indique qu'un poisson s'est fait prendre se met parfois à tinter toute seule, comme ça, sans raison. Le vent, peut-être ?

FIN DU JEU

Le jeu est terminé et l'ordinateur propose une nouvelle partie, dès que le score d'un des joueurs est égal à 30.

Le niveau de difficulté établit le temps pendant lequel les poissons tirent sur les lignes des pêcheurs.

MODIFICATIONS POSSIBLES

| | Instructions BASIC | N° de ligne |
|------------------------------------|--------------------|-------------|
| Nombre de poissons à attraper | IF E(K) > 29 | 5010 |
| Fausses alertes | IF RND < 0.98 | 1060 |
| Durée de la promenade des poissons | C1=INT(RND*50) | 1000 |

JEU SANS MANETTE

Pour utiliser le programme tel qu'il est donné, on doit employer les quatre flèches du clavier (premier joueur) et la manette de jeu (second joueur).

Si l'on ne dispose pas de manette, le second joueur devra se servir des touches 'W' (haut), 'S' (droite), 'A' (gauche), et 'Z' (bas).

Les modifications à réaliser sont les suivantes :

- Dans les lignes 1020, 1080 et 2030, remplacer

```
K(0)=JOY(0)
```

par :

```
K(0)=0
```


- Ajouter les trois lignes :

```
1025 K(0) = -(V = CHR$(87)) - 2*(V = CHR$(90)) - 4*(V = CHR$(65)) - 8*(V = CHR$(83))
```

```
1085 K(0) = -(V = CHR$(87)) - 2*(V = CHR$(90)) - 4*(V = CHR$(65)) - 8*(V = CHR$(83))
```

```
2035 K(0) = -(V = CHR$(87)) - 2*(V = CHR$(90)) - 4*(V = CHR$(65)) - 8*(V = CHR$(83))
```

LE CŒUR DU PROGRAMME



```

940 '
950 '
960 '           LE POISSON PILOTE
970 '
980 '
990 GOTO 8000
1000 C1=INT(RND*50):C2=0:I=2^INT(RND*4):FF=I(I):
GG=J(I)
1010 PAPER 0:PEN 3:FOR KO=K1 TO K2 STEP K3:C2=C2
+1:LOCATE F(KO),G(KO)
1020 PRINT CHR$(250+H(KO)):K(0)=JOY(0):K(1)=0:V=
INKEY$:IF V="" THEN 1040
1030 K(1)=-((V=CHR$(240))-2*(V=CHR$(241))-4*(V=CH
R$(242))-8*(V=CHR$(243)))
1040 IF K(0) OR K(1) THEN 6000
1050 IF C2>C1 AND F(KO)=FF AND G(KO)=GG THEN 200
0
1060 LOCATE F(KO),G(KO):PRINT " ":IF RND<0.98 TH
EN NEXT:GOTO 1110
1070 FOR J=1 TO 12:FOR K=1 TO 70:NEXT:SOUND 1,50
,1
1080 K(0)=JOY(0):K(1)=0:V=INKEY$:IF V="" THEN 11
00
1090 K(1)=-((V=CHR$(240))-2*(V=CHR$(241))-4*(V=CH
R$(242))-8*(V=CHR$(243)))
1100 IF K(0) OR K(1) THEN 6000 ELSE NEXT
1110 K2=-36*(K2=0):K1=K0-K3:K3=-K3:GOTO 1010
1970 '
1980 '           Le poisson a mordu
1990 '
2000 LOCATE FF,GG:PRINT " ":FOR J=1 TO NV:PEN 1:
P=2+2*(I=2 OR I=4):PAPER P
2010 SOUND 1,50,1:LOCATE A(I),B(I):PRINT S:PAPER
0:LOCATE C(I),D(I):PRINT " "
2020 PAPER P:LOCATE A(I),B(I):PRINT R(I):PAPER 0
:PEN 2:LOCATE C(I),D(I)
2030 PRINT R:K(0)=JOY(0):K(1)=0:V=INKEY$:IF V=""
THEN 2050
2040 K(1)=-((V=CHR$(240))-2*(V=CHR$(241))-4*(V=CH
R$(242))-8*(V=CHR$(243)))
2050 IF K(0)=I AND K(1)=I THEN 4000 ELSE IF K(0)
=I THEN K=0:GOTO 3000
2060 IF K(1)=I THEN K=1:GOTO 3000 ELSE NEXT:GOTO
1000

```

```

2970 '
2980 '      Poisson peche par l'un
2990 '
3000 K6=0:FOR J=1 TO 7
3010 P=2+2*(I=2 OR I=4):PEN 1:PAPER P:LOCATE A(I
),B(I):PRINT S
3020 LOCATE C(I),D(I):PAPER 0:PEN 3:PRINT TO:K5=
23+2*K:PEN 1:PAPER 2
3030 LOCATE 1,K5:PRINT U(K)::LOCATE 13,K5:PRINT
"+3":SOUND 1,300+20*J,3
3040 IF SQ(1)>127 THEN 3040 ELSE PAPER P:LOCATE
A(I),B(I):PRINT R(I)
3050 LOCATE C(I),D(I):PAPER 0:PEN 3:PRINT T1:PEN
3:PAPER 2
3060 LOCATE 1,K5:PRINT U(K)::LOCATE 13,K5:PRINT
"+3":SOUND 1,400+20*J,3
3070 IF SQ(1)>127 THEN 3070 ELSE IF K6 THEN RETU
RN ELSE NEXT:PAPER 0:PEN 2
3080 LOCATE C(I),D(I):PRINT R:E(K)=E(K)+3:GOSUB
5000:GOTO 4030
3970 '
3980 '      Poisson peche par les deux
3990 '
4000 K6=1:FOR J=1 TO 4:FOR K=0 TO 1:GOSUB 3010:N
EXT:NEXT:PAPER 0:PEN 2
4020 LOCATE C(I),D(I):PRINT R:E(0)=E(0)+3:E(1)=E
(1)+3:GOSUB 5000
4030 IF INKEY$<>"" THEN 4030 ELSE 1000
4970 '
4980 '      Scores
4990 '
5000 K6=0:PEN 3:PAPER 2:FOR K=0 TO 1:LOCATE 9,23
+2*K:PRINT E(K);"  ";
5010 IF E(K)>29 THEN LOCATE 10,10+2*K:PRINT U(K)
;" A GAGNE":K6=1
5020 NEXT:IF K6 THEN 7000 ELSE RETURN
5970 '
5980 '      Ligne relevee trop tot
5990 '
6000 LOCATE F(K0),G(K0):PRINT " ":FOR J=1 TO 8:F
OR K=0 TO 1:IF K(K)=0 THEN 6040
6010 K5=25-2*K:PAPER 0:PEN J:LOCATE 10,14:PRINT
"ET ALORS , ";U(K);" ?"

```

```

6020 PAPER 2:LOCATE 13,K5:PRINT "+1";:SOUND 1,30
0-20*J,3
6030 IF SQ(1)>127 THEN 6030
6040 NEXT:NEXT:FOR K=0 TO 1:IF K(K) THEN E(1-K)=
E(1-K)+1
6050 NEXT:GOSUB 5000:GOTO 1110
6970 '
6980 '     Nouvelle partie
6990 '
7000 PAPER 0:PEN 3:LOCATE 7,7:PRINT "UNE AUTRE P
ARTIE ?"
7010 V=INKEY$:IF V="0" THEN RUN ELSE IF V="N" TH
EN CLS:END
7020 PEN INT(RND*3)+1:LOCATE 12,8:PRINT "(0 ou N
)":FOR I=1 TO 90:NEXT:GOTO 7010
7970 '
7980 '     Initialisation
7990 '
8000 MODE 1:INK 0,0:INK 1,16:INK 2,9:INK 3,23:BO
RDER 0:PAPER 0:SYMBOL AFTER 230
8010 DEFINT A-P,X-Z:DEFSTR R-W:DIM F(36),G(36),H
(36)
8020 SYMBOL 231,254,124,56,16,16,144,144,80:SYMB
OL 232,81,49,50,18,20,20,24,24
8030 SYMBOL 233,252,131,128,128,128,128,128:
SYMBOL 234,0,192,56,7,0,0,0,0
8040 SYMBOL 235,0,0,0,0,224,24,6,1:SYMBOL 236,0,
0,0,0,7,24,96,128
8050 SYMBOL 237,0,3,28,224,0,0,0,0:SYMBOL 238,63
,193,1,1,1,1,1,1
8060 SYMBOL 239,48,48,48,48,48,48,240,240:SYMBOL
240,0,0,0,0,0,0,63,63
8070 SYMBOL 241,24,28,62,126,255,255,255,255:SYM
BOL 242,124,254,124,252,252,120,24,24
8080 SYMBOL 243,12,12,12,12,12,12,15,15:SYMBOL 2
44,0,0,0,0,0,0,252,252
8090 SYMBOL 246,24,56,124,126,255,255,255,255:SY
MBOL 247,62,127,62,63,63,30,24,24
8100 SYMBOL 248,3,3,1,0,0,0,0,0:SYMBOL 249,192,1
92,128,0,0,0,0,0
8110 SYMBOL 250,255,255,255,195,195,195,195,195:
SYMBOL 251,0,0,0,60,195,60,0,0
8120 SYMBOL 252,16,16,40,40,40,40,16,16:SYMBOL 2
53,0,2,12,20,40,80,96,128

```

```

8130 SYMBOL 254,0,64,48,40,20,10,6,1:R(1)=CHR$(2
33)+CHR$(234)+CHR$(235)
8140 R(2)=CHR$(236)+CHR$(237)+CHR$(238):R(8)=R(1
):R(4)=R(2)
8150 R=CHR$(248)+CHR$(249):T0=" "+CHR$(254):T1=C
HR$(253)+" "
8160 RESTORE 10020:FOR I=0 TO 36:READ F(I),G(I),
H(I):NEXT:RESTORE
8170 PEN 3:FOR I=0 TO 1:PRINT "NOM DU JOUEUR":I+
1::LINE INPUT U(I)
8180 U(I)=LEFT$(U(I),8):PRINT:NEXT:PEN 1:LOCATE
1,20
8190 PRINT "CHOISISSEZ LA DIFFICULTE ( DE 1 A 3
)"
8200 W=INKEY$:Q=RND:IF W="" THEN 8200 ELSE NV=VA
L(W):IF NV<1 OR NV>3 THEN 8200
8210 CLS:NV=3*(4-NV):BORDER 9:K2=36:K3=1:S=CHR$(
95)+CHR$(95)+CHR$(95)
8970 '
8980 '      Decor
8990 '
9000 J=14:PAPER 2:FOR I=2 TO 8:LOCATE 1,I:PRINT
SPACE$(J):LOCATE 25+J,16+I
9010 PRINT SPACE$(16-J):J=J-2:NEXT:J=1:FOR I=1 T
O 15 STEP 2:LOCATE 1,I+9
9020 PRINT SPACE$(J):LOCATE 16+J,I:PRINT SPACE$(
25-J):J=J+2:LOCATE 1,I+10
9030 PRINT SPACE$(J)::LOCATE 16+J,I+1:PRINT SPAC
E$(25-J):J=J+1:NEXT
9040 WINDOW 1,16,1,1:CLS:WINDOW 25,40,25,25:CLS:
WINDOW 1,40,1,25
9050 FOR I=2 TO 4 STEP 2:READ A(I),B(I):PEN 1:A=
A(I)-5:B=B(I)+3:LOCATE A,B
9060 PRINT CHR$(247):LOCATE A,B+1:PRINT CHR$(246
):LOCATE A,B+2
9070 PRINT CHR$(143)+CHR$(244):PEN 3:LOCATE A,B+
3:PRINT CHR$(250)::PEN 1
9080 PRINT CHR$(243):MOVE 16*A+64,16*(28-B):DRAW
R -64,-64,1
9090 MOVER 0,2:DRAWR 64,64:MOVE 16*A+110,16*(28-
B):DRAWR 0,-32,1:PAPER 0
9100 LOCATE A+5,B-3:PRINT R(2):PEN 2:LOCATE A+7,
B:PRINT R:PAPER 2:NEXT

```

```

9110 FOR I=1 TO 8 STEP 7:READ A(I),B(I):PEN 1:A=
A(I)+6:B=B(I)+2:LOCATE A,B
9120 PRINT CHR$(242):LOCATE A,B+1:PRINT CHR$(241
):LOCATE A-1,B+2
9130 PRINT CHR$(240)+CHR$(143):LOCATE A-1,B+3:PR
INT CHR$(239):;PEN 3
9140 PRINT CHR$(250):MOVE 16*A-16,16*(24-B):DRAW
R -48,48,1
9150 MOVER 2,0:DRAWR 48,-48:MOVE 16*A-112,16*(27
-B):DRAWR 0,-48,1:PEN 1
9160 LOCATE A-6,B-2:PRINT R(1):PAPER 0:PEN 2:LOC
ATE A-7,B+2:PRINT R:PAPER 2
9170 NEXT:FOR I=1 TO 5:READ A:FOR J=1 TO 3:READ
B:PEN 1-2*(RND<0.5)
9180 LOCATE B,A:PRINT CHR$(232):LOCATE B,A-1:PRI
NT CHR$(231):NEXT:NEXT:PEN 3
9190 LOCATE 1,23:PRINT U(0):LOCATE 10,23:PRINT "
0":LOCATE 1,25:PRINT U(1);
9200 LOCATE 10,25:PRINT "0":;FOR I=0 TO 3:READ I
(2^I),J(2^I):NEXT
9210 FOR I=2 TO 4 STEP 2:C(I)=A(I)+2:D(I)=B(I)+3
:NEXT
9220 FOR I=1 TO 8 STEP 7:C(I)=A(I)-1:D(I)=B(I)+4
:NEXT:GOTO 1000
9970 '
9980 '      Datas du parcours
9990 '
10000 DATA 21,19,6,9,21,1,33,9,2,3,29,37,5,2,6,3
4
10010 DATA 7,1,31,39,19,4,7,11,23,33,37,40,20,6,
23,23,8,13,32,14
10020 DATA 29,22,3,31,21,3,33,20,3,35,19,3,37,18
,3,38,16,4
10030 DATA 36,14,4,34,14,1,32,14,1,31,12,2,30,10
,4,28,9,4
10040 DATA 26,8,1,24,6,4,22,6,1,20,6,1,19,4,2,17
,2,4
10050 DATA 15,2,3,13,3,2,11,4,3,9,5,3,7,6,1,5,7,
3
10060 DATA 4,9,2,5,11,4,7,12,2,8,13,4,11,15,4,13
,16,4
10070 DATA 15,17,2,17,19,4,19,20,4,21,22,1,23,23
,4,25,24,1,27,23,3

```

**LA VIE
DE CHÂTEAU**

Le but du jeu est d'empêcher les soucoupes qui apparaissent en haut de l'écran de venir se poser sur le château. On dispose pour cela d'une navette et d'un hélicoptère.

LA NAVETTE

On la fait monter ou descendre en poussant la manette vers le haut ou vers le bas. Lorsque l'on se trouve au niveau d'une soucoupe rouge, l'appui sur le bouton FIRE provoque la destruction de cette dernière. Tout irait pour le mieux dans le meilleur des mondes si l'ordinateur ne nous réservait quelques surprises désagréables dont voici un aperçu :

- Les soucoupes touchées sont automatiquement remplacées par de nouveaux vaisseaux qui repartent du haut du téléviseur.
- Certaines soucoupes sont pendant de courts instants rendues invulnérables. Cela se voit au fait qu'elles passent du rouge au jaune. Il est fortement conseillé de ne pas rester trop longtemps à leur hauteur car elles sont armées et laissent bien peu de chances aux navettes qui ne veulent pas changer d'altitude.
- On voit de temps en temps descendre du ciel une soucoupe bleue qui vient s'intercaler entre la navette et la flotte ennemie. Ne vous avisez pas de tirer sur cette catégorie de soucoupes ; la riposte serait foudroyante. Laissez-les donc en paix et continuez de ne vous intéressez qu'aux soucoupes rouges. Quand vous jugerez que le nombre de vaisseaux bleus est devenu trop important pour que vous puissiez continuer le combat sans risque, vous pousserez la manette sur la gauche pour appeler l'hélicoptère à votre rescousse.

L'HÉLICOPTÈRE

Il est sorti d'un souterrain et prend son envol dans le bruit de ses pales. Il a, par rapport à la navette, une supériorité très réelle : il est plus lourdement armé et peut donc détruire tous les types de soucoupes. Profitez-en pour réduire en cendres tous les vaisseaux que vous pourrez et en particulier ceux qui étaient prêts à se poser. Hélas, la durée de vie de l'hélicoptère est limitée. Il sera lui-même transformé en poussière par une soucoupe venue d'on ne sait où quand le temps qui lui est imparti, pas bien important au départ il est vrai, sera tombé à zéro. Le jeu reprendra alors avec la navette.

FIN DU JEU

Pour être le vainqueur de ce jeu, il faut arriver au total de 30 soucoupes touchées. On est déclaré perdant si l'on épuise sa réserve de missiles (40 au départ), si les deux navettes sont abattues par des soucoupes ou si une soucoupe arrive à se poser.

Le niveau de difficulté règle d'une part le temps au bout duquel les soucoupes jaunes tirent et d'autre part la vitesse de déplacement des soucoupes.

MODIFICATIONS POSSIBLES

| | Instructions BASIC | N° de ligne |
|---|--------------------|--------------|
| Nombre de missiles | MI=40 | 8060 |
| Nombre de soucoupes à détruire | IFG=30 | 3060 et 5090 |
| Nombre de navettes | IFP=1 | 6030 |
| Fréquence d'apparition des soucoupes bleues | IFRND < 0.8 | 3100 |
| Fréquence d'apparition des soucoupes jaunes | IFRND < 0.7 | 1050 |
| Durée de vie de l'hélicoptère | IF TIME-Q > 1500 | 5000 |

JEU SANS MANETTE

On fera monter ou descendre la navette en enfonçant l'une des deux flèches verticales. L'arrivée de l'hélicoptère sera commandée par l'appui de la flèche gauche. Et la barre d'espacement déclenchera le tir.

- Dans la ligne 2000, remplacer

```
D=JOY(0)
```

par :

```
W=INKEY$:D=- (W=CHR$(240))-2*(W=CHR$(241))-4*(W=CHR$(242))
```

- Dans la ligne 3000, remplacer

```
IF JOY(0) < 16
```

par :

```
IF W < > CHR$(32)
```

- Dans la ligne 5000, remplacer

```
D=JOY(0)
```

par :

```
W=INKEY$:D=- (W=CHR$(240))-2*(W=CHR$(241))
```


- Dans la ligne 5030, remplacer

```
IF JOY(0) < 16
```

par :

```
IF W < > CHR$(32)
```

LE CŒUR DU PROGRAMME



```

940 '
950 '
960 '      LA   VIE   DE   CHATEAU
970 '
980 '
990 GOTO 7000
1000 PAPER 0:PEN 2:Y=8:FOR I=1 TO 8:LOCATE 2,I:P
RINT R2:SOUND 1,80-7*I,5
1010 IF SQ(1)>127 THEN 1010 ELSE LOCATE 2,I:PRIN
T "   ":NEXT:LOCATE 2,9:PRINT R2
1020 C=INT(RND*6)+3:PEN 1:LOCATE 4*C+5,Y(C)+1:PR
INT "   ":A(C,Y(C))=0
1030 Y(C)=Y(C)+1:LOCATE 4*C+5,Y(C)+1:PRINT R:IF
Y(C)>14 THEN 6040
1040 A(C,Y(C))=1:FOR I=1 TO NV:GOSUB 2000:IF D=4
THEN 4000 ELSE GOSUB 3000:NEXT
1050 IF RND<0.7 OR A(2,Y)=2 THEN 1020 ELSE FOR I
=3 TO 8
1060 IF A(I,Y)=0 THEN NEXT:GOTO 1020 ELSE PEN 2:
LOCATE 4*I+5,Y+1
1070 PRINT R:Y0=Y:FOR J=1 TO 30*NV:GOSUB 2000:IF
D THEN 1090
1080 NEXT:PEN 1:LOCATE 4*I+5,Y0+1:PRINT R:GOTO 6
000
1090 PEN 1:LOCATE 4*I+5,Y0+1:PRINT R:IF D=4 THEN
4000 ELSE 1020
1970 '
1980 '      Mouvements navette
1990 '
2000 D=JOY(0):IF D=4 THEN RETURN ELSE D=(D=1)-(D
=2):IF D=0 THEN RETURN
2010 PEN 2:LOCATE 2,Y+1:PRINT "   ":Y=Y+D:Y=Y-(Y=
-1)+(Y=16):LOCATE 2,Y+1
2020 PRINT R2:FOR K=1 TO 150:NEXT:RETURN
2970 '
2980 '      Tir navette
2990 '
3000 IF JOY(0)<16 THEN RETURN ELSE PEN 2:LOCATE
5,Y+1:FOR I=0 TO 8
3010 IF A(I,Y) THEN 3030 ELSE PRINT "----":NEXT
3020 MOVE 64,16*(24-Y)+8:DRAWR 576,0,0:GOTO 3070
3030 MOVE 64,16*(24-Y)+8:DRAWR 64*I,0,0:IF A(I,Y
)=2 THEN 6000

```

```

3040 SOUND 7,0,0,0,2,0,30:FOR K=1 TO 12:LOCATE 4
*I+5,Y+1
3050 PEN K:PRINT R4:NEXT:G=G+1:PAPER 3:PEN 0:LOC
ATE 37,22:PRINT G
3060 A(I,Y)=0:Y(I)=0:IF G=30 THEN S="GAGNE":GOTO
6040
3070 MI=MI-1:PAPER 3:PEN 0:LOCATE 37,24:PRINT MI
:IF MI THEN 3100
3080 FOR I=1 TO 15:LOCATE 28,24:PRINT "MISSILES:
0":SOUND 1,300+30*I,5
3090 IF SQ(1)>127 THEN 3090 ELSE PEN I:NEXT:GOTO
6040
3100 PAPER 0:IF RND<0.8 THEN RETURN ELSE PEN 3:E
NT 1,70,1,1
3110 IF SQ(2)>127 THEN 3110 ELSE ENV 1,8,1,10,3,
-3,5:SOUND 1,420,0,4,1
3120 FOR I=0 TO Y:LOCATE 9,I+1:PRINT R:FOR J=1 T
O 50:NEXT:LOCATE 9,I+1
3130 PRINT "      ":NEXT:FOR I=8 TO 12:LOCATE I,Y+
1:PRINT " ";R
3140 IF INKEY$<>"" THEN 3140 ELSE FOR J=1 TO 150
:NEXT:NEXT:A(2,Y)=2:RETURN
3970 '
3980 '      Sortie helicoptere
3990 '
4000 IF M=5 THEN 1020 ELSE FOR I=Y TO 0 STEP -1:
PEN 2:LOCATE 2,I+1:PRINT R2
4010 PEN INT(RND*3)+1:LOCATE 1,16:PRINT "APPEL H
ELICO":SOUND 1,50,1
4020 FOR J=1 TO 30:NEXT:LOCATE 2,I+1:PRINT "  ":
NEXT:LOCATE 1,16
4030 PRINT SPACE$(12):FOR I=4 TO 1 STEP -1:PEN 2
:LOCATE I,24:PRINT R6
4040 PEN 1:LOCATE I,25:PRINT R7:SOUND 1,200+3*I,
8:SOUND 2,205+3*I,8
4050 IF SQ(1)>127 THEN 4050 ELSE NEXT:PEN 2:FOR
I=23 TO 9 STEP -1
4060 LOCATE 1,I+1:PRINT "      ":FOR J=1 TO 5:LOCA
TE 1,I:PRINT R(H)
4070 H=1-H:NEXT:SOUND 1,800,3:NEXT:Y=8:M=M+1:Q=T
IME
4970 '
4980 '      Jeu avec l'helicoptere
4990 '

```

```

5000 IF TIME-Q>1500 THEN 5110 ELSE D=JOY(0):D=(D
=1)-(D=2):IF D=0 THEN 5030
5010 LOCATE 1,Y+1:PRINT "      ":Y=Y+D:Y=Y-(Y=-1)+
(Y=16)
5020 FOR I=1 TO 10:LOCATE 1,Y+1:PRINT R(H):H=1-H
:NEXT:GOTO 5000
5030 H=1-H:LOCATE 1,Y+1:PRINT R(H):IF JOY(0)<16
THEN 5000
5040 PEN 2:LOCATE 5,Y+1:FOR I=0 TO 8:IF A(I,Y) T
HEN 5060
5050 PRINT "----":NEXT:MOVE 64,16*(24-Y)+8:DRAW
R 576,0,0:GOTO 5000
5060 MOVE 64,16*(24-Y)+8:DRAWR 64*I,0,0
5070 SOUND 7,0,0,0,2,0,30:FOR K=1 TO 12:LOCATE 4
*I+5,Y+1
5080 PEN K:PRINT R4:NEXT:G=G+1:PAPER 3:PEN 0:LOC
ATE 37,22:PRINT G:PAPER 0
5090 PEN 2:A(I,Y)=0:Y(I)=0:IF G=30 THEN S="GAGNE
":GOTO 6040 ELSE 5000
5110 FOR I=0 TO Y:PEN 3:LOCATE 9,I+1:PRINT R:PEN
2:LOCATE 1,Y+1
5120 PRINT R(H):H=1-H:SOUND 1,400+INT(RND*200),6
5130 IF SQ(1)>127 THEN 5130 ELSE LOCATE 9,I+1:PR
INT "      ":NEXT:PEN 3
5140 LOCATE 9,Y+1:PRINT R:FOR I=1 TO 5:K=INT(RND
*3)+1:FOR J=8 TO 5 STEP -1:PEN K
5150 LOCATE J,Y+1:PRINT "-":PEN 2:LOCATE 1,Y+1:P
RINT R(H):H=1-H:SOUND 1,150,5
5160 IF SQ(1)>127 THEN 5160 ELSE LOCATE J,Y+1:PR
INT "      ":NEXT:NEXT
5170 PEN 2:FOR I=Y+1 TO 25:LOCATE 1,I:PRINT R4::
FOR J=1 TO 30:NEXT
5180 LOCATE 1,I:PRINT "      ":NEXT:PEN 3:FOR I=Y+
1 TO 1 STEP -1
5190 LOCATE 9,I:PRINT R:SOUND 1,400+INT(RND*200)
,6
5200 IF SQ(1)>127 THEN 5200 ELSE LOCATE 9,I:PRIN
T "      ":NEXT
5210 R6=RIGHT$(R6,LEN(R6)-4):R7=RIGHT$(R7,LEN(R7)
)-4):GOTO 1000
5970 '
5980 '      Fin du jeu
5990 '

```

```

6000 ENT 1,70,1,1:ENV 1,2,5,1,1,0,70:SOUND 1,100
,70,0,1,1:FOR K =1 TO 5
6010 FOR J=1 TO 12:MOVE 64*I+62,16*(24-Y)+8:DRAW
56,16*(24-Y)+8,J
6020 NEXT:NEXT:SOUND 7,0,0,0,2,0,30:PEN 2:LOCATE
2,Y+1:PRINT R3:P=P+1
6030 IF SQ(2)>127 THEN 6030 ELSE LOCATE 2,Y+1:PR
INT " ":IF P=1 THEN 1000
6040 PAPER 3:WINDOW 12,28,1,3:CLS:PEN 0:LOCATE 2
,2:PRINT "VOUS AVEZ ":S
6050 IF INKEY$<>"" THEN 6050 ELSE WINDOW 1,40,1,
25:PAPER 0
6060 W=INKEY$:IF W="0" THEN RUN ELSE IF W="N" TH
EN CLS:END
6070 LOCATE 5,6:PEN INT(RND*3)+1:FOR I=1 TO 150:
NEXT
6080 PRINT "UNE AUTRE PARTIE ? ( O ou N )":GOTO
6060
6970 '
6980 '      Initialisation
6990 '
7000 MODE 1:INK 0,0:INK 1,16:INK 2,24:INK 3,10:P
APER 0:BORDER 0:DEFINT A-P,X-Z
7010 DIM A(8,15):DEFSTR R-W:ENV 2,14,-1,6:S="PER
DU":SYMBOL AFTER 236
7020 SYMBOL 237,127,127,12,30,63,63,30,12:SYMBOL
238,254,254,48,120,252,252,120,48
7030 SYMBOL 239,0,36,36,36,36,36,0:SYMBOL 240
,168,2,144,0,162,8,130,40
7040 SYMBOL 241,0,240,14,3,31,15,28,0:SYMBOL 242
,0,0,0,192,240,255,24,0
7050 SYMBOL 243,0,0,0,192,127,192,0,0:SYMBOL 244
,63,0,0,3,255,3,0,0
7060 SYMBOL 245,252,4,255,240,240,240,255,0:SYMB
OL 246,0,0,240,48,254,240,192,0
7070 SYMBOL 248,0,0,0,3,255,3,0,0:SYMBOL 249,7,4
,255,240,240,255,0
7080 SYMBOL 250,255,0,240,48,254,240,192,0:SYMBOL
L 251,0,0,0,1,3,7,28,127
7090 SYMBOL 252,7,31,255,255,153,255,192,255:SYM
BOL 253,224,252,255,255,153,255,3,255
7100 SYMBOL 254,0,0,0,128,192,224,56,254:R=CHR$(
251)+CHR$(252)+CHR$(253)+CHR$(254)

```

```

7110 R(0)=CHR$(243)+CHR$(248)+CHR$(249)+CHR$(250)
):R2=CHR$(241)+CHR$(242)
7120 R(1)=CHR$(243)+CHR$(244)+CHR$(245)+CHR$(246)
):R3=CHR$(240)+CHR$(240)
7130 R4=R3+CHR$(240)+CHR$(240):R5=CHR$(237)+CHR$(
208)+CHR$(208)+CHR$(238)
7140 R6=R(0)+R(1)+R(0)+R(1)+R(0)+" ":R7=R5+R5+R5
+R5+R5+" "
7150 PEN 3:LOCATE 1,20:PRINT "CHOISISSEZ LA DIFF
ICULTE ( DE 1 A 3 )"
7160 W=INKEY$:Q=RND:IF W="" THEN 7160 ELSE NV=VA
L(W):IF NV<1 OR NV>3 THEN 7160
7970 '
7980 '      Decor
7990 '
8000 CLS:PAPER 3:WINDOW 5,40,18,25:CLS:PEN 0:FOR
I=1 TO 40
8010 LOCATE INT(RND*40)+1,INT(RND*5)+1:PRINT CHR
$(239);:NEXT:PAPER 0
8020 WINDOW 10,16,18,20:CLS:WINDOW 1,40,1,25:PAP
ER 3:FOR I=5 TO 39 STEP 2
8030 J=17-3*( I>9 AND I<17 ):LOCATE I,J:PRINT "
":NEXT
8040 PEN 2:PAPER 0:LOCATE 5,24:PRINT R6:PEN 1:LO
CATE 5,25:PRINT R7;
8050 PAPER 3:PEN 0:LOCATE 18,22:PRINT "SOUCOUPES
TOUCHEES: 0":LOCATE 28,24
8060 PRINT "MISSILES: 40";:MI=40:NV=4-NV:GOTO 10
00

```

**LE BASSET,
LE HÉRON ET
LE LAPIN**

L'ordinateur est à la fois juge et partie dans ce jeu. Juge car c'est lui qui établit les scores des deux joueurs et les départage au moment du verdict final. Partie car il a les moyens de réduire à zéro les prétentions des joueurs trop pressés... ou trop maladroits.

PREMIÈRE PHASE

Les lapins sont dans leurs terriers et n'ont aucune idée de ce qui se passe à l'extérieur. Les deux joueurs peuvent alors faire avancer leurs bassets aussi loin qu'ils le veulent en poussant la manette sur la droite ou en enfonçant la flèche droite. Il se peut, ce faisant, que l'un des chien-chiens atteigne l'entrée d'un terrier. Cela mettra son propriétaire très en colère, le fera passer par toutes les couleurs mais n'empêchera pas l'ordinateur d'attribuer 5 points au toutou émérite, ou plus exactement à l'heureux joueur qui le promène. Une fois ceci fait, le basset reprendra sa place sur la ligne de départ et essaiera de réitérer son exploit.

DEUXIÈME PHASE

Tout ne se passe pas toujours très bien pour les chiens, car les lapins, se souvenant sans doute qu'ils doivent monter la garde, mettent parfois le nez dehors. Tant mieux pour les joueurs si aucun d'eux n'est à ce moment là en train de faire avancer son animal de compagnie : les lapins rentreront leurs oreilles, réintégreront leurs habitats et le jeu reprendra son cours normal. Mais sachez que quand un lapin surprend l'un des joueurs (ou les deux) avec la main sur la manette ou sur la flèche droite, il s'empresse de donner l'alerte. Comment ? Nul ne le sait. Mais toujours est-il que l'on voit apparaître en haut et à droite de l'écran un héron qui se dirige sans tarder vers le basset repéré. Malheureux toutou, la pierre qu'il recevra sur la tête lui fera voir 36 chandelles ! Quant au joueur malhabile, il sera obligé de retourner sur la ligne de départ et son adversaire, sans avoir rien fait, aura gagné un point.

Vous remarquerez sans doute que les communications entre les lapins et leur ami le héron lanceur de cailloux, ne marchent pas toujours très bien. Il arrive que le héron se montre, survole les bassets et disparaisse sans procéder à un bombardement. C'est énervant car on n'est jamais certain d'avoir relâché la manette à temps, n'est-ce pas ?

FIN DU JEU _____

Lorsque l'un des joueurs atteint le score de 30 points, l'ordinateur affiche le nom du gagnant et propose une nouvelle partie.

Le niveau de difficulté détermine l'intervalle de temps qui sépare deux sorties successives des lapins.

MODIFICATIONS POSSIBLES _____

| | Instructions BASIC | N° de ligne |
|---|---|--------------|
| Score à obtenir pour gagner | IF Q(0) < 30 AND Q(1) < 30 IF Q(0) > 29 | 4000 4010 |
| Fréquence d'apparition des "hérons inoffensifs" | IF RND < 0.1 | 1100 |

JEU SANS MANETTE _____

Pour utiliser le programme tel qu'il est donné, on doit employer la flèche droite du clavier (premier joueur) et la manette de jeu (second joueur).

Si l'on ne dispose pas de manette, le second joueur devra faire avancer son basset en enfonçant la barre d'espace.


Les modifications à réaliser sont les suivantes :

- Dans les lignes 1040 et 1090, remplacer

INKEY(75)

par :

INKEY(47)



```

940 '
950 '
960 '          LE          BASSET
970 '
980 '
990 GOTO 5000
1000 F=E*(4*INT(RND*15)+4):PEN 1:FOR I=18 TO 24
STEP 6:PEN 1-2*(I=24)
1010 LOCATE 38,I-1:PRINT " ":LOCATE 38,I:PRINT U
:NEXT
1020 F=F-1:IF INKEY(1) THEN 1040 ELSE B=B+1
1030 PEN 1:LOCATE B-1,17:PRINT " ";U(G):G=1-G:IF
B=35 THEN 2000
1040 IF INKEY(75) THEN 1060 ELSE C=C+1
1050 PEN 3:LOCATE C-1,23:PRINT " ";U(H):H=1-H:IF
C=35 THEN 2000
1060 IF F THEN 1020 ELSE ENV 1,8,1,6,3,-3,5:SOUN
D 1,390,0,4,1
1070 PEN 1:LOCATE 38,17:PRINT CHR$(240):LOCATE 3
8,18:PRINT CHR$(241);" "
1080 PEN 3:LOCATE 38,23:PRINT CHR$(240):LOCATE 3
8,24:PRINT CHR$(241);" "
1090 FOR I=1 TO 150:IO=(INKEY(1)=0):I1=(INKEY(75
)=0)
1100 IF IO+I1 THEN 3000 ELSE NEXT:IF RND<0.1 THE
N 3000 ELSE 1000
1970 '
1980 '          Un basset arrive au but
1990 '
2000 M=-(C=35):FOR I=1 TO 20:PEN INT(RND*3)+1:PA
PER 0:LOCATE 38,18+6*M
2010 PRINT U:PAPER 2:LOCATE 19,19+6*M:PRINT "+5"
;:SOUND 1,55,3
2020 IF SQ(1)>127 THEN 2020 ELSE NEXT:PAPER 0:PE
N 1+2*M:LOCATE 38,18+6*M
2030 PRINT U:PAPER 2:LOCATE 19,19+6*M:PRINT " "
:Q(M)=Q(M)+5:PEN 0
2040 LOCATE 14,19+6*M:PRINT Q(M);:PAPER 0:PEN 2*
M+1:FOR I=34 TO 1 STEP -1
2050 LOCATE I,17+6*M:PRINT U(J)+" ":J=1-J:SOUND
1,20+I,3
2060 IF SQ(1)>127 THEN 2060 ELSE NEXT:IF M THEN
C=1 ELSE B=1

```

```

2070 GOSUB 4000:GOTO 1000
2970 '
2980 '      Deplacement du heron
2990 '
3000 IF SQ(1)>127 THEN 3000 ELSE J=0:B1=B:C1=C:F
OR I=37 TO 1 STEP -1
3010 PEN 3:LOCATE I,1:PRINT V(J):" ":J=1-J:SOUND
  1,800+INT(RND*200),5
3020 IF SQ(1)>127 THEN 3020 ELSE IF I<>B THEN 31
10
3030 FOR K=1 TO 7:PEN K-(K=4):LOCATE I,1:SOUND 1
,800+INT(RND*200),5
3040 IF SQ(1)>127 THEN 3040 ELSE PRINT V(J):J=1-
J:NEXT:IF I0=0 THEN 3110
3050 FOR K=2 TO 17:LOCATE I+1,K-1:PRINT " ":LOCA
TE I+1,K:PRINT CHR$(227)
3060 SOUND 1,30+K,2:LOCATE I,1:PRINT V(J):J=1-J:
NEXT:FOR K=1 TO 10:PAPER 0
3070 PEN K-2*(K=4 OR K=8):LOCATE B,17:PRINT U(J)
:J=1-J:SOUND 1,50+20*K,5
3080 IF SQ(1)>127 THEN 3080 ELSE PAPER 2:LOCATE
19,25:PRINT "+1":NEXT
3090 Q(1)=Q(1)+1:PEN 0:LOCATE 14,25:PRINT Q(1):P
APER 0:GOSUB 4000
3100 LOCATE B,17:PRINT " ":PEN 1:LOCATE 1,17:P
RINT U(0):B1=1
3110 IF I<>C THEN 3220
3120 FOR K=1 TO 7:PEN K-(K=4):LOCATE I,1:SOUND 1
,800+INT(RND*200),5
3130 IF SQ(1)>127 THEN 3130 ELSE PRINT V(J):J=1-
J:NEXT:IF I1=0 THEN 3220
3140 FOR K=4 TO 22 STEP 2:PAPER 0:PEN 3:LOCATE I
+1,K-2:PRINT " ":LOCATE I+1,K
3150 PRINT CHR$(227):SOUND 1,30+K,2:LOCATE I,1:P
RINT V(J):J=1-J
3160 PAPER 2:LOCATE I+1,18:PRINT " ":NEXT:PAPER
0
3170 LOCATE I+1,22:PRINT " ":FOR K=1 TO 10:PAPER
0
3180 PEN K-2*(K=4 OR K=8):LOCATE C,23:PRINT U(J)
:J=1-J:SOUND 1,50+20*K,5
3190 IF SQ(1)>127 THEN 3190 ELSE PAPER 2:LOCATE
19,19:PRINT "+1":NEXT

```

```

3200 Q(0)=Q(0)+1:PEN 0:LOCATE 14,19:PRINT Q(0):P
APER 0:GOSUB 4000
3210 LOCATE C,23:PRINT "   ":PEN 3:LOCATE 1,23:P
RINT U(0):C1=1
3220 NEXT:B=B1:C=C1:LOCATE 1,1:PRINT "   ":GOTO
1000
3970 '
3980 '   Nouvelle partie
3990 '
4000 IF Q(0)<30 AND Q(1)<30 THEN RETURN ELSE PAP
ER 2:WINDOW 12,31,3,5
4010 CLS:PEN 0:LOCATE 2,2:IF Q(0)>29 THEN PRINT
W(0); ELSE PRINT W(1);
4020 PRINT " A GAGNE":U="UNE AUTRE PARTIE ? ( O o
u N )"
4030 IF INKEY$<>"" THEN 4030 ELSE WINDOW 1,40,1,
25:PAPER 0
4040 PEN INT(RND*3)+1:LOCATE 9,12:PRINT U:W=INKE
Y$:IF W="O" THEN RUN
4050 IF W="N" THEN CLS:END ELSE FOR I=1 TO 150:N
EXT:GOTO 4040
4970 '
4980 '   Initialisation
4990 '
5000 MODE 1:INK 0,0:INK 1,16:INK 2,9:INK 3,23:BO
RDER 0:PAPER 0
5010 DEFINT B-Q:DEFSTR R-Z:SYMBOL AFTER 239
5020 SYMBOL 240,120,230,33,32,176,120,24,24:SYMB
OL 241,24,24,48,96,64,32,16,112
5030 SYMBOL 242,12,16,32,39,127,255,255,0:SYMBOL
243,0,24,252,254,255,255,254,0
5040 SYMBOL 244,0,0,0,12,255,0,0,0:SYMBOL 245,4,
2,1,0,255,3,1,0
5050 SYMBOL 246,1,2,140,112,255,252,240,0:SYMBOL
247,0,0,12,255,0,0,0,0
5060 SYMBOL 248,0,0,0,255,3,5,9,18:SYMBOL 249,0,
0,0,255,252,240,0,0
5070 SYMBOL 250,0,0,30,63,127,113,64,96:SYMBOL 2
51,112,120,96,96,224,224,192,64
5080 SYMBOL 252,0,0,30,63,127,113,144,136:SYMBOL
253,0,0,0,128,255,255,31,1
5090 SYMBOL 254,14,15,12,24,240,240,192,32:U=CHR
$(242)+CHR$(243)

```

```

5100 U(0)=CHR$(252)+CHR$(253)+CHR$(254):U(1)=CHR
$(250)+CHR$(253)+CHR$(251)
5110 V(0)=CHR$(247)+CHR$(248)+CHR$(249):V(1)=CHR
$(244)+CHR$(245)+CHR$(246)
5120 PEN 3:FOR I=0 TO 1:LOCATE 1,4+2*I::PRINT "
NOM DU JOUEUR";I+1;
5130 LINE INPUT W(I):W(I)=LEFT$(W(I),10):NEXT
5140 PEN 1:LOCATE 1,20:PRINT "CHOISISSEZ LA DIFF
ICULTE (DE 1 A 3)"
5150 W=INKEY$:A=RND:IF W="" THEN 5150 ELSE E=VAL
(W):IF E<1 OR E>3 THEN 5150
5970 '
5980 '      Decor
5990 '
6000 CLS:PAPER 2:WINDOW 1,40,18,19:CLS:WINDOW 1,
40,24,25:CLS
6010 WINDOW 1,40,1,25:FOR I=18 TO 24 STEP 6:PAPE
R 2:PEN 0:LOCATE 3,I+1
6020 J=-(I=24):PRINT W(J)::LOCATE 13,I+1:PRINT "
: 0":PAPER 0:PEN 1+2*J
6030 LOCATE 38,I:PRINT U:PEN 2*J+1:LOCATE 1,I-1:
PRINT U(J):NEXT:FOR I=1 TO 40
6040 PLOT RND*640,RND*130+250,INT(RND*3)+1:DRAWR
2,0:NEXT
6050 B=1:C=1:G=1:H=0:E=4-E:GOTO 1000

```

VOL DE NUIT

Votre mission : défendre avec une énergie farouche la ville que les soucoupes ennemies ont pour mission, elles, de détruire totalement. Si vous n'agissez pas, vous verrez les vaisseaux adverses traverser l'espace à une vitesse inimaginable et procéder au bombardement systématique de tous les immeubles dont vous devez assurer la garde. Mais vous avez des chances très réelles de repousser les envahisseurs si vos réflexes sont sûrs et... si vous ne vous embrouillez pas dans les commandes de votre laser. Un laser un peu spécial à vrai dire : il a une puissance de feu extraordinaire et l'on n'a jamais vu une soucoupe finir autrement qu'en éclats quand elle a eu affaire à lui. Seulement, comme toutes les machines trop sophistiquées, il est un peu capricieux ; il n'accepte de fonctionner que si l'on respecte les règles qu'il a lui-même, semble-t-il, mystérieusement établies. Voici ces règles : dès que le radar associé au laser détecte l'approche d'une soucoupe, il fait apparaître sur son écran une succession de flèches de directions diverses. Visiblement il s'affole, mais, pour votre part, gardez votre sang-froid, vous n'avez pour l'instant rien à faire. Au bout d'un moment, le radar se calme et une petite soucoupe apparaît alors sur son écran de contrôle. Sa position (à droite ou à gauche) indique de quel côté du ciel va sortir le prochain vaisseau ennemi. Lorsque celui-ci se montre, laissez-le arriver dans la ligne de mire de votre laser et appuyez sur le bouton FIRE. Comment, rien ne se passe ? Ah oui, j'ai oublié de vous dire que pour déclencher la mise à feu d'un missile il faut appuyer sur le bouton FIRE et, en même temps, pousser la manette du côté de la flèche qui est encore dessinée sur l'écran du radar. C'est une drôle de contrainte, non ?

Ne terminons pas l'analyse de ce jeu sans mentionner le comportement assez particulier des soucoupes dans le cas où un missile explose dans leur voisinage immédiat. Endommagées, elles n'ont plus qu'une idée en tête : venir se jeter en zigzaguant sur la base de tir. Mieux vaut, à ce moment-là, dépenser sans compter ses missiles pour essayer de détruire ces soucoupes kamikazées avant qu'elles n'atteignent leur but.

FIN DU JEU _____

Le jeu s'arrête et l'on a perdu dans l'un des cas suivants :

- toute la ville a été bombardée (10 immeubles détruits) ;
- une soucoupe endommagée a réussi à détruire la base de tir dans sa chute ;

- le nombre de missiles est passé à 0 (on dispose de 25 de ceux-ci en début de partie).

On ne peut être déclaré vainqueur que si l'on arrive à toucher 10 soucoupes. La vitesse de déplacement des soucoupes est proportionnelle au niveau de difficulté choisi par l'utilisateur.

MODIFICATIONS POSSIBLES _____

| | Instructions BASIC | N° de ligne |
|--------------------------------|--------------------|-------------|
| Nombre de missiles | MI=25 | 10220 |
| Nombre de soucoupes à détruire | IFC < 10 | 6020 |

JEU SANS MANETTE _____

Les soucoupes sont abattues si l'on appuie, à leur passage, sur la flèche qui s'est dessinée en dernier lieu sur l'écran du radar.

- Ajouter la ligne 1015 suivante :

```
K=240-(K=2)-2*(K=4)-3*(K=8)
```

- Dans les lignes 2010 et 3010, remplacer

```
IF JOY(0)=16+K
```

par :

```
IF INKEY$=CHR$(K)
```

- Dans la ligne 7030, remplacer

```
IF JOY (0) > 15
```

par :


```
IF INKEY$ < > ""
```

- Dans la ligne 8060, remplacer

```
IF JOY(0) < 16
```

par :

```
IF INKEY$ = ""
```



```

940 '
950 '
960 '          VOL     DE     NUIT
970 '
980 '
990 GOTO 10000
1000 K=J1*2+15*(J1=8):PEN 3:LOCATE X(J1),Y(J1):P
RINT U:LOCATE X(K),Y(K)
1010 PRINT U(K):SOUND 1,50,2:FOR I=1 TO 10:NEXT:
J1=K:IF RND<0.9 THEN 1000
1020 Z=INT(RND*2)*2-1:PAPER 2:PEN 0:FOR I=1 TO N
1:LOCATE 18-5*(Z=-1),16
1030 PRINT CHR$(255):SOUND 1,200,2:FOR J=1 TO 80
:NEXT:LOCATE 18-5*(Z=-1),16
1040 PRINT " ":FOR J=1 TO 10:NEXT:NEXT:FOR I=1 T
O 3*NO:NEXT
1050 L=1:M=0:PAPER 0:PEN 1:LOCATE 2-36*(Z=-1),1:
PRINT S
1060 IF INKEY$<>"" THEN 1060 ELSE IF Z=-1 THEN 3
000
1970 '
1980 '          Gauche vers droite
1990 '
2000 FOR X=2 TO 22 STEP 2:LOCATE X,1:PRINT " ":
S:FOR I=1 TO NO:NEXT
2010 IF M=0 THEN IF JOY(0)=16+K THEN GOSUB 5000:
L=1:ON N GOTO 6000,7000
2020 NEXT:M=0:GOTO 4000
2970 '
2980 '          Droite vers gauche
2990 '
3000 FOR X=36 TO 16 STEP -2:LOCATE X,1:PRINT S:"
":FOR I=1 TO NO:NEXT
3010 IF M=0 THEN IF JOY(0)=16+K THEN GOSUB 5000:
L=1:ON N GOTO 6000,7000
3020 NEXT:M=0:X=X+2
3970 '
3980 '          Bombardement
3990 '
4000 IF M OR TEST(16*X-2,82)=0 OR X<8 OR X>32 TH
EN 4030 ELSE M=1:B=B+1
4010 FOR I=2 TO 20:LOCATE X,I:PRINT CHR$(209)+"
":SOUND 1,20+I,2

```

```

4020 LOCATE X,I:PRINT " ":NEXT:IF B=10 THEN 9000
4030 SOUND 1,800,3:FOR I=1 TO 100:NEXT:LOCATE X,
1:PRINT " ":X=X+2*Z
4040 IF X=0 OR X=40 THEN 1000 ELSE LOCATE X,1:PR
INT S:GOTO 4000
4970 '
4980 '     Tir
4990 '
5000 N=0:M=1:ENT 1,70,1,1:ENV 1,2,4,1,1,0,70:SOU
ND 1,30,30,0,1,1:PEN 3
5010 FOR I=14 TO L+1 STEP -2:LOCATE 20,I:PRINT C
HR$(209):FOR K=1 TO 20
5020 NEXT:LOCATE 20,I:PRINT " ":NEXT:GOSUB 5070
:IF P THEN RETURN
5030 IF (X=18 AND Z=1) OR (X=20 AND Z=-1) THEN N
=1:RETURN
5040 FOR I=1 TO 8:PEN I:LOCATE 20,L:PRINT T:NEXT
:PEN 1
5050 IF (X=16 AND Z=1) OR (X=22 AND Z=-1) THEN N
=2:RETURN
5060 IF (X=20 AND Z=1) OR (X=18 AND Z=-1) THEN N
=2:RETURN ELSE RETURN
5070 MI=MI-1:PEN 3:LOCATE 36,25:PRINT MI::IF MI
THEN RETURN
5080 FOR I=1 TO 13:PEN I:LOCATE 27,25:SOUND 1,20
0+10*I,6
5090 IF SQ(1)>127 THEN 5090 ELSE PRINT "MISSILE
: 0":NEXT:GOTO 9000
5970 '
5980 '     Soucoupe touchee
5990 '
6000 ENV 2,14,-1,6:SOUND 7,0,0,0,2,0,30:FOR I=1
TO 3:FOR J=1 TO 12
6010 PEN J:LOCATE 19,L:PRINT T+T:NEXT:NEXT:C=C+1
6020 PEN 3:LOCATE 20,25:PRINT C::P=0:IF C<10 THE
N 1000
6030 FOR I=1 TO 15:PEN I:LOCATE 1,25:SOUND 1,50+
20*I,6
6040 IF SQ(1)>127 THEN 6040 ELSE PRINT "SOUCOUE
S TOUCHEES: 10":
6050 NEXT:W="GAGNE":GOTO 9000
6970 '
6980 '     Chute d'une soucoupe

```

```

6990 '
7000 P=1:X=X-2*(Z=1):IF RND<0.5 THEN 8000 ELSE L
OCATE X,1:PRINT " "
7010 FOR L=2 TO 14:X=X+2*(INT(RND*3)-1):X=X-2*(X
=16)+2*(X=24)
7020 PEN 1:LOCATE X,L:PRINT S:FOR K=1 TO NO:NEXT
7030 IF JOY(0)>15 THEN GOSUB 5000:IF X=20 THEN 6
000
7040 LOCATE X,L:PRINT " ":NEXT:GOTO 8040
7970 '
7980 '     Chute d'une soucoupe (bis)
7990 '
8000 N=0:FOR I=X TO 30 STEP 2:LOCATE I,1:PRINT "
"+S
8010 FOR K=1 TO NO:NEXT:NEXT:FOR L=1 TO 15 STEP
2:M=0:FOR J=32 TO 8 STEP -2
8020 GOSUB 8060:PEN 1:IF N THEN 6000 ELSE NEXT:L
=L+2:M=0:FOR J=8 TO 32 STEP 2
8030 GOSUB 8060:PEN 1:IF N THEN 6000 ELSE IF J<>
20 OR L<>15 THEN NEXT:NEXT
8040 FOR I=1 TO 15:LOCATE 20,15:PEN I:PRINT T:S0
UND 1,200+50*I,5
8050 IF SQ(1)>127 THEN 8050 ELSE NEXT:GOTO 9000
8060 LOCATE J,L:PRINT S:FOR K=1 TO NO:NEXT:IF JO
Y(0)<16 OR M THEN 8080
8070 GOSUB 5000:M=1:IF J=20 THEN N=1
8080 IF INKEY$<>" THEN 8080 ELSE LOCATE J,L:PRI
NT " ":RETURN
8970 '
8980 '     Nouvelle partie
8990 '
9000 PAPER 1:PEN 0:WINDOW 12,28,1,3:CLS:LOCATE 2
,2:PRINT "VOUS AVEZ ";W
9010 IF INKEY$<>" THEN 9010 ELSE WINDOW 1,40,1,
25:PAPER 0
9020 W=INKEY$:IF W="0" THEN RUN ELSE IF W="N" TH
EN CLS:END
9030 LOCATE 6,8:PRINT "UNE AUTRE PARTIE ? ( O ou
N )"
9040 PEN INT(RND*3)+1:FOR I=1 TO 150:NEXT:GOTO 9
020
9970 '
9980 '     Initialisation

```

```

9990 '
10000 MODE 1:INK 0,0:INK 1,16:INK 2,18:INK 3,23:
BORDER 0:PAPER 0
10010 DEFINIT A-P,X-Z:DEFSTR R-W:SYMBOL AFTER 233
10020 SYMBOL 234,3,3,3,7,31,60,120,240:SYMBOL 23
5,0,0,0,128,224,240,120,60
10030 SYMBOL 236,20,65,20,65,20,65,8,34:SYMBOL 2
37,0,128,192,96,48,24,252,252
10040 SYMBOL 238,24,48,96,192,128,0,0,0:SYMBOL 2
39,0,1,3,6,12,24,63,63
10050 SYMBOL 240,24,12,6,3,1,0,0,0:SYMBOL 241,99
,51,27,15,7,3,0,0
10060 SYMBOL 242,24,48,96,192,128,0,0,0:SYMBOL 2
43,0,0,3,7,15,27,51,99
10070 SYMBOL 244,0,0,0,128,192,96,48,24:SYMBOL 2
45,0,32,65,70,45,31,34,0
10080 SYMBOL 246,0,4,130,98,180,248,68,0:SYMBOL
247,192,240,252,255,255,255,255
10090 SYMBOL 248,255,255,255,255,255,255,248,192
:SYMBOL 249,3,15,63,255,255,255,255,255
10100 SYMBOL 250,255,255,255,255,255,255,31,3:SY
MBOL 251,255,255,153,153,153,255,255,255
10110 SYMBOL 252,255,255,159,159,159,255,255,255
:SYMBOL 253,255,255,249,249,249,255,255,255
10120 SYMBOL 254,255,255,255,255,255,255,255,255
:SYMBOL 255,0,129,90,60,126,255
10130 U(1)=CHR$(243)+CHR$(244)+CHR$(8)+CHR$(8)+C
HR$(10)+CHR$(209)+CHR$(32)
10140 U(2)=CHR$(209)+CHR$(32)+CHR$(8)+CHR$(8)+CH
R$(10)+CHR$(241)+CHR$(242)
10150 U(4)=CHR$(239)+CHR$(210)+CHR$(8)+CHR$(8)+C
HR$(10)+CHR$(240)+CHR$(32)
10160 U(8)=CHR$(210)+CHR$(237)+CHR$(8)+CHR$(8)+C
HR$(10)+CHR$(32)+CHR$(238)
10170 U=CHR$(32)+CHR$(32)+CHR$(8)+CHR$(8)+CHR$(1
0)+CHR$(32)+CHR$(32)
10180 T=CHR$(236)+CHR$(236):S=CHR$(245)+CHR$(246
):FOR I=0 TO 3
10190 J=2^I:READ X(J),Y(J):NEXT:DATA 20,17,20,19
,19,18,21,18
10200 PEN 3:LOCATE 1,20:PRINT "CHOISISSEZ LA DIF
FICULTE ( DE 1 A 3 )"
10210 V=INKEY$:Q=RND:IF V="" THEN 10210 ELSE N=V
AL(V):IF N<1 OR N>3 THEN 10210

```

```

10220 NO=(3-N)*30:N1=17-5*N:MI=25:W="PERDU":J1=1
10970 '
10980 '   Decor
10990 '
11000 CLS:FOR I=0 TO 2:LOCATE 2*I+1,7+3*I:PRINT
CHR$(251)+CHR$(247)
11010 FOR J=8+3*I TO 22-I:X=INT(RND*3):LOCATE 2*
I+1,J:PRINT CHR$(251+X)+CHR$(254)
11020 NEXT:LOCATE 2*I+1,J:PRINT CHR$(254)+CHR$(2
48):NEXT:FOR I=0 TO 2
11030 LOCATE 35+2*I,13-3*I:PRINT CHR$(249)+CHR$(
253):FOR J=14-3*I TO 20+I
11040 X=INT(RND*3):LOCATE 35+2*I,J:PRINT CHR$(25
4)+CHR$(251+X):NEXT
11050 LOCATE 35+2*I,J:PRINT CHR$(250)+CHR$(254):
NEXT:MOVE 96,78
11060 DRAWR 448,0,1:PEN 2:FOR I=7 TO 34 STEP 3:L
OCATE I,22:PRINT CHR$(236)
11070 MOVE 16*I-9,46:DRAWR 0,-16,1:MOVER 2,0:DRA
WR 0,16:NEXT:PEN 3:LOCATE 1,25
11080 PRINT "SOUCOUPES TOUCHEES: 0";:LOCATE 27,2
5:PRINT "MISSILLES: 25";
11090 FOR I=8 TO 32 STEP 2:Y=INT(RND*4)+16:PAPER
 0:PEN 1:LOCATE I,Y
11100 PRINT CHR$(214)+CHR$(215):PAPER 3:PEN 0:FO
R J=Y+1 TO 20
11110 LOCATE I,J:PRINT CHR$(128-20*(RND<0.7))+CH
R$(128-20*(RND<0.7)):NEXT
11120 I=I-6*(I=16):NEXT:PAPER 2:WINDOW 18,23,16,
21:CLS:PAPER 0
11130 WINDOW 19,22,17,20:CLS:WINDOW 1,40,1,25:PE
N 1
11140 PEN 1:LOCATE 20,15:PRINT CHR$(234)+CHR$(23
5):GOTO 1000

```

LE COMBAT DES CHEFS

Voici un programme qui va permettre à deux joueurs de comparer la vivacité de leurs réflexes. Chacun est aux commandes d'un avion — via le joystick ou le clavier — et choisit le sens de déplacement de celui-ci. Il a la possibilité de tirer des missiles (en nombre non limité) dans la direction de son adversaire en appuyant sur le bouton FIRE ou sur la touche COPY. A Chaque fois qu'un vaisseau est touché, une phase de jeu se termine, l'ordinateur attribue un point au valeureux vainqueur et démarre une nouvelle partie. Notons que la durée de ce jeu n'est pas décidée par la machine : l'exécution du programme ne sera interrompue que lorsque l'un des joueurs aura enfoncé deux fois de suite la touche ESC.

En dehors du fait que l'on peut être atteint par un missile ennemi, voyons les autres ennuis que l'on peut avoir en chemin : tout d'abord, il existe le risque de percuter l'autre avion ; l'ordinateur n'entre alors pas dans les détails et, plutôt que de rechercher si la responsabilité de la collision appartient à l'un ou à l'autre, décrète nulle la partie en cours. Ce n'est pas trop grave, comme on le voit. Par contre, il vaut mieux éviter le voisinage immédiat des petits losanges rouges qui ont tendance à se générer spontanément sur le téléviseur : ce sont des mines. Naturellement, si un avion en rencontre une, il est détruit et un point est donné au pilote du mobile adverse. Mais, de plus, en explosant, ces mines détruisent tout ce qui est autour d'elles et l'on peut très bien voir son avion réduit en cendres parce que l'adversaire aura profité de notre passage à proximité d'une mine pour tirer un missile sur cette dernière et la mettre à feu. OK ?

Il n'y a rien d'autre à ajouter sur ce programme sinon que les avions se déplacent d'eux-mêmes si l'on n'intervient pas et recherchent automatiquement une nouvelle direction lorsqu'ils se trouvent devant un élément fixe du décor.

FIN DU JEU

De lui-même, le programme boucle continuellement. Les utilisateurs arrêteront donc le jeu quand ils le souhaiteront ; le gagnant sera alors celui des deux dont le score sera le plus élevé.

MODIFICATIONS POSSIBLES

| | Instructions BASIC | N° de ligne |
|-------------------------------|--------------------|-------------|
| Apparition de nouvelles mines | IF RND < 0.1 | 1220 |


JEU SANS MANETTE

Pour utiliser le programme tel qu'il est donné, on doit employer les quatre flèches et la touche COPY du clavier (premier joueur) et la manette de jeu (second joueur). Si l'on ne dispose pas de manette, le second joueur devra se servir des touches 'W' (haut), 'S' (droite), 'A' (gauche) et 'Z' (bas). Le tir sera déclenché en enfonçant la barre d'espacement.

Les modifications à réaliser sont les suivantes :

- Dans la ligne 1000, remplacer INKEY(72) par INKEY(59)
- Dans la ligne 1010, remplacer INKEY(73) par INKEY(71)
- Dans la ligne 1020, remplacer INKEY(74) par INKEY(69)
- Dans la ligne 1030, remplacer INKEY(75) par INKEY(60)
- Dans la ligne 1210, remplacer INKEY(76) par INKEY(47)

LE CŒUR DU PROGRAMME



```

940 '
950 '
960 '      LE      COMBAT      DES      CHEFS
970 '
980 '
990 GOTO 8000
1000 IO=1:PEN 2:IF INKEY(72)=0 THEN C=0:D=-1:Q=0
:GOTO 1040
1010 IF INKEY(73)=0 THEN C=0:D=1:Q=1:GOTO 1040
1020 IF INKEY(74)=0 THEN C=-1:D=0:Q=2:GOTO 1040
1030 IF INKEY(75)=0 THEN C=1:D=0:Q=3
1040 G=V+C:H=W+D:L=A(G,H):IF L<>4 THEN 1060 ELSE
  A(V,W)=4:LOCATE 2*V+1,W+1
1050 PRINT "  ":V=G:W=H:LOCATE 2*V+1,W+1:PRINT Z
(Q):A(V,W)=6:GOTO 1100
1060 IF L=6 THEN 4000 ELSE IF L=8 THEN GOSUB 500
0:GOTO 3000
1070 S=RND:IF S<0.25 THEN C=0:D=-1:Q=0:GOTO 1040
1080 IF S<0.5 THEN C=0:D=1:Q=1:GOTO 1040
1090 IF S<0.75 THEN C=-1:D=0:Q=2:GOTO 1040 ELSE
C=1:D=0:Q=3:GOTO 1040
1100 IF INKEY(9)=0 THEN G=X+E:H=Y+F:M=E:N=F:GOSU
B 2000:IF I THEN 3000
1110 IO=2:PEN 3:IF INKEY(0)=0 THEN E=0:F=-1:R=0:
GOTO 1150
1120 IF INKEY(2)=0 THEN E=0:F=1:R=1:GOTO 1150
1130 IF INKEY(8)=0 THEN E=-1:F=0:R=2:GOTO 1150
1140 IF INKEY(1)=0 THEN E=1:F=0:R=3
1150 G=X+E:H=Y+F:L=A(G,H):IF L<>4 THEN 1170 ELSE
  A(X,Y)=4:LOCATE 2*X+1,Y+1
1160 PRINT "  ":X=G:Y=H:LOCATE 2*X+1,Y+1:PRINT Z
(R):A(X,Y)=6:GOTO 1210
1170 IF L=6 THEN 4000 ELSE IF L=8 THEN GOSUB 500
0:GOTO 3000
1180 S=RND:IF S<0.25 THEN E=0:F=-1:R=0:GOTO 1150
1190 IF S<0.5 THEN E=0:F=1:R=1:GOTO 1150
1200 IF S<0.75 THEN E=-1:F=0:R=2:GOTO 1150 ELSE
E=1:F=0:R=3:GOTO 1150
1210 IF INKEY(76)=0 THEN G=V+C:H=W+D:M=C:N=D:GOS
UB 2000:IF I THEN 3000
1220 IF RND<0.1 THEN 6000 ELSE 1000
1970 '
1980 '      Tir

```

```

1990 '
2000 PEN 1:ZM=CHR$(154-55*(M=0)):I=0
2010 L=A(G,H):IF L<>4 THEN 2030 ELSE LOCATE 2*G+
1,H+1:PRINT ZM
2020 G=G+M:H=H+N:LOCATE 2*(G-M)+1,H-N+1:PRINT "
":GOTO 2010
2030 IF L=2 THEN RETURN ELSE IF L=8 THEN GOSUB 5
000:RETURN ELSE I=3-I0:RETURN
2970 '
2980 '     Avion touche
2990 '
3000 ENV 2,14,-1,6:SOUND 7,0,0,0,2,0,30:X=-X*(I=
1)-V*(I=2)
3010 Y=-Y*(I=1)-W*(I=2):FOR J=1 TO 3:FOR T=1 TO
12:LOCATE 2*X+1,Y+1
3020 PEN T:PRINT CHR$(245)+CHR$(245):NEXT:NEXT:P
(I)=P(I)+1
3030 PEN I+1:LOCATE 10-21*(I=2),25:PRINT P(I)::G
OTO 7000
3970 '
3980 '     Accident
3990 '
4000 ENV 1,25,1,10,3,-3,5:SOUND 1,400,0,4,1:WIND
OW 14,27,10,12:PAPER 3
4010 CLS:FOR I= 1 TO 12:PEN I:LOCATE 2,2:PRINT "
PARTIE NULLE"
4020 FOR J=1 TO 200:NEXT:NEXT:WINDOW 1,40,1,25:P
APER 0:GOTO 7000
4970 '
4980 '     Explosion mine
4990 '
5000 I=0:PEN 1:FOR K=-2 TO 2:FOR J=-2 TO 2:IF A(
G+J,H+K)<>6 THEN 5020
5010 IF G+J=V AND H+K=W THEN I=2:RETURN ELSE I=1
:RETURN
5020 LOCATE 2*(G+J)+1,H+K+1:PRINT CHR$(246)+" ":
SOUND 1,50,1:A(G+J,H+K)=4
5030 NEXT:NEXT:WINDOW 2*G-3,2*G+6,H-1,H+3:CLS:WI
NDOW 1,40,1,25:RETURN
5970 '
5980 '     Nouvelle mine
5990 '
6000 K=INT(RND*14)+3:J=INT(RND*16)+3:IF A(K,J)<>
4 THEN 1000

```

```

6010 PEN 1:LOCATE 2*K+1,J+1:PRINT CHR$(246):A(K,
J)=8:GOTO 1000
6970 '
6980 '      Fin d'une phase
6990 '
7000 GOTO 9050
7970 '
7980 '      Initialisation
7990 '
8000 MODE 1:INK 0,0:INK 1,16:INK 2,10:INK 3,24:P
APER 0:BORDER 0
8010 PEN 3:DEFINT A-R,T-Y:DEFSTR Z:DIM A(19,23)
8020 SYMBOL 245,168,2,144,0,146,8,130,40:SYMBOL
246,0,0,2,7,7,2,0,0
8030 SYMBOL 247,1,1,3,7,13,25,1,2:SYMBOL 248,128
,128,192,224,176,152,128,64
8040 SYMBOL 249,2,1,25,13,7,3,1,1:SYMBOL 250,64,
128,152,176,224,192,128,128
8050 SYMBOL 251,0,1,3,31,31,3,1,0:SYMBOL 252,192
,128,8,240,240,8,128,192
8060 SYMBOL 253,3,1,32,31,31,32,1,3:SYMBOL 254,0
,128,192,248,248,192,128,0
8070 Z(0)=CHR$(247)+CHR$(248):Z(1)=CHR$(249)+CHR
$(250):Z(2)=CHR$(251)+CHR$(252)
8080 Z(3)=CHR$(253)+CHR$(254):ZO=CHR$(143)+CHR$(
143):LOCATE 1,10
8090 LINE INPUT "NOM DU PREMIER JOUEUR ":Z1:Z1=L
EFT$(Z1,9):LOCATE 1,13
8100 LINE INPUT "NOM DU DEUXIEME JOUEUR ":Z2:Z2=
LEFT$(Z2,9)
8970 '
8980 '      Decor
8990 '
9000 CLS:PAPER 1:WINDOW 1,40,1,1:CLS:WINDOW 1,40
,24,24:CLS
9010 WINDOW 1,2,1,24:CLS:WINDOW 39,40,1,24:CLS:W
INDOW 1,40,1,25:PAPER 0:PEN 2
9020 FOR X=0 TO 19:A(X,0)=2:A(X,23)=2:NEXT:FOR Y
=1 TO 22:A(0,Y)=2:A(19,Y)=2
9030 NEXT:LOCATE 1,25:PRINT Z1;:LOCATE 11,25:PRI
NT "0 POINTS";
9040 PEN 3:LOCATE 22,25:PRINT Z2;:LOCATE 32,25:P
RINT "0 POINTS";

```

```
9050 FOR J=1 TO 22:FOR I=1 TO 18:A(I,J)=4:NEXT:N  
EXT:PEN 1  
9060 WINDOW 3,38,2,23:CLS:WINDOW 1,40,1,25:FOR I  
=1 TO 20:X=INT(RND*16)+2  
9070 Y=INT(RND*20)+2:LOCATE 2*X+1,Y+1:PRINT CHR$(  
143)+CHR$(143)  
9080 A(X,Y)=2:NEXT:PEN 2:LOCATE 3,2:PRINT Z(3):P  
EN 3:LOCATE 37,23:PRINT Z(2)  
9090 V=1:W=1:X=18:Y=22:C=1:D=0:Q=3:E=-1:F=0:R=2:  
GOTO 1000
```

***POUR UN CATALOGUE COMPLET
DE NOS PUBLICATIONS***

FRANCE
6-8, Impasse du Curé
75881 PARIS CEDEX 18
Tél. : (1) 42.03.95.95
Télex : 211801

U.S.A.
2344 Sixth Street
Berkeley, CA 94710
Tel. : (415) 848.8233
Telex : 336311

ALLEMAGNE
Vogelsanger. WEG 111
4000 Düsseldorf 30
Postfach N° 30.09.61
Tel. : (0211) 61 80 2-0
Telex : 08588163



Paris • Berkeley • Düsseldorf

Achevé d'imprimer le 12 mai 1986 sur les presses de l'Imprimerie «La Source d'Or»
63200 Marsat - Dépôt légal : 2^e trimestre 1986 - Imprimeur n° 2027

Cet ouvrage contient des programmes de jeux écrits pour les ordinateurs Amstrad CPC 464, 664 et 6128.

Chaque programme est accompagné d'une analyse pédagogique de la structure des phases essentielles et de tableaux résumant la fonction et les valeurs possibles des principales variables. De quoi comprendre, en s'amusant, comment se réalisent des programmes de jeux.

L'AUTEUR

GEORGES FAGOT BARRALY, professeur de mathématiques et d'informatique, est spécialiste de l'utilisation des micro-ordinateurs dans l'enseignement. Il a écrit plusieurs livres de jeux et est également l'auteur de nombreuses publications sur la programmation en assembleur.

0208 0586 98 F



9 782736 102081




Document **numérisé**
avec amour par :

AMSTRAD

CPC 

MÉMOIRE ÉCRITE



<https://acpc.me/>